



AMSTRAD



JEUX DE REFLEXION

GEORGES FAGOT - BARRALY



AMSTRAD

JEUX DE RÉFLEXION

DANS LA MÊME COLLECTION

Amstrad jeux d'action, P. Monsaut
Amstrad premiers programmes, R. Zaks
Amstrad 56 programmes, S.R. Trost
Amstrad guide du BASIC et de l'AMSDOS, J.-L. Gréco/M. Laurent
Amstrad exploré, J. Braga
Amstrad programmation en assembleur, G. Fagot-Barraly
Amstrad guide du graphisme, J. Wynford
Amstrad CP/M 2.2, A. d'Hardancourt
Amstrad astrologie, numérologie, biorythmes, P. Bourgault
Amstrad graphisme en trois dimensions, T. Lachant-Robert
Amstrad Multiplan, Amstrad
Amstrad CP/M plus, A. d'Hardancourt
Amstrad Astrocalc, G. Blanc/P. Destrebecq
Amstrad gagnez aux courses, J.-C. Despoine
Amstrad créer de nouvelles instructions, J.-C. Despoine
Amstrad Locoscript, B. Le Dü
Amstrad mise au point des programmes BASIC, C. Vivier/Y. Jacob
Amstrad jeux en assembleur, E. Ravis
Amstrad programmes en langage machine, S. Webb (à paraître)
Amstrad Logo, A. d'Hardancourt (à paraître)
Amstrad guide du DOS, Amstrad (à paraître)
Amstrad introduction à la programmation en assembleur du Z80, A. d'Hardancourt (à paraître)
Amstrad systèmes d'exploitation, Amstrad (à paraître)
Amstrad routines en assembleur, J.-C. Despoine (à paraître)
Amstrad mieux programmer en assembleur, T. Lachant-Robert (à paraître)

GEORGES FAGOT-BARRALY

AMSTRAD

JEUX DE RÉFLEXION



Paris • San Francisco • Düsseldorf • Londres

SYBEX est indépendant de tout constructeur.

Tous les efforts ont été faits pour fournir dans ce livre une information complète et exacte. Néanmoins, SYBEX n'assume de responsabilités ni pour son utilisation, ni pour les contrefaçons de brevets ou atteintes aux droits de tierces personnes qui pourraient résulter de cette utilisation.

© SYBEX, 1986.

Tous droits réservés. Toute reproduction même partielle, par quelque procédé que ce soit, est interdite sans autorisation préalable. Une copie par xérographie, photographie, film, bande magnétique ou autre, constitue une contrefaçon passible des peines prévues par la loi sur la protection des droits d'auteur.

ISBN 2-7361-0207-X

SOMMAIRE

| | |
|---------------------------|-----|
| Introduction | 7 |
| Puissance 4 | 11 |
| Mistigri | 21 |
| Tours de Hanoi | 31 |
| Solitaire | 39 |
| Hexapion | 47 |
| De visu | 59 |
| L'intrus | 67 |
| Pendu | 75 |
| La dernière étoile | 83 |
| Pousse-pousse | 91 |
| La mare aux diables | 99 |
| Labyrinthe | 109 |
| Nim | 117 |
| Le mot | 125 |
| Les fruits | 135 |
| Le carré chinois | 145 |
| Symétrie | 155 |
| Les brigands | 165 |
| Anagramme | 175 |
| Au suivant | 183 |

I N T R O D U C T I O N

Ce livre regroupe vingt programmes de jeux de réflexion écrits pour les micro-ordinateurs Amstrad. Ces ordinateurs possèdent d'exceptionnelles caractéristiques graphiques et sonores et ont des vitesses d'exécution tout à fait honorables. Dès lors, efficacement programmés, ils ont toutes les qualités pour devenir nos partenaires privilégiés dans le seul domaine où l'informatique arrive à nous faire oublier qu'elle est la fille rieuse des deux monuments de froideur que sont l'électronique et les mathématiques. Je veux parler du domaine de l'ordinateur joueur, de ce monde dans lequel les programmeurs réussissent le tour de force de nous faire croire que l'on est assis devant une machine qui n'est pas tenue de suivre la route qui lui a été tracée. Dans ce monde, l'ordinateur ne fait plus peur et certaines questions n'appellent pas la réponse stéréotypée que l'on a l'habitude d'entendre. Est-ce qu'il arrive à ces machines de prendre des initiatives ? Peuvent-elles, dans leur précipitation, commettre des erreurs ? Sont-elles capables de commenter autrement que par *vrai* ou *faux*, comme dans les programmes que l'on se plaît à appeler "pédagogiques", nos réponses ? Bref, sont-elles capables de faire montre de fraîcheur et de spontanéité ? Pour ma part, je pense que oui, mais je n'empêche personne de penser que chez moi, le cœur l'emporte sur la raison.

Revenons au sujet de cet ouvrage. Vingt programmes de réflexion donc, tous plus passionnants les uns que les autres. De thèmes classiques ou résolument nouveaux, ils exigeront de la part de leurs utilisateurs une vigilance et une faculté de méditation à toute épreuve. Un haut niveau dans le domaine du *self-control* sera le bienvenu. Pas d'inquiétude exagérée toutefois. Il n'y a, dans ce livre, aucun programme dans lequel l'ordinateur soit imbattable. De plus, la plupart du temps, le choix du niveau de difficulté est donné au joueur. Celui-ci pourra donc, quel que soit son âge ou son habileté, trouver "chausure à son pied".

Le lecteur ne manquera pas de s'apercevoir que tous les programmes sont écrits sur le même modèle, respectant tous un certain nombre de règles. Cette homogénéité n'est pas due à une quelconque manie de l'auteur. Elle tient au fait que la conception de ces jeux

s'est réalisée dans un plan d'ensemble et qu'il était alors aisé de définir à l'avance les contraintes que devaient respecter les différents programmes.

Voici la liste de ces règles et leurs raisons d'être :

- La partie du clavier regroupant les quatre flèches de déplacement du curseur est utilisée de façon privilégiée. Il est tellement plus facile d'amener, avec les touches fléchées, un petit repère clignotant dans une case de l'écran que de taper au clavier les coordonnées de ladite case, séparées, de surcroît, par une virgule, non ?
- Le joueur est guidé, à chaque instant, par des phrases qui clignent sur l'écran dans différentes couleurs. Il sait donc toujours quels sont les choix qui s'offrent à lui, précisément.
- S'il arrive que l'on veuille jouer sur une portion de l'image qui n'est pas compatible avec la règle du jeu, l'ordinateur ne nous en tient pas rigueur et fait celui qui n'a rien vu. Tout ce qui pourrait ressembler à un rappel à l'ordre, avec caractères en double grandeur, couleurs agressives et musique tonitruante, a été banni.
- La structure des programmes est linéaire : l'ordre dans lequel sont rangés les numéros de lignes correspond à la façon dont s'enchaînent les différentes actions. Il y a tout d'abord l'initialisation, puis la constitution du décor, la première phase de jeu, la seconde, etc. On ne trouvera pas, dans ce livre, ce que l'on appelle parfois des astuces de programmation qui consistent en réalité à enchevêtrer, à grand renfort de GOTO, diverses lignes de programme, sous prétexte que le listing final sera moins long d'une dizaine d'instructions.
- Les programmes occupent tous à peu près le même volume mémoire, 2 à 3 kilo octets sans les REM. Cet ordre de grandeur, un peu élevé quand il est relatif à des jeux de réflexion, permet de ne pas sacrifier ce qui est essentiel dans un jeu : le décor. Un effort tout particulier a été fait pour rendre la présentation de ces jeux la plus attrayante possible. Cela prend de la place dans les programmes bien sûr, mais le résultat est à la hauteur de la peine que l'on s'est donnée.
- Tous les listings font apparaître une abondante documentation. Chaque sous-programme étant précédé d'un commentaire, la faculté est offerte aux adeptes de la programmation de retrouver sans difficulté les phases de jeu qu'ils souhaitent analyser ou modifier.

Notons qu'aucun branchement n'est "réalisé sur une REM" et que, de ce fait, toutes les lignes dont le numéro est suivi d'une apostrophe peuvent très bien ne pas être tapées au clavier.

- Chaque jeu est accompagné, c'est la moindre des choses, d'un mode d'emploi détaillé. S'y ajoutent des indications concernant la manière dont les parties se terminent ainsi qu'une analyse pédagogique des phases essentielles du programme BASIC. Une liste des principales variables est fournie et, pour chacune de celles-ci, on indique la fonction et le domaine des valeurs qu'elle peut prendre. Cette liste constituera une aide précieuse pour les lecteurs qui voudront comprendre selon quel mécanisme fonctionne tel ou tel programme.
- Des idées sur les modifications possibles des jeux sont toujours données, pour les utilisateurs qui veulent aller plus loin. Ils pourront ainsi, en fonction de leur inspiration, adapter les programmes de ce livre et en faire des versions plus personnelles.
- Le dernier point a son importance : le noir et le bleu foncé ont été choisis comme couleurs de fond pour tous les jeux ; c'est plus reposant pour la vue des joueurs et cela accroît la durée de vie des téléviseurs.

Voilà. Et maintenant, place aux jeux !

PUISSANCE 4

A tour de rôle, chacun des deux joueurs va devoir placer un pion de sa couleur sur l'une des cases du damier dessiné sur l'écran. Le but est d'aligner quatre de ses jetons horizontalement, verticalement ou encore en diagonale.

Ce jeu, qui ressemble fort au célèbre *Morpion* tant apprécié des lycéens, s'en distingue pourtant par une particularité plutôt contraignante : on ne peut porter son choix que sur certaines cases vides, celles qui sont tout à fait en bas de la grille ou celles qui se trouvent au-dessus d'une case contenant déjà elle-même un pion. En cela, ce programme respecte le principe du jeu vendu sous le nom de *Puissance 4*, jeu qui consiste à introduire un pion en haut d'une colonne et à le laisser descendre jusqu'à ce qu'il rencontre un obstacle.

Bien, allez-y, faites porter votre choix ¹ sur une case. L'ordinateur se charge de vérifier si elle n'est pas déjà occupée et si c'est bien la plus basse possible. Inutile donc d'essayer de retenir une case qui n'obéirait pas à la règle du jeu, le programme ferait la sourde oreille. Par contre, si votre choix apparaît comme possible, ce même programme fera apparaître en haut du téléviseur un jeton de votre couleur et le fera dégringoler jusqu'à la case choisie.

Si d'aventure ce coup vous permet d'emporter la partie, l'ordinateur indiquera que vous êtes le vainqueur et proposera de rejouer. Sinon, il préviendra votre adversaire, par un message clignotant, que c'est maintenant à lui de désigner une case.

Remarquez qu'il a été prévu la possibilité, si l'on est seul, de jouer contre l'ordinateur. Ce dernier attend alors que l'on ait fait sienne une case puis il choisit une colonne, y fait tomber un pion de sa couleur et nous redonne la main. Naturellement, il vérifie à chaque instant s'il n'y a pas un vainqueur. Dans le cas du jeu en solitaire, on peut déterminer le niveau de difficulté de la partie. Le niveau 1 correspond à un jeu facile, contre une machine qui n'est pas toujours suffisamment concentrée, et le niveau 3 à un jeu plus compliqué, contre une machine très attentive (mais pas imbattable toutefois).

1. Il suffit de déplacer, avec les quatre flèches, la petite étoile jaune clignotante jusqu'à ce qu'elle se trouve dans la case voulue. A ce moment-là, on appuie sur la barre d'espacement.

■ FIN DU JEU

Le jeu se termine lorsque l'un des joueurs (ou l'ordinateur) a réussi à aligner quatre de ses pions ou lorsque la totalité des cases a été remplie. Un message indiquant que la partie est déclarée nulle est, dans ce dernier cas, affiché sur l'écran.

Le niveau de difficulté (variable NV dont la valeur est définie à la ligne 190) intervient dans le programme à la ligne 4020. Au niveau 1, par exemple, l'ordinateur joue correctement six fois sur dix ($RND < 0.6$) et joue... au hasard les quatre autres fois.

■ PHASES ESSENTIELLES

Sous-programme 2000

Le joueur choisit une case.

2000 à 2070 L'ordinateur attend qu'une case soit désignée par le joueur et change de temps en temps la couleur du message invitant ce joueur à se décider. Quand la barre d'espacement est enfoncée, il calcule les coordonnées de la case choisie et vérifie la validité de l'essai.

2080 Si le coup est possible, un pion dégringole dans la grille jusqu'à la case voulue (GOSUB 6000). Le nombre de pions alignés avec celui qui vient d'être joué est alors calculé (GOSUB 5000) et, si ce nombre est égal à 4, un branchement à la ligne 7000 (fin du jeu) est ordonné.

| Variables | Fonction | Valeurs possibles |
|-----------|---|-------------------|
| Z | Couleur du message | 0 ou 2 |
| A\$ | Touche enfoncée | Touche quelconque |
| XP et YP | Coordonnées de l'étoile clignotante | 5 à 29 et 4 à 24 |
| B et C | Coordonnées de la case choisie | 0 à 6 et 0 à 5 |
| D(B) | Ordonnée de la case jouable dans la colonne B | 0 à 6 |

| Variables | Fonction | Valeurs possibles |
|-----------|---------------------------------|-------------------|
| BO | Abscisse de la case choisie | 0 à 6 |
| J | Couleur d'un pion | 0 ou 2 |
| H | Nombre maximum de pions alignés | 1 à 4 |

Sous-programme 4000

L'ordinateur choisit une case.

4000 à 4030 La phrase "Chut... je réfléchis" apparaît sur l'écran et l'ordinateur se lance à la recherche d'une case intéressante... sauf s'il se déconcentre (approximativement quatre fois sur dix au niveau 1 et deux fois sur dix au niveau 2).

4040 à 4080 Le programme prend les colonnes une par une et détermine le nombre de pions qu'alignerait le joueur s'il choisissait telle ou telle colonne. Il ne lui reste plus ensuite qu'à placer le pion de l'ordinateur dans la colonne que le joueur n'aurait pas manqué de considérer comme la plus intéressante.

4090 à 4130 Après la défensive, voici l'offensive : le programme, s'il estime que les cases de l'adversaire ne le menacent pas directement, cherche à aligner ses propres pions.

4140 à 4170 Le pion de l'ordinateur est dessiné (GOSUB 6000) et il ne reste qu'à voir si, par hasard, le coup joué ne conduirait pas l'ordinateur à être déclaré vainqueur (GOSUB 5000 avec E=3)

| Variables | Fonction | Valeurs possibles |
|-----------|---|--|
| B | Numéro de la colonne inspectée | 0 à 6 |
| E | Indicateur de la case occupée sur la grille | 0 : case vide 1 : 1 ^{er} joueur 2 : 2 ^e joueur 3 : ordinateur |

| Variables | Fonction | Valeurs possibles |
|-----------|--|-------------------|
| U et V | Numéros de la colonne la plus intéressante | 0 à 6 |
| S(B) | Nombre de pions qu'alignerait le joueur en jouant dans la colonne B | 1 à 4 |
| T(B) | Nombre de pions qu'alignerait l'ordinateur en jouant dans la colonne B | 1 à 4 |
| G | Nombre total de coups joués | 0 à 42 |

Sous-programme 5000

Recherche de la valeur d'une case.

5000 à 5070 On attribue à une case une valeur égale au nombre maximal de pions qui seraient alignés si l'on jouait justement dans cette case.

| Variables | Fonction | Valeurs possibles |
|-----------|---|-------------------|
| J | Abscisse de la case étudiée | 0 à 6 |
| K | Ordonnée de la case étudiée | 0 à 5 |
| L | Valeur ajoutée à l'abscisse pour passer à une case contiguë | - 1 , 0 , 1 |
| M | Idem pour l'ordonnée | - 1 , 0 , 1 |
| P | Compteur des cases alignées | 1 à 4 |
| R | Drapeau permettant de repartir en sens contraire | - 1 et 1 |
| F(K,J) | Indications sur le propriétaire de la case analysée | Idem E |

■ MODIFICATIONS POSSIBLES

Augmentation du nombre de colonnes

Il n'y aurait pas de difficulté majeure à faire passer le nombre de colonnes à 8 ou même à 9 ; seuls quelques points de détail sont à modifier.

Instauration d'un chronomètre

Il pourrait être intéressant de limiter le temps de réflexion imparti au joueur. Pas trop, toutefois, si l'on veut garder une chance raisonnable de battre l'ordinateur.

```
50 '
60 '
70 '          PUISSANCE  QUATRE
80 '
90 '
100 DEFINT A-Z:MODE 1:INK 0,0:INK 1,10:INK 2,18:INK
3,24:BORDER 0:PAPER 0
110 P$=CHR$(214)+CHR$(215)+CHR$(8)+CHR$(8)+CHR$(10)+
CHR$(213)+CHR$(212)
120 CLS:PEN 2:LOCATE 2,5:PRINT "CONTRE L'ORDINATEU
R :  TAPEZ 1"
130 LOCATE 2,9:PRINT "AVEC DEUX JOUEURS :  TAPEZ 2
"
140 A$=INKEY$:IF A$="" THEN 140 ELSE A=VAL(A$)
150 IF A<1 OR A>2 THEN 140 ELSE PEN 3:PRINT:PRINT
:FOR J=1 TO A
160 PRINT "NOM DU JOUEUR";J;:LINE INPUT A$(J)
170 A$(J)=LEFT$(A$(J),9):PRINT:NEXT:IF A=2 THEN 1000
180 PEN 2:LOCATE 2,20:PRINT "CHOISISSEZ LA DIFFICU
LTE (DE 1 A 3)"
190 A$=INKEY$:IF A$="" THEN 190 ELSE NV=VAL(A$)

200 IF NV<1 OR NV>3 THEN 190
970 '
980 '          Decor
990 '
1000 CLS:PAPER 1:WINDOW 2,29,1,24:CLS:WINDOW 30,40,1,
5
1010 CLS:WINDOW 1,40,1,25:PAPER 0:FOR I=16 TO 464 S
TEP 64
1020 MOVE I,16:DRAW I,399,0:NEXT:FOR I=15 TO 399 ST
EP 64
1030 MOVE 16,I:DRAW 464,I:NEXT:PEN 2:LOCATE 31,2:PRI
NT "PUISSANCE"
1040 LOCATE 35,4:PRINT "4":PEN 3:LOCATE 2,25
1060 PRINT A$(1);"  CONTRE  ";
```



```

1070 IF A=1 THEN PRINT "L'ORDINATEUR";ELSE PRINT
A$(2);
1080 IF A=1 THEN PEN 2:LOCATE 32,23:PRINT "NIVEAU
";NV
1090 FOR I=0 TO 6:D(I)=6:NEXT:XP=17:YP=24:B=3:C=5

1970 '
1980 '      Premier joueur
1990 '
2000 E=1:I=100:Z=0
2010 A$=INKEY$:IF A$<>" " THEN 2050 ELSE I=I+1:IF
I<100 THEN 2010
2020 Z=2-Z:PEN Z:PAPER 0:LOCATE 32,12:PRINT "A VOU
S"
2030 LOCATE 31,14:PRINT A$(1):I=0:PAPER 1:PEN 1+Z
2040 LOCATE XP,YP:PRINT "*":PAPER 0:GOTO 2010
2050 IF A$<>CHR$(32) THEN GOSUB 7000:GOTO 2010
2060 PAPER 0:LOCATE 32,12:PRINT SPACE$(7):LOCATE 3
1,14:PRINT SPACE$(9)
2070 LOCATE XP,YP:PAPER 1:PRINT " ":IF D(B)<>C+1 TH
EN I=100:GOTO 2010
2080 B0=B:J=2:GOSUB 6000:GOSUB 5000:IF H=4 THEN 8
000
2970 '
2980 '      Deuxieme joueur
2990 '
3000 IF A=1 THEN 4000 ELSE E=2:I=100:Z=0
3010 A$=INKEY$:IF A$<>" " THEN 3050 ELSE I=I+1:IF
I<100 THEN 3010
3020 Z=2-Z:PEN Z:PAPER 0:LOCATE 32,12:PRINT "A VOU
S"
3030 LOCATE 31,14:PRINT A$(2):I=0:PAPER 1:PEN 1+Z
3040 LOCATE XP,YP:PRINT "*":PAPER 0:GOTO 3010

3050 IF A$<>CHR$(32) THEN GOSUB 7000:GOTO 3010
3060 PAPER 0:LOCATE 32,12:PRINT SPACE$(7):LOCATE 3
1,14:PRINT SPACE$(9)
3070 LOCATE XP,YP:PAPER 1:PRINT " ":IF D(B)<>C+1 TH
EN I=100:GOTO 3010
3080 J=0:GOSUB 6000:GOSUB 5000:IF H=4 THEN 8000

3090 G=G+2:IF G=42 THEN 8000 ELSE 2000
3970 '

```

```

3980 '          L'ordinateur joue
3990 '
4000 PAPER 0:PEN 3:LOCATE 31,12:PRINT "CHUT.. JE"

4010 LOCATE 31,14:PRINT "REFLECHIS"
4020 IF RND<0.4+0.2*NV THEN 4040 ELSE B=INT(RND*7
)
4030 IF D(B) THEN FOR I=1 TO 4000:NEXT:GOTO 4140
4040 E=1:U=0:FOR B=0 TO 6
4050 IF D(B)=0 THEN S(B)=0:GOTO 4080
4060 D(B)=D(B)-1:GOSUB 5000:S(B)=H:D(B)=D(B)+1
4070 IF S(B)>S(U) THEN U=B
4080 NEXT:B=U
4090 E=3:V=0:FOR B=0 TO 6
4100 IF D(B)=0 THEN T(B)=0:GOTO 4130
4110 D(B)=D(B)-1:GOSUB 5000:T(B)=H:D(B)=D(B)+1
4120 IF T(B)>T(V) THEN V=B
4130 NEXT:IF S(U)>T(V) THEN B=U ELSE B=V
4140 LOCATE 31,12:PRINT SPACE$(9):LOCATE 31,14
4150 PRINT SPACE$(9):E=3:J=0
4160 GOSUB 6000:GOSUB 5000:IF H=4 THEN 8000
4170 G=G+2:IF G=42 THEN 8000 ELSE B=B0:GOTO 2000

4970 '
4980 '          Recherche valeurs cases
4990 '
5000 RESTORE:H=1:FOR I=1 TO 4:P=1:R=1:READ L,M
5010 J=B:K=D(B)
5020 J=J+L:K=K+M:IF J<0 OR J>6 OR K<0 THEN 5050
5030 IF K>5 OR F(J,K)<>E THEN 5050
5040 P=P+1:GOTO 5020
5050 R=-R:IF R=-1 THEN L=-L:M=-M:GOTO 5010
5060 IF P>H THEN H=P
5070 NEXT:RETURN:DATA 1,0,1,1,0,1,-1,1
5970 '
5980 '          Affichage pion
5990 '
6000 D(B)=D(B)-1:PAPER 1:K=INT(RND*6)+2:FOR I=0 TO D
(B)
6010 PEN J:LOCATE 4*B+3,4*I+2:PRINT P$:SOUND 1,50*
K+5*I,10
6020 IF SQ(1)>127 THEN 6020 ELSE IF I=D(B) THEN 60
40

```

```

6030 PEN 1:LOCATE 4*B+3,4*I+2:PRINT P$
6040 NEXT:F(B,D(B))=E:RETURN
6970 '
6980 '      Deplacement etoile
6990 '
7000 AS=ASC(A$):IF AS<240 OR AS>243 THEN RETURN
7010 PAPER 1:LOCATE XP,YP:PEN 1:PRINT "*"
7020 IF AS=242 AND XP>5 THEN XP=XP-4:B=B-1
7030 IF AS=243 AND XP<29 THEN XP=XP+4:B=B+1
7040 IF AS=240 AND YP>4 THEN YP=YP-4:C=C-1
7050 IF AS=241 AND YP<24 THEN YP=YP+4:C=C+1
7060 LOCATE XP,YP:PEN 3:PRINT "^":PAPER 0:RETURN
7970 '
7980 '      Fin du jeu
7990 '
8000 WINDOW 29,40,11,15:PAPER 2:CLS
8010 WINDOW 1,40,1,25:PEN 0:IF H<4 THEN 8040
8020 IF E=3 THEN A$="L'ORDINATEUR" ELSE A$=A$(E)
8030 LOCATE 29,12:PRINT A$:LOCATE 31,14:PRINT "A G
AGNE":GOTO 8050
8040 LOCATE 29,13:PRINT "PARTIE NULLE"
8050 FOR I=1 TO 3000:NEXT:PAPER 0
8060 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>""
THEN CLS:END
8070 I=INT(RND*3)+1:PEN I:LOCATE 2,16
8080 PRINT "UNE AUTRE PARTIE: TOUCHE R":FOR I=1
TO 200:NEXT:GOTO 8060

```

MISTIGRI

Après que l'ordinateur nous ait demandé le nombre de joueurs ainsi que leurs noms et nous ait fait choisir le niveau de difficulté, 25 carrés apparaissent en bleu sur l'écran. Ils représentent des cartes au dos desquelles sont dessinées 13 figurines, les 12 premières sont reproduites en double et la dernière n'est autre que le ... Mistigri.

Pour l'instant, toutes ces cartes sont à l'envers et, de ce fait, il nous est impossible de deviner ce qui se cache au dos de l'une ou de l'autre. Choisissons ¹ une première case comme nous y invite un message clignotant sur l'écran. Immédiatement, la carte se retourne et un petit dessin apparaît. Puisque le programme nous demande alors d'indiquer une autre case, exécutons-nous. Un nouveau graphique se dessine dans la deuxième case. Supposons d'abord que les deux dessins obtenus soient identiques : l'ordinateur accorde immédiatement un point au joueur perspicace et engage le joueur suivant à tenter sa chance. Les deux cartes trouvées resteront visibles durant toute la partie et les désigner ne servira plus à rien.

Autre supposition : les deux cases retournées n'ont pas le même contenu. Aucun point n'est accordé et le tour devra passer au joueur suivant. Auparavant, l'ordinateur aura pris la peine d'effacer les petites figurines que l'on venait de faire apparaître. Si vous avez eu le temps de graver dans votre mémoire ces deux dessins et bien sûr leurs emplacements, vous aurez pris une option sur la victoire. Toute la difficulté de ce jeu consiste en effet à essayer de se souvenir que dans telle case il y a, par exemple, un écureuil ; dans telle autre se trouve une girafe, etc. Si, au cours du jeu, on tombe sur le deuxième écureuil, ou la deuxième girafe, on sera en mesure de reconstituer la paire.

Passons maintenant au Mistigri. Ne vous avisez surtout pas de retourner la carte sous laquelle il se cache. Cela ne le ferait que ricaner et vous verriez votre compte débité de deux points. D'accord, direz-vous, nous nous ferons prendre une fois mais pas deux ; dès qu'elle sera découverte, la case maudite ne sera plus retenue par personne, c'est l'évidence. Détrompez-vous, le Mistigri a le pouvoir de changer de cachette. Par la même occasion, il bouscule un peu la façon dont sont rangées les autres cartes. S'il prend la place d'un petit personnage, il faut bien que celui-ci se retrouve ailleurs, n'est-ce pas ?

1. Il suffit de déplacer, avec les quatre flèches, la petite étoile verte clignotante jusqu'à ce qu'elle se trouve dans la case voulue. A ce moment-là, on appuie sur la barre d'espacement.

Cela dit, si vous êtes un peu attentif, vous remarquerez que le Mistigri, lorsqu'il change de coin, ne peut s'empêcher de se faire remarquer. Il apparaît puis disparaît sur différentes cases de la grille. Cela vous permettra de savoir, approximativement, dans quelle région de l'écran il se cache.

■ FIN DU JEU

Il faut, pour arriver au terme d'une partie, que les 12 paires soient retrouvées ; 24 carrés de la grille contiennent à ce moment-là un dessin, la 25^e case étant celle du Mistigri. L'ordinateur affiche alors le nom du (ou des) gagnant(s) et propose une nouvelle partie. Le gagnant est, cela va de soit, celui qui a le plus de points, c'est-à-dire celui qui a découvert le plus grand nombre de couples identiques et qui n'est pas tombé trop souvent sur le Mistigri.

C'est en fonction du niveau de difficulté qu'est réglé le temps pendant lequel les dessins restent visibles avant d'être effacés et qu'est déterminée la fréquence des déplacements du Mistigri.

■ PHASES ESSENTIELLES

Sous-programme 2000

Les cases sont mélangées.

- 2000 Les valeurs 1 à 12 sont attribuées aux 12 premières cases, la valeur 13 à la suivante (Mistigri), puis les valeurs 1 à 12 sont à nouveau attribuées aux 12 dernières cases. Les paires sont donc formées mais pas encore mélangées.
- 2010 à 2070 On échange 40 fois de suite la case de coordonnées (X,Y) avec la case de coordonnées (Z,T). Ces 4 variables sont évidemment tirées au sort à chaque étape. En même temps, un message d'attente s'imprime dans diverses couleurs sur l'écran et les différentes figurines se dessinent puis s'effacent sur la grille, tout cela ne servant qu'à "faire joli".
- 2080 Les coordonnées du Mistigri sont recherchées dans le damier.

| Variables | Fonction | Valeurs possibles |
|---------------|---|-------------------|
| B(I,J) | Valeur de la case (I,J) | 1 à 13 |
| K | Valeur successivement attribuée aux cases | 1 à 13 |
| X , Y , Z , T | Coordonnées des cases à échanger | 1 à 5 |
| U | Numéro d'un dessin G\$ | 1 à 13 |
| V | Numéro d'une couleur | 1 à 3 |
| W | Numéro de la ligne pour B\$ | 11 ou 16 |
| B\$ | Message "Chut , je réfléchis" | Chaîne caractères |
| R\$ | Effacement des dessins G\$ | Chaîne caractères |
| C | Variable intermédiaire pour l'échange | 1 à 5 |
| XM , YM | Coordonnées du Mistigri | 1 à 5 |

Sous-programme 3000

Le joueur choisit deux cases.

3000 à 3070 Une première case est choisie et la vérification de la validité de l'essai est réalisée. Le dessin caché au dos de la carte désignée apparaît.

3080 à 3140 C'est au tour d'une deuxième carte d'être affichée.

3150 à 3200 Dans le cas où le Mistigri a été découvert, le programme le fait savoir bruyamment et retire deux points au joueur maladroit.

| Variables | Fonction | Valeurs possibles |
|-----------|-------------------------------------|-------------------|
| A | Nombre de joueurs | 1 à 5 |
| D | Numéro du joueur concerné | 1 à 5 |
| A\$ | Touche enfoncée | Touche quelconque |
| A\$(D) | Nom du joueur concerné | Chaîne caractères |
| XP et YP | Coordonnées de l'étoile clignotante | 6 à 26 et 5 à 25 |

| Variables | Fonction | Valeurs possibles |
|-----------|---|-------------------|
| E | Compteur permettant l’affichage régulier du message “Désignez une case” | 0 à 60 |
| F , G | Coordonnées de la case choisie | 1 à 5 |
| Z | Couleur du dessin qui apparaît | 1 ou 2 |
| B(F,G) | Valeur de la case (F,G) | 1 à 13 |
| FO et GO | Coordonnées de la première case | 1 à 5 |
| N(D) | Score du joueur numéro D | max : 12 |

Sous-programme 5000

Déplacements du Mistigri.

5000 à 5080 Une partie de la grille (9 cases) est choisie aléatoirement et l’ordinateur vérifie s’il peut y cacher le Mistigri. Si c’est le cas, il fait apparaître puis disparaître le dessin de celui-ci dans diverses cases environnantes.

| Variables | Fonction | Valeurs possibles |
|-----------|---|-------------------|
| X , Y | Coordonnées du coin supérieur gauche de la surface à étudier puis coordonnées des cases où se dessinent les Mistigris | 1 à 3 puis 1 à 5 |
| K | Drapeau indiquant qu’une case libre a été trouvée | 0 ou 1 |
| C | Variable intermédiaire permettant l’échange | 1 à 5 |

■ MODIFICATIONS POSSIBLES

Augmentation du nombre de cases

Cela peut se réaliser facilement à condition de faire en sorte que le total des cartes soit un nombre impair.

Augmentation du nombre de Mistigris

Le passage d'un Mistigri à trois devra se faire en attribuant la valeur 13 à trois cases du damier.

```
50 '
60 '
70 '          MISTIGRI
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    O:PAPER 0:SYMBOL AFTER 203
110 SYMBOL 204,3,3,3,3,1,1,63,63:SYMBOL 205,192,192,1
    92,192,128,128,252,252
120 SYMBOL 206,3,3,3,3,12,12,48,48:SYMBOL 207,192,192
    ,192,192,48,48,12,12
130 SYMBOL 208,4,6,7,7,7,7,7,7:SYMBOL 209,0,0,0,128,19
    2,224,240,248
140 SYMBOL 210,7,7,4,255,127,63,31,15:SYMBOL 211,252,
    254,0,255,254,252,248,240
150 SYMBOL 212,0,0,0,255,4,4,63,235:SYMBOL 213,0,0,0,2
    24,0,0,0,129
160 SYMBOL 214,127,63,31,0,0,0,0,0:SYMBOL 215,255,129
    ,0,0,0,0,0,0
170 SYMBOL 216,24,56,48,48,24,139,103,31:SYMBOL 217,
    24,28,12,12,24,209,230,248
180 SYMBOL 218,7,63,71,135,27,32,64,64:SYMBOL 219,22
    4,252,226,225,216,4,2,2
190 SYMBOL 220,15,15,7,3,1,97,145,9:SYMBOL 221,224,22
    4,192,128,0,3,4,8
200 SYMBOL 222,5,5,3,1,1,1,1,1:SYMBOL 223,16,32,32,64,
    64,128,0,0
210 SYMBOL 224,32,80,136,136,4,4,2,2:SYMBOL 225,8,20,
    36,36,64,64,128,128
220 SYMBOL 226,3,7,5,7,7,7,3,3:SYMBOL 227,128,192,64,1
    92,192,192,128,128
230 SYMBOL 228,15,31,63,127,255,63,63,38:SYMBOL 229,
    248,248,252,254,255,252,252,100
240 SYMBOL 230,38,38,38,62,62,62,255,255:SYMBOL 231,
    100,100,100,124,124,124,255,255
250 SYMBOL 232,0,0,0,0,0,15,63,255:SYMBOL 233,0,0,0,0,
    0,1,194,254
```

```

260 SYMBOL 234,255,63,15,0,0,0,0:SYMBOL 235,254,194
,1,0,0,0,0,0
270 SYMBOL 236,60,60,195,195,195,195,3,3:SYMBOL 237,
0,0,0,12,14,15,12,12
280 SYMBOL 238,12,12,48,48,48,48,15,15:SYMBOL 239,60,
63,62,60,252,252,252,248
290 SYMBOL 240,0,0,0,0,0,0,0:SYMBOL 241,14,12,8,24,1
6,48,96,96
300 SYMBOL 242,0,32,31,31,22,16,16,16:SYMBOL 243,192,
192,128,128,128,128,128,128
310 SYMBOL 244,0,0,48,240,48,48,63,31:SYMBOL 245,0,0,
0,0,0,2,252,252
320 SYMBOL 246,13,12,12,12,12,12,0,0:SYMBOL 247,236,1
2,12,12,12,12,0,0
330 SYMBOL 248,0,0,0,1,3,6,255,255:SYMBOL 249,48,96,1
92,128,1,2,252,252
340 SYMBOL 250,6,3,1,0,0,0,0:SYMBOL 251,2,1,128,192,
96,48,0,0
350 SYMBOL 252,32,32,127,127,231,231,255,255:SYMBOL
253,4,4,254,254,231,231,255,255
360 SYMBOL 254,127,63,56,24,31,15,7,3:SYMBOL 255,254,
252,28,24,248,240,224,192
370 K=204:DIM G$(13):FOR I=1 TO 13
380 G$(I)=CHR$(K)+CHR$(K+1)+CHR$(8)+CHR$(8)+CHR$(10)+
CHR$(K+2)+CHR$(K+3)
390 K=K+4:NEXT:OPEN 3:LOCATE 2,5:PRINT "NOMBRE DE J
OUEURS (MAX 5)"
400 A$=INKEY$:IF A$="" THEN X=RND:GOTO 400
410 A=VAL(A$):IF A<1 OR A>5 THEN 400 ELSE PEN 2
420 FOR I=1 TO A:LOCATE 2,6+2*I:PRINT "NOM DU JOU
EUR";I;
430 LINE INPUT A$(I):A$(I)=LEFT$(A$(I),9):NEXT
440 OPEN 3:LOCATE 2,20:PRINT "CHOISISSEZ LA DIFFICU
LTE (DE 1 A 3)"
450 A$=INKEY$:IF A$="" THEN 450 ELSE NV=VAL(A$)
460 IF NV<1 OR NV>3 THEN 450 ELSE B$="CHUT ... JE
MELANGE LES DESSINS"
470 R$=" "+CHR$(8)+CHR$(8)+CHR$(10)+" ":XP=16:YP=
15:F=3:G=3
970 '
980 ' Decor
990 '
1000 CLS:PAPER 1:FOR I=3 TO 23 STEP 5:FOR J=2 TO

```

```

22 STEP 5
1010 WINDOW I,I+3,J,J+3:CLS:NEXT:NEXT:WINDOW 29,40,2,
4
1020 CLS:WINDOW 1,40,1,25:PEN 2:LOCATE 31,3:PRINT "M
ISTIGRI"
1030 PEN 3:PAPER 0:FOR I=1 TO A:LOCATE 28,19+I
1040 PRINT A$(I);TAB(38);"O";:NEXT
1970 '
1980 '      Melange des dessins
1990 '
2000 K=1:FOR I=1 TO 5:FOR J=1 TO 5:B(I,J)=K:K=K+1:IF
K=14 THEN K=1
2010 NEXT:NEXT:FOR I=1 TO 40:X=INT(RND*5)+1
2020 Y=INT(RND*5)+1:Z=INT(RND*5)+1:T=INT(RND*5)+1
2030 U=INT(RND*13)+1:V=INT(RND*3)+1:W=5*INT(RND*2)+
11
2040 PEN 2:PAPER 1:LOCATE 5*X-1,5*Y-2:PRINT G$(U);:
C=B(X,Y)
2050 B(X,Y)=B(Z,T):B(Z,T)=C:PEN V:PAPER 0:LOCATE 1,W:PR
INT B$:SOUND 1,200,2
2060 IF SQ(1)>127 THEN 2060 ELSE PAPER 1:LOCATE 5
*X-1,5*Y-2:PRINT R$;:NEXT
2070 PAPER 0:LOCATE 1,11:PRINT SPACE$(32):LOCATE 1,
16:PRINT SPACE$(32)
2080 FOR I=1 TO 5:FOR J=1 TO 5:IF B(I,J)=13 THEN X
M=I:YM=J ELSE NEXT:NEXT
2090 PEN 2:LOCATE 28,10:PRINT "DESIGNEZ UNE":LOCAT
E 37,12:PRINT "CASE"
2970 '
2980 '      Un joueur joue
2990 '
3000 E=60:Z=1:D=D+1:IF D=A+1 THEN D=1
3010 A$=INKEY$:IF A$<>"" THEN 3050 ELSE E=E+1:IF
E<60 THEN 3010
3020 Z=3-Z:PEN Z:PAPER 0:LOCATE 28,12:PRINT "PREMIE
RE"
3030 LOCATE 28,19+D:PRINT A$(D);:PAPER 1
3040 LOCATE XP,YP:PRINT "*";:PAPER 0:E=0:GOTO 3010
3050 IF A$<>CHR$(32) THEN GOSUB 6000:GOTO 3010
3060 IF B(F,G)=0 THEN 3010 ELSE PEN 2:PAPER 1
3070 LOCATE 5*F-1,5*G-2:PRINT G$(B(F,G));:FO=F:GO=G:I
F B(F,G)=13 THEN 3160
3080 A$=INKEY$:IF A$<>"" THEN 3120 ELSE E=E+1:IF

```

```

E<60 THEN 3080
3090 Z=3-Z:PEN Z:PAPER 0:LOCATE 28,12:PRINT "DEUXIE
ME"
3100 LOCATE 28,19+D:PRINT A$(D);:PAPER 1
3110 LOCATE XP,YP:PRINT "*"::PAPER 0:E=0:GOTO 3080
3120 IF A$<>CHR$(32) THEN GOSUB 6000:GOTO 3080
3130 IF F=FO AND G=GO OR B(F,G)=0 THEN 3080
3140 PEN 2:PAPER 1:LOCATE 5*F-1,5*G-2:PRINT G$(B(F,
G));
3150 IF B(F,G)<>13 THEN 4000 ELSE LOCATE 5*FO-1,5*
GO-2:PRINT R$;
3160 LOCATE XP,YP:PRINT " ";;FOR I=0 TO 14 :PEN I
3170 LOCATE 5*F-1,5*G-2:PRINT G$(13);:SOUND 1,200+3
*I,10
3180 IF SQ(1)>127 THEN 3180 ELSE NEXT:LOCATE 5*F-
1,5*G-2
3190 PRINT R$;:PAPER 0:PEN 3:LOCATE 28,19+D:PRINT
A$(D);
3200 N(D)=N(D)-2:LOCATE 37,19+D:PRINT N(D);:GOTO 3000

3970 '
3980 '      Mariage et score
3990 '
4000 LOCATE XP,YP:PRINT " ";;PAPER 0:PEN 3:LOCATE 2
8,19+D
4010 PRINT A$(D):IF B(F,G)<>B(FO,GO) THEN 4100
4020 PAPER 1:FOR I=1 TO 15 STEP 2:PEN I:LOCATE 5*
F-1,5*G-2:PRINT R$;
4030 LOCATE 5*FO-1,5*GO-2:PRINT R$;:SOUND 1,300+3*
I,1
4040 IF SQ(1)>127 THEN 4040 ELSE LOCATE 5*F-1,5*G
-2
4050 PRINT G$(B(F,G));:LOCATE 5*FO-1,5*GO-2
4060 PRINT G$(B(F,G));:SOUND 1,300+3*I,2
4070 IF SQ(1)>127 THEN 4070 ELSE NEXT:N(D)=N(D)+1
4080 LOCATE 37,19+D:PAPER 0:PRINT N(D);:B(F,G)=0:B(FO,
GO)=0
4090 P=P+1:IF P=12 THEN 7000 ELSE 3000
4100 FOR I=1 TO 1000*(4-NV):NEXT:PAPER 1:LOCATE 5*
F-1,5*G-2
4110 PRINT R$;:LOCATE 5*FO-1,5*GO-2:PRINT R$;:IF R
ND>0.4+0.2*NV THEN 3000
4970 '

```

```

4980 '      Le mistigri bouge
4990 '
5000 X=INT(RND*3)+1:Y=INT(RND*3)+1
5010 K=0:FOR I=X TO X+2:FOR J=Y TO Y+2
5020 IF B(I,J) THEN K=1:IF RND<0.2 THEN 5040
5030 NEXT:NEXT:IF K THEN 5010 ELSE 3000
5040 FOR II=1 TO 5:X=I-1+INT(RND*3):Y=J-1+INT(RND*
3)
5050 IF B(X,Y)=0 THEN 5080 ELSE LOCATE 5*X-1,5*Y-2
5060 PEN 0:PRINT G$(13);SOUND 1,300,1
5070 IF SQ(1)>127 THEN 5070 ELSE LOCATE 5*X-1,5*Y
-2:PRINT R$;
5080 NEXT:C=B(XM,YM):B(XM,YM)=B(I,J):B(I,J)=C:XM=I:YM=J:
GOTO 3000
5970 '
5980 '      Deplacement etoile
5990 '
6000 AS=ASC(A$):IF AS<240 OR AS>243 THEN RETURN
6010 PAPER 1:PEN 1:LOCATE XP,YP:PRINT "*"
6020 IF AS=242 AND XP>6 THEN XP=XP-5:F=F-1
6030 IF AS=243 AND XP<26 THEN XP=XP+5:F=F+1
6040 IF AS=240 AND YP>5 THEN YP=YP-5:G=G-1
6050 IF AS=241 AND YP<25 THEN YP=YP+5:G=G+1
6060 PEN 2:LOCATE XP,YP:PRINT "*";PAPER 0:RETURN
6970 '
6980 '      Fin du jeu
6990 '
7000 J=1:FOR I=1 TO A:IF N(I)>N(J) THEN J=I
7010 NEXT:WINDOW 2,27,2,6:PAPER 2:CLS:WINDOW 1,40,1,2
5:PAPER 0
7020 LOCATE 5,3:FOR I=1 TO A:IF N(I)=N(J) THEN PRIN
T A$(I);" A GAGNE"
7030 NEXT
7040 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>""
THEN CLS:END
7050 PEN INT(RND*3)+1:PAPER 0:LOCATE 1,16
7060 PRINT "UNE AUTRE PARTIE : TOUCHE R";:FOR I=1
TO 50:NEXT:GOTO 7040

```

TOURS DE HANOI

A la question “Combien de disques voulez-vous ?”, supposons que l’on ait répondu 3. L’écran s’efface alors et le décor se plante. On peut voir trois rectangles dessinés côte à côte. Sur ceux de droite il n’y a rien, mais sur celui de gauche sont posés trois disques bleus, le plus gros des trois supportant les deux autres. Tout le travail va consister à déplacer ces trois disques et à faire en sorte qu’ils se retrouvent sur le rectangle de droite. Essayons :

- jetez votre dévolu ¹ sur le rectangle gauche ; c’est celui sur lequel sont posés les trois disques. Vous voyez que le plus petit de ceux-ci se soulève d’un seul coup. Il restera en l’air tant que vous n’aurez pas exécuté ce que l’ordinateur vous demande en clignotant : “Reposez le disque”.
- choisissez donc un autre rectangle, pourquoi pas celui de droite ? Le petit disque qui flottait va sagement se poser sur le rectangle de droite.
- continuez à respecter la marche à suivre donnée par l’ordinateur et, puisqu’il insiste en faisant clignoter un message bicolore, désignez une nouvelle fois l’un des trois rectangles. Celui de gauche encore ? pourquoi pas. C’est cette fois au disque moyen de se soulever et d’attendre patiemment que l’on veuille bien le reposer.
- mettez-le aussi sur le rectangle de droite toujours en vous servant des flèches. Aïe, le programme n’apprécie pas et nous rappelle à l’ordre avec un bruit de sonnette ! “Il est interdit de poser un disque sur un disque plus petit” dit-il !

C’est sur cette notion bien précise qu’est basé le jeu. Vous pourrez soulever n’importe quel disque en désignant le rectangle sur lequel il est posé, mais vous n’aurez le droit de le reposer que sur un emplacement vide ou sur un disque de diamètre supérieur. Rien ne vous empêchera, par exemple, de poser le petit disque sur le disque moyen ou même sur le gros disque mais le contraire ne sera jamais réalisable.

1. On déplace les deux étoiles clignotantes jusqu’au rectangle choisi en appuyant sur les flèches horizontales. A ce moment-là, si l’on appuie sur la flèche montante, un disque se soulève et, si l’on appuie sur la flèche descendante, un disque se pose.

■ FIN DU JEU

Puisque le but du jeu est de placer tous les disques sur le rectangle de droite, le message “Vous avez gagné” ne pourra apparaître que lorsque cette situation se sera présentée. Notez toutefois que l’on peut abandonner une partie trop compliquée en appuyant simplement sur la touche F. L’ordinateur proposera alors une nouvelle partie et vous pourrez recommencer en choisissant un nombre de disques moins élevé. Le nombre maximal de disques avec lequel le jeu peut démarrer est de 6. Est-il besoin de préciser qu’il est plutôt difficile de gagner dans ce cas-là ?

■ PHASES ESSENTIELLES

Sous-programme 2000

Le joueur soulève un disque.

2000 à 2020 Tant qu’aucune flèche n’est enfoncée, un message invite le joueur à se décider en clignotant à intervalles réguliers.

2030 Si l’une des flèches horizontales a été pressée, les étoiles clignotantes changent de place.

2040 à 2120 Le disque voulu est soulevé à condition, naturellement, qu’il s’en trouve un sur le rectangle choisi. Sinon l’ordinateur fait celui qui n’entend rien.

2130 Un point est ajouté au total des essais (GOSUB 3150).

| Variables | Fonction | Valeurs possibles |
|-----------|--|-------------------|
| E | Compteur pour le message | 0 à 60 |
| A\$ | Touche enfoncée | Touche quelconque |
| C | Numéro d’une couleur | 1 à 3 |
| XP | Abscisse des étoiles clignotantes | 7, 20 ou 33 |
| X | Numéro du rectangle retenu | 0 à 2 |
| I | Nombre de disques sur le rectangle de gauche | 0 à N |

| Variables | Fonction | Valeurs possibles |
|-----------|---|-------------------|
| A(1,I) | Demi-largeur du disque placé en haut de la pile gauche | 1 à 6 |
| J | Nombre de disques sur le rectangle du milieu | 0 à N |
| A(2,J) | Demi-largeur du disque placé en haut de la pile du milieu | 1 à 6 |
| K | Nombre de disques sur le rectangle de droite | 0 à N |
| A(3,K) | Demi-largeur du disque placé en haut de la pile de droite | 1 à 6 |
| Z | Abscisse du disque soulevé | 2 à 33 |
| T\$ | Dessin du disque soulevé | Chaîne caractères |

Sous-programme 3000

Le joueur repose le disque.

3000 à 3030 Le disque soulevé lors de la phase précédente se met à clignoter dans la même couleur que le message demandant au joueur de choisir un rectangle.

3040 L'ordinateur regarde s'il est nécessaire de modifier l'emplacement des étoiles clignotantes.

3050 à 3150 Dans le cas où l'on essaie de poser un disque sur un disque plus petit, le programme se branche à la ligne 4000 ; il faudra ensuite faire porter son choix sur un autre rectangle. Lorsque le coup est accepté par l'ordinateur, le disque vient se dessiner à l'endroit voulu.

| Variables | Fonction | Valeurs possibles |
|-----------|--------------------------------|-------------------|
| A | Demi-largeur du disque soulevé | 1 à 6 |
| N | Nombre total de disques | 3 à 6 |
| S | Nombre d'essais | 0 à ? |

■ MODIFICATIONS POSSIBLES

Augmentation du nombre maximum de disques

Ce maximum, établi à 6 dans ce programme, pourra être porté à un niveau supérieur. (A réserver aux longues soirées d'hiver...)

Utilisation de ce programme par plusieurs joueurs

L'ordinateur devra garder en mémoire le nombre d'essais de chacun des participants et déterminer le gagnant, c'est-à-dire le joueur qui aura déplacé tous ses disques le plus rapidement.

```
50 '
60 '
70 '          TOURS   DE   HANOI
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    0:PAPER 0
110 PEN 3:LOCATE 1,15:PRINT "COMBIEN DE DISQUES V
    OULEZ-VOUS ?"
120 LOCATE 8,17:PRINT "( DE 3 A 6 )"
130 A$=INKEY$:IF A$="" THEN 130 ELSE N=VAL(A$)
140 IF N<3 OR N>6 THEN 130 ELSE FOR I=1 TO N:A(1,
    I)=I:NEXT
150 SYMBOL 253,255,255,255,255,255,255,255,0:SYMBOL
    254,7,15,31,63,31,15,7,0
160 SYMBOL 255,224,240,248,252,248,240,224,0:FOR I=1
    TO 10:D$=D$+CHR$(253):NEXT
970 '
980 '          Decor
990 '
1000 CLS:PEN 1:FOR I=2 TO 28 STEP 13:FOR J=14 TO
    16
1010 LOCATE I,J:PRINT "*****":NEXT:NEXT:FOR
    I=1 TO N
1020 LOCATE 1+I,14-I:PRINT CHR$(254)+MID$(D$,1,12-2
    *I)+CHR$(255)
1030 NEXT:WINDOW 12,29,1,3:PAPER 1:CLS:WINDOW 1,40,1,
    25:PAPER 0
```

```

1040 LOCATE 1,25:PRINT "POUR ABANDONNER , APPUYE
R SUR F";
1050 PEN 2:LOCATE 13,2:PRINT "TOURS DE HANOI":L
OCATE 19,23
1060 PRINT "NOMBRE D'ESSAIS ":"I=N:E=60:XP=7:C=1:J=
0
1970 '
1980 '      Un disque est souleve
1990 '
2000 A$=INKEY$:IF A$="F" THEN 6020 ELSE IF A$<>""
THEN 2030
2010 E=E+1:IF E<60 THEN 2000 ELSE PEN C:LOCATE XP
,15:PRINT "***"
2020 LOCATE 12,19:PRINT "SOULEVEZ UN DISQUE":E=0:C
=4-C:GOTO 2000
2030 IF A$<>CHR$(240) THEN GOSUB 5000:GOTO 2000 E
LSE ON X GOTO 2070,2100
2040 IF I=0 THEN 2000 ELSE A=A(1,I):PEN 1:LOCATE 2,
14-I
2050 PRINT SPACE$(12):T$=CHR$(254)+LEFT$(D$,12-2*A)
+CHR$(255)
2060 Z=1+A:LOCATE Z,5:PRINT T$:A(1,I)=0:I=I-1:GOTO 21
30
2070 IF J=0 THEN 2000 ELSE A=A(2,J):PEN 1:LOCATE 1
5,14-J
2080 PRINT SPACE$(12):T$=CHR$(254)+LEFT$(D$,12-2*A)
+CHR$(255)
2090 Z=14+A:LOCATE Z,5:PRINT T$:A(2,J)=0:J=J-1:GOTO
2130
2100 IF K=0 THEN 2000 ELSE A=A(3,K):PEN 1:LOCATE 2
8,14-K
2110 PRINT SPACE$(12):T$=CHR$(254)+LEFT$(D$,12-2*A)
+CHR$(255)
2120 Z=27+A:LOCATE Z,5:PRINT T$:A(3,K)=0:K=K-1
2130 GOSUB 3150:C=2:E=60
2970 '
2980 '      Un disque est repose
2990 '
3000 A$=INKEY$:IF A$="F" THEN 6020 ELSE IF A$<>""
THEN 3040
3010 E=E+1:IF E<60 THEN 3000 ELSE PEN C:LOCATE XP
,15:PRINT "***"
3020 LOCATE 12,19:PRINT "REPOSEZ LE DISQUE ":LOCA

```

```

TE Z,5
3030 PRINT T$:E=0:C=3-C:GOTO 3000
3040 IF A$<>CHR$(241) THEN GOSUB 5000:GOTO 3000 E
LSE ON X GOTO 3080,3110
3050 IF A<A(1,I) THEN 4000 ELSE I=I+1:A(1,I)=A:PEN 0:
LOCATE Z,5
3060 PRINT T$:PEN 1:LOCATE 1+A,14-I
3070 PRINT CHR$(254)+LEFT$(D$,12-2*A)+CHR$(255):GOT
O 3140
3080 IF A<A(2,J) THEN 4000 ELSE J=J+1:A(2,J)=A:PEN 0
:LOCATE Z,5
3090 PRINT T$:PEN 1:LOCATE 14+A,14-J
3100 PRINT CHR$(254)+LEFT$(D$,12-2*A)+CHR$(255):GOT
O 3140
3110 IF A<A(3,K) THEN 4000 ELSE K=K+1:A(3,K)=A:PEN 0
:LOCATE Z,5
3120 PRINT T$:PEN 1:LOCATE 27+A,14-K
3130 PRINT CHR$(254)+LEFT$(D$,12-2*A)+CHR$(255):IF
K=N THEN 6000
3140 GOSUB 3150:E=60:C=3:GOTO 2000
3150 PEN 2:S=S+1:LOCATE 36,23:PRINT S:RETURN
3970 '
3980 '      Figure interdite
3990 '
4000 WINDOW 1,33,19,20:PAPER 1:CLS:FOR R=1 TO 8
4010 PEN R:LOCATE 1,1:PRINT "INTERDIT DE POSER UN
E GROS DISQUE"
4020 LOCATE 7,2:PRINT "SUR UN PETIT DISQUE";:SOUN
D 1,200+5*I,8
4030 IF SQ(1)>127 THEN 4030 ELSE NEXT:PAPER 0:CLS
4040 WINDOW 1,40,1,25:GOSUB 3150:GOTO 3000
4970 '
4980 '      Deplacememt etoile
4990 '
5000 AS=ASC(A$):IF AS<242 AND AS>243 THEN RETURN
5010 PEN 1:LOCATE XP,15:PRINT "***"
5020 IF AS=242 AND XP>7 THEN XP=XP-13:X=X-1
5030 IF AS=243 AND XP<33 THEN XP=XP+13:X=X+1
5040 PEN 2:LOCATE XP,15:PRINT "***":RETURN
5970 '
5980 '      Fin du jeu
5990 '
6000 WINDOW 12,29,5,7:PAPER 1:CLS:WINDOW 1,40,1,25

```

```
6010 PEN 3:LOCATE 13,6:PRINT "VOUS AVEZ GAGNE":FOR I=1 TO 2000:NEXT I
6020 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>" " THEN CLS:END
6030 PEN INT(RND*3)+1:PAPER 0:LOCATE 9,19
6040 PRINT "UNE AUTRE PARTIE : TOUCHE R":FOR I=1 TO 100:NEXT I:GOTO 6020
```

SOLITAIRE

Avant que le jeu ne commence, l'ordinateur demande le nombre de joueurs et leur nom. Il nous donne ensuite le choix entre différents temps de réflexion. Prenons, pour commencer, la durée la plus longue ; cela nous laissera le temps de bien voir comment fonctionne le programme. Dès que l'on a appuyé sur la touche voulue, l'écran s'efface et une grille se dessine devant nos yeux. Elle contient 49 cases (soit 7 sur 7) mais une partie de celles-ci, (celles qui sont situées dans les angles) ne servira à rien dans le jeu. Les cases utiles apparaissent en bleu et l'on peut voir que l'ordinateur a déposé sur chacune d'elles un pion jaune. A vrai dire, il y en a tout de même une qui a été laissée libre ; c'est la seule qui, pour l'instant, n'est pas occupée par un jeton.

Regardez le message qui se colore de différentes façons en bas de l'écran. Il nous invite à choisir ¹ l'un des pions du damier. Désignez celui que vous voulez et vous remarquerez qu'il change immédiatement de couleur. Grâce à son aspect clignotant, nous ne risquons pas d'oublier sa position.

Le message qui s'affiche à la limite inférieure du téléviseur a changé ; l'ordinateur nous demande maintenant de désigner une case vide. Notre choix est restreint, il n'y a pas trente-six possibilités ; amenez donc le carré lumineux sur la seule case vide qui existe sur la grille et enfoncez la barre d'espacement. Et vous verrez... qu'il n'y a rien à voir ! Le programme ignore totalement notre action et répète inlassablement la même phrase : "Choisissez une case vide". Vous n'aurez pas plus de succès si vous essayez de placer le petit pavé lumineux sur une case déjà occupée.

Revenez donc au point de départ en ramenant le petit carré clignotant à proximité de votre premier pion et enfoncez la barre d'espacement. Le pion reprendra sa couleur normale et vous pourrez alors désigner un de ses semblables dans la grille. Si vous le faites au hasard, vous n'aurez probablement pas plus de chance que la première fois. Choisissez obligatoirement un jeton qui soit sur la même verticale ou sur la même horizontale que la case vide et qui soit séparé de celle-ci par une, et une seule case intermédiaire. Lorsqu'ensuite vous indiquerez la case libre, il viendra se placer à cet endroit et le jeton de la case intermédiaire aura disparu. Il vous faudra ensuite renouveler cette opération le plus de fois possible puisque le gagnant de

1. Il suffit de déplacer, avec les quatre flèches, le petit carré jaune clignotant jusqu'à ce qu'il se trouve dans la case voulue. A ce moment-là, on appuie sur la barre d'espacement.

ce jeu est celui qui aura fait disparaître le maximum de pions de l'écran.

Si vous ne voulez pas que votre partie soit brutalement interrompue par l'ordinateur, tenez compte de ce que la durée de chaque coup est limitée. En laissant le chronomètre qui apparaît à droite de l'image arriver à zéro, vous donnez automatiquement la main au joueur suivant.

■ FIN DU JEU

Dès que la situation est bloquée et qu'il n'est plus possible de déplacer un pion, on laisse le laps de temps imparti arriver à son terme. L'ordinateur indique alors le score des différents participants et propose une grille nouvelle au joueur suivant.

Le temps de réflexion accordé à chaque essai varie entre 8 et 25 secondes selon le choix qui a été fait en début de partie.

■ PHASES ESSENTIELLES

Sous-programme 2000

Le joueur choisit le pion qu'il veut déplacer.

2000 à 2020 La phrase "Choisissez un pion" change de couleur à chaque fois que la variable E atteint la valeur 40, c'est-à-dire approximativement toutes les secondes. A ce moment-là le compteur du chronomètre est décrémenté ; s'il s'annule, le jeu se termine.

2030 à 2050 La barre d'espacement vient d'être enfoncée et le programme se doit de vérifier que le coup est valable. La case choisie contient-elle bien un pion ?

| Variables | Fonction | Valeurs possibles |
|-----------|--------------------------------------|-------------------|
| A\$ | Touche enfoncée | Touche quelconque |
| E | Compteur pour l'affichage du message | 0 à 40 |
| T | Compteur du chronomètre | 0 à 24 |

| Variables | Fonction | Valeurs possibles |
|-----------|---------------------------------|--|
| X , Y | Coordonnées de la case choisie | 1 à 7 |
| A(X,Y) | Valeur de la case (X,Y) | 0 : hors grille 1 : pion 2 : case vide |
| G\$ | Dessin d'un pion | Chaîne caractères |
| C | Couleur du message | 1 à 3 |
| XO et YO | Coordonnées de la première case | 1 à 7 |

Sous-programme 3000

Le joueur choisit une case libre.

3000 à 3030 Ces lignes sont équivalentes des lignes 2000 à 2020. Cependant le pion qui a été retenu dans le sous-programme précédent s'est mis à clignoter.

3040 à 3110 L'ordinateur procède à toutes les vérifications nécessaires pour s'assurer de la validité de la case désignée. Respecte-t-elle toutes les contraintes imposées par la règle du jeu ?

3120 à 3130 Le compte des pions gagnés est établi.

| Variables | Fonction | Valeurs possibles |
|-----------|---|-------------------|
| AX , AY | Ecart entre les coordonnées des deux cases pointées | 0 à 6 |
| U , V | Coordonnées de la case qui disparaîtra | 1 à 7 |
| H\$ | Effacement d'un pion | Chaîne caractères |
| B | Numéro du joueur concerné | 1 à 9 |
| S(B) | Score du joueur concerné | 0 à 32 |

■ MODIFICATIONS POSSIBLES

Suppression du chronomètre

Les joueurs auraient alors un temps de réflexion plus important.

Fixation d'un nombre minimum de pions à supprimer pour gagner

Ce nombre, tenant compte du fait qu'il y a 33 cases, pourrait être de l'ordre de 28 ou de 30.

Augmentation du nombre de cases vides en début de partie

Pourquoi, au lieu de n'en libérer qu'une seule, ne pas en libérer trois ou quatre par exemple.

Modification du dessin de la grille

L'augmentation de la surface de jeu et, en conséquence, du nombre de jetons ne devrait pas être d'une difficulté insurmontable.

```
50 '
60 '
70 '          SOLITAIRE
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    O:PAPER 0
110 PEN 2:LOCATE 2,1:PRINT "NOMBRE DE JOUEURS ?"
120 A$=INKEY$:IF A$="" THEN 120 ELSE N=VAL(A$):IF
N=0 THEN 120
130 PEN 3:FOR I=1 TO N:LOCATE 1,2*I+2:PRINT "NOM D
U JOUEUR";I;
140 LINE INPUT B$(I):B$(I)=LEFT$(B$(I),9):NEXT
150 PEN 2:LOCATE 2,23:PRINT "CHOISISSEZ LE TEMPS
DE REFLEXION"
160 PRINT "1: REDUIT          2: MOYEN          3: LONG"
170 A$=INKEY$:IF A$="" THEN 170 ELSE NV=VAL(A$):IF
NV<1 OR NV>3 THEN 170
180 G$=CHR$(214)+CHR$(215)+CHR$(8)+CHR$(8)+CHR$(10)
+CHR$(213)+CHR$(212)
190 H$=" " +CHR$(8)+CHR$(8)+CHR$(10)+" " :B=1
```

```

200 FOR I=3 TO 5:FOR J=1 TO 7:A(I,J)=1:A(J,I)=1:NEXT:
NEXT
970 '
980 '      Decor
990 '
1000 CLS:PAPER 1:WINDOW 11,20,1,22:CLS:WINDOW 5,26,7,
16
1010 CLS:WINDOW 30,40,1,3:CLS:WINDOW 1,40,1,25:PAPER
0
1020 FOR I=72 TO 408 STEP 48:MOVE I,399:DRAW I,48,
0:NEXT
1030 FOR I=56 TO 392 STEP 48:MOVE 64,I:DRAW 416,I,
0:NEXT
1040 PEN 1:LOCATE 1,1:PRINT B$(B):PEN 2:LOCATE 31,2
1050 PRINT "SOLITAIRE":LOCATE 29,12:PRINT "SCORE:
O":PEN 3
1060 LOCATE 23,19:PRINT "TEMPS:      SECONDES":PAPE
R 1:PEN 3
1070 FOR I=3 TO 5:FOR J=1 TO 7:LOCATE 3*I+3,3*J-1:
PRINT G$
1080 LOCATE 3*J+3,3*I-1:PRINT G$:NEXT:NEXT
1090 I=INT(RND*3)+3:J=INT(RND*3)+3:LOCATE 3*I+3,3*J
-1
1100 PRINT H$:A(I,J)=2:X=4:Y=4:XP=257:YP=208:C=3:E=40
:T=8*NV
1970 '
1980 '      Choix du pion a deplacer
1990 '
2000 A$=INKEY$:IF A$<>" " THEN 2030 ELSE E=E+1:IF
E<40 THEN 2000
2010 GOSUB 4060:E=0:T=T-1:PAPER 0:PEN 3:LOCATE 29,
19:PRINT T:IF T=0 THEN 5000
2020 C=3-C:PEN C:LOCATE 7,24:PRINT "CHOISISSEZ UN
PION      ":GOTO 2000
2030 IF A$<>CHR$(32) THEN GOSUB 4000:GOTO 2000
2040 IF A(X,Y)<>1 THEN 2000 ELSE PAPER 1:PEN 2:LOC
ATE 3*X+3,3*Y-1
2050 PRINT G$:PAPER 0:A(X,Y)=2:E=40:T=8*NV:C=2:X0=X:
Y0=Y
2970 '
2980 '      Choix d'une case vide
2990 '
3000 A$=INKEY$:IF A$<>" " THEN 3040 ELSE E=E+1:IF

```

```

E<40 THEN 3000
3010 GOSUB 4060:T=T-1:PAPER 0:PEN 3:LOCATE 29,19:P
RINT T:IF T=0 THEN 5000
3020 PEN C:LOCATE 7,24:PRINT "CHOISISSEZ UNE CASE
VIDE"
3030 PAPER 1:LOCATE 3*X0+3,3*Y0-1:PRINT G$:PAPER
0:E=0:C=5-C:GOTO 3000
3040 IF A$<>CHR$(32) THEN GOSUB 4000:GOTO 3000
3050 IF A(X,Y)<>2 THEN 3000 ELSE PEN 3:PAPER 1
3060 IF X=X0 AND Y=Y0 THEN 3100 ELSE AX=ABS(X-X0
):AY=ABS(Y-Y0)
3070 IF (AX<>2 OR AY) AND (AX OR AY<>2) THEN 3000
3080 U=(X+X0)/2:V=(Y+Y0)/2:IF A(U,V)<>1 THEN 3000
3090 LOCATE 3*X0+3,3*Y0-1:PRINT H$
3100 LOCATE 3*X+3,3*Y-1:PRINT G$:A(X,Y)=1
3110 IF X=X0 AND Y=Y0 THEN 3130 ELSE LOCATE 3*U
+3,3*V-1
3120 PRINT H$:A(U,V)=2:S(B)=S(B)+1:T=8*N*V
3130 PAPER 0:PEN 2:LOCATE 35,12:PRINT S(B):E=40:C=3:
GOTO 2000
3970 '
3980 '      Deplacement point lumineux
3990 '
4000 AS=ASC(A$):IF AS<240 OR AS>243 THEN RETURN E
LSE C=1:GOSUB 4060
4010 IF AS=242 AND A(X-1,Y) THEN X=X-1:XP=XP-48
4020 IF AS=243 AND A(X+1,Y) THEN X=X+1:XP=XP+48
4030 IF AS=240 AND A(X,Y-1) THEN Y=Y-1:YP=YP+48
4040 IF AS=241 AND A(X,Y+1) THEN Y=Y+1:YP=YP-48
4050 C=2:GOSUB 4060:RETURN
4060 MOVE XP,YP:DRAW 3,3,C:MOVER -3,0:DRAW 3,-3,C:
RETURN
4970 '
4980 '      Fin du jeu
4990 '
5000 CLS:PEN 3:FOR I=1 TO N:LOCATE 2,2*I:PRINT B$(I
)
5010 PEN 3:LOCATE 15,2*I:PRINT S(I):NEXT I:B=B+1:IF B=
N+1 THEN 5040
5020 PEN 2:LOCATE 1,20:PRINT B$(B);" ENFONCEZ UNE
TOUCHE"
5030 IF INKEY$="" THEN 5030 ELSE 200
5040 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>""

```

```
THEN CLS:END
5050 PEN INT(RND*3)+1:LOCATE 2,22:PRINT "UNE AUTRE
      PARTIE : TOUCHE R"
5060 FOR I=1 TO 100:NEXT:GOTO 5040
```

HEXAPION

Vous avez précisé à l'ordinateur à quel niveau de difficulté devait se dérouler la partie. Vous devez alors avoir devant les yeux une grille de 9 cases. Les 3 cases du haut sont occupées par des pions représentant des têtes de diables jaunes ; ce sont les pions de votre adversaire, en l'occurrence l'ordinateur. Les 3 cases du bas contiennent des petits bonshommes verts, ce sont celles du joueur. Les 3 cases centrales, pour leur part, ne contiennent rien du tout et n'appartiennent pour le moment à personne.

Respectez pas à pas les indications données par le programme et, puisque cela vous est demandé, choisissez ¹ l'une de vos cases avec le crayon optique. Si vous essayez de désigner autre chose qu'un petit bonhomme vert, l'ordinateur l'ignorera. Dès que l'un de vos pions a été indiqué, il se met à clignoter, imité en cela par une phrase écrite en bas du téléviseur. "Reposez votre pion" vous demande-t-on dans diverses couleurs. Choisissez donc une autre case de l'écran et, si votre coup respecte la règle du jeu, vous verrez le bonhomme cesser de clignoter et venir se placer à l'endroit qui vient d'être retenu. Vous savez maintenant comment on déplace un pion. Sachez aussi, cela pourra vous servir, que si vous aviez changé d'avis vous auriez pu revenir sur la case clignotante. L'essai aurait été annulé et vous auriez eu ainsi toute latitude pour déplacer un autre bonhomme.

C'est ensuite au tour de la machine de jouer. Elle fait semblant de réfléchir un court instant puis se décide : une tête de diable passe d'une case à une autre dans un roulement de bip-bip. L'ordinateur, naturellement, aura respecté toutes les règles du jeu ; règles que voici :

- un petit bonhomme ne peut se déplacer vers le haut que s'il a devant lui une case vide ;
- un petit bonhomme peut avancer en diagonale à condition de tomber, ce faisant, sur une case contenant un diable jaune. Le pion adverse est alors "mangé" et le bonhomme prend sa place ;
- dès que le joueur a réussi à poser l'un de ses pions dans l'une des trois cases supérieures, il est déclaré gagnant.

1. Il suffit de déplacer, avec les quatre flèches, la petite étoile verte clignotante jusqu'à ce qu'elle se trouve dans la case voulue. A ce moment-là, on appuie sur la barre d'espacement.

Les règles symétriques sont valables pour l'ordinateur ; il partira de la partie haute de la grille et essaiera de faire en sorte qu'un de ses pions atteigne l'une des cases du bas. Les parties sont généralement de courte durée et dès que l'une d'elles est terminée, on passe à la suivante, chacun reprenant ses positions de départ. Vous remarquerez que l'ordinateur garde le souvenir de ses mauvaises expériences... Ce qui veut dire que si vous avez gagné une partie contre lui et que vous essayez de lui refaire un peu plus tard un coup identique, il ne recommencera pas les mêmes erreurs et jouera autrement. La machine donne l'impression de devenir de plus en plus intelligente.

■ FIN DU JEU

Une partie se termine lorsque l'un des adversaires a réussi à placer un pion sur la ligne opposée à sa propre ligne de départ ou lorsque l'un des joueurs se trouve dans une situation telle qu'aucun mouvement de sa part n'est possible. Il est alors contraint à l'abandon par le programme.

Chaque partie gagnée rapporte un point et c'est à la fin des sept parties consécutives qu'est proclamé le vainqueur du jeu.

Le niveau de difficulté détermine la façon de jouer de l'ordinateur. Au niveau 1, il ne voit pas toujours très bien quelle est la solution la plus intéressante pour lui. Et au niveau 3, il ne commet plus que de très rares erreurs.

■ PHASES ESSENTIELLES

Sous-programme 2000

Le joueur choisit un des bonshommes jaunes.

2000 à 2030 La phrase "Désignez un de vos pions", colorée de deux couleurs différentes, indique au joueur qu'il doit prendre une décision.

2040 à 2060 L'ordinateur vérifie simplement que la case que l'on vient d'indiquer contient bien un lutin jaune.

| Variables | Fonction | Valeurs possibles |
|-----------|-------------------------------------|--|
| A\$ | Touche enfoncée | Touche quelconque |
| A | Compteur pour la couleur du message | 0 à 40 |
| XP et YP | Coordonnées de l'étoile clignotante | 8, 13 et 18 |
| X , Y | Coordonnées de la case choisie | 1 à 3 |
| P | Numéro de la case retenue | 1 à 9 |
| G(P) | Valeur de la case numérotée P | 0 : case libre 1 : case ordinateur 2 : case joueur |
| H\$ | Effacement d'un bonhomme | Chaîne caractères |
| XO et YO | Coordonnées de la première case | 1 à 3 |
| C | Couleur du bonhomme clignotant | 1 à 3 |

Sous-programme 3000

Le joueur repose son bonhomme.

3000 à 3030 Le pion du joueur clignote en même temps qu'une phrase précise qu'il faut indiquer une autre case.

3040 à 3090 Le numéro de la case désignée est calculé et l'ordinateur se livre alors à toutes les vérifications voulues. Si le coup est acceptable, le pion du joueur est déplacé. Lorsque ce pion atteint la ligne supérieure de la grille, le joueur est déclaré gagnant.

| Variables | Fonction | Valeurs possibles |
|-----------|------------------------------------|-------------------|
| Q | Numéro de la deuxième case retenue | 1 à 9 |
| G(Q) | Valeur de la case numéro Q | Idem G(P) |
| F\$ | Dessin d'un bonhomme | Chaîne caractères |
| X1 et Y1 | Coordonnées de la deuxième case | 1 à 3 |

Sous-programme 4000

L'ordinateur analyse le jeu.

4000 à 4020 Le programme vérifie qu'il reste bien au moins un pion à l'ordinateur.

4030 à 4090 Il regarde si celui-ci peut jouer quelque part. Si ce n'est pas le cas, le joueur est déclaré vainqueur par abandon de la machine.

4100 à 4170 La moitié des diverses configurations de la surface de jeu est explorée. Puis c'est au tour des configurations symétriques d'être analysées. Quand l'ordinateur retrouve la situation qui est celle du moment, il en déduit sa façon de jouer. Il est possible alors, qu'il s'aperçoive qu'il a pris un mauvais chemin ; pas têtù, il abandonne aussitôt.

| Variables | Fonction | Valeurs possibles |
|-----------|--|-------------------|
| N | Numéro de la configuration étudiée | 1 à 19 |
| H | Numéro du data lu | 1 à 9 |
| L | Valeur lue en data | Idem G(P) |
| F | Drapeau configuration directe/symétrique | 1 et 2 |
| U | Numéros des cases symétriques | 1 à 9 |
| E\$(N,I) | Chemin à parcourir par le pion de l'ordinateur | 0 à 59 |

Sous-programme 5000

L'ordinateur décide où jouer.

5000 à 5090 Le programme détermine le pion de l'ordinateur qui va devoir être déplacé, s'il existe, et le fait clignoter en musique.

5100 à 5250 Le pion est alors déplacé jusqu'à sa nouvelle position. Il ne reste plus qu'à savoir si cette action laisse une chance au joueur.

| Variables | Fonction | Valeurs possibles |
|---------------|--|-------------------|
| M | Numéro du chemin à parcourir | 1 à 3 |
| NO et MO | Valeurs de N et M mises en réserve | 1 à 19 et 1 à 3 |
| P et Q | Numéros des cases de départ et d'arrivée du pion de l'ordinateur | 1 à 9 |
| X , Y , Z , T | Coordonnées des cases de départ et d'arrivée | 1 à 3 |

■ MODIFICATIONS POSSIBLES

Durée du jeu

Il est tout à fait possible de ne pas limiter la durée d'un jeu à 7 parties, de ne pas faire intervenir de niveau de difficulté et de laisser le joueur résister le plus longtemps possible aux assauts de l'ordinateur.

```

50 '
60 '
70 '                HEXAPION
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    0:PAPER 0
110 SYMBOL 248,32,32,127,127,231,231,255,255:SYMBOL
    249,4,4,254,254,231,231,255,255
120 SYMBOL 250,127,63,56,24,31,15,7,3:SYMBOL 251,254,
    252,28,24,248,240,224,192
130 SYMBOL 252,3,3,3,3,1,1,63,63:SYMBOL 253,192,192,1
    92,192,128,128,252,252
140 SYMBOL 254,3,3,3,3,12,12,48,48:SYMBOL 255,192,192
    ,192,192,48,48,12,12
150 F$=CHR$(252)+CHR$(253)+CHR$(8)+CHR$(8)+CHR$(10)+
    CHR$(254)+CHR$(255)
160 G$=CHR$(248)+CHR$(249)+CHR$(8)+CHR$(8)+CHR$(10)
    +CHR$(250)+CHR$(251)
170 H$="  "+CHR$(8)+CHR$(8)+CHR$(10)+"  "
180 PEN 2:LOCATE 1,10:PRINT "CHOISISSEZ LA DE DIFF
    ICULTE ( DE 1 A 3)"

```

```

190 A$=INKEY$:IF A$="" THEN 190 ELSE NV=VAL(A$):IF
  NV=0 OR NV>3 THEN 190
200 CLS:DIM E$(19,3):FOR I=1 TO 19:FOR J=1 TO 9:REA
  D A:NEXT
210 FOR J=1 TO 3:READ E$(I,J):NEXT:NEXT
220 FOR J=1 TO 5*(NV-1):I=INT(RND*11)+1:E$(I,1)="O":N
  EXT
970 '
980 '      Decor
990 '
1000 PAPER 1:WINDOW 3,20,3,20:CLS:WINDOW 26,35,3,5:CL
  S
1010 WINDOW 25,36,8,12:CLS:WINDOW 23,38,15,19:CLS
1020 PAPER 0:WINDOW 4,19,4,19:CLS:WINDOW 1,40,1,25:PA
  PER 1
1030 PEN 2:LOCATE 26,9:PRINT "NIVEAU ";NV:LOCATE 2
  6,11
1040 PRINT "PARTIES 0":PEN 3:LOCATE 24,16:PRINT "O
  RDINATEUR 0"
1050 LOCATE 26,18:PRINT "JOUEUR      0":PAPER 0
1060 PEN 2:LOCATE 27,4:PRINT "HEXAPION":PEN 1:LOCA
  TE 1,24
1070 PRINT "DEPLACEMENT EN AVANT: VERS CASE LIBR
  E"
1080 PRINT "EN DIAGONALE: PRISE DU PION ADVERSE";
1090 PAPER 1:FOR I=5 TO 15 STEP 5:FOR J=5 TO 15
  STEP 5
1100 WINDOW I,I+3,J,J+3:CLS:NEXT:NEXT:WINDOW 1,40,1,2
  5:FOR I=6 TO 16 STEP 5
1110 PEN 3:LOCATE I,6:PRINT G$:PEN 2:LOCATE I,16:PR
  INT F$:NEXT
1120 FOR I=1 TO 3:G(I)=1:G(I+3)=0:G(I+6)=2:NEXT:A=40:X
  =2:Y=2:XP=13:YP=13:C=0
1970 '
1980 '      Le joueur choisit un pion
1990 '
2000 A$=INKEY$:IF A$<>"" THEN 2040 ELSE A=A+1:IF
  A<40 THEN 2000
2010 PAPER 0:PEN C:A=0:C=2-C:LOCATE 1,22
2020 PRINT "DESIGNEZ UN DE VOS PIONS"
2030 PAPER 1:LOCATE XP,YP:PRINT "*":GOTO 2000
2040 IF A$<>CHR$(32) THEN GOSUB 6000:GOTO 2000
2050 P=X+3*(Y-1):IF G(P)<2 THEN 2000

```

```

2060 PAPER 1:LOCATE 5*X+1,5*Y+1:PRINT H$:A=40:X0=X
:Y0=Y:C=3
2970 '
2980 '      Le joueur repose le pion
2990 '
3000 A$=INKEY$:IF A$<>" " THEN 3040 ELSE A=A+1:IF
A<40 THEN 3000
3010 PAPER 1:PEN C:A=0:C=3-C:LOCATE 5*X0+1,5*Y0+1:
PRINT F$
3020 PAPER 0:LOCATE 1,22:PRINT "REPOSEZ VOTRE PIO
N";SPACE$(8)
3030 PAPER 1:LOCATE XP,YP:PRINT "*":GOTO 3000
3040 IF A$<>CHR$(32) THEN GOSUB 6000:GOTO 3000 EL
SE Q=X+3*(Y-1)
3050 IF Q=P THEN PAPER 1:PEN 2:LOCATE 5*X+1,5*Y+1
:PRINT F$:A=40:C=2:GOTO 2000
3060 IF ( G(Q)=2 OR P<Q OR P*Q=21 ) OR ( P-Q<>3 AN
D G(Q)<>1 ) THEN 3000
3070 IF ( P-Q=3 AND G(Q) ) OR P-Q>4 THEN 3000
3080 PAPER 1:LOCATE 5*X0+1,5*Y0+1:PRINT H$:PEN 2:L
OCATE 5*X+1,5*Y+1:PRINT F$
3090 G(P)=0:G(Q)=2:FOR I=1 TO 3:IF G(I)=2 THEN 7030
ELSE NEXT:X1=X:Y1=Y
3970 '
3980 '      Ordinateur cherche ou jouer
3990 '
4000 FOR I=1 TO 9:IF G(I)<>1 THEN NEXT:GOTO 6060
4010 PAPER 2:PEN 0:LOCATE 1,1:PRINT " CHUT .... JE
REFLECHIS "
4020 LOCATE XP,YP:PAPER 1:PRINT " ":FOR I=1 TO 500
:NEXT
4030 FOR I=1 TO 6:IF G(I)<>1 THEN 4090
4040 IF G(I+3)=0 THEN 4100
4050 IF I=2 THEN IF G(4)=2 OR G(6)=2 THEN 4100
4060 IF I=5 THEN IF G(7)=2 OR G(9)=2 THEN 4100
4070 IF I=1 OR I=3 THEN IF G(5)=2 THEN 4100
4080 IF I=4 OR I=6 THEN IF G(8)=2 THEN 4100
4090 NEXT:GOTO 7020
4100 RESTORE:FOR N=1 TO 19:FOR H=1 TO 9
4110 READ L:IF G(H)=L THEN NEXT:F=1:GOTO 4170
4120 FOR I=H+1 TO 12:READ L:NEXT I:NEXT N
4130 RESTORE:FOR N=1 TO 19:FOR H=1 TO 9
4140 READ L:U=4-6*(H>3)-6*(H>6)-H

```

```

4150 IF G(U)=L THEN NEXT:F=2:GOTO 4170
4160 FOR I=H+1 TO 12:READ L:NEXT I:NEXT N:END
4170 FOR I=1 TO 3:IF E$(N,I)="0" THEN NEXT:GOTO 70
20
4970 '
4980 '      Ordinateur place son pion
4990 '
5000 M=INT(RND*3)+1:IF E$(N,M)="0" THEN 5000
5010 NO=N:MO=M:P=VAL(LEFT$(E$(N,M),1))
5020 Q=VAL(RIGHT$(E$(N,M),1)):IF F=1 THEN 5060
5030 IF P=1 OR P=3 THEN P=4-P
5040 IF P=4 OR P=6 THEN P=10-P
5050 IF P=7 OR P=9 THEN P=16-P
5060 PAPER 0:PEN 3:Y=INT((P-1)/3)+1:X=P MOD 3:IF X=0
    THEN X=3
5070 LOCATE 1,1:PRINT SPACE$(24):PAPER 1:FOR I=1 TO
    6:SOUND 1,200+I,5
5080 IF SQ(1)>127 THEN 5080 ELSE LOCATE 5*X+1,5*Y
    +1:PRINT G$:SOUND 1,220+I,5
5090 IF SQ(1)>127 THEN 5090 ELSE LOCATE 5*X+1,5*Y
    +1:PRINT H$:NEXT
5100 IF F=1 THEN 5130 ELSE IF Q=1 OR Q=3 THEN Q
    =4-Q
5110 IF Q=4 OR Q=6 THEN Q=10-Q
5120 IF Q=7 OR Q=9 THEN Q=16-Q
5130 T=INT((Q-1)/3)+1:Z=Q MOD 3:IF Z=0 THEN Z=3
5140 G(P)=0:G(Q)=1:FOR I=1 TO 6:SOUND 1,250+I,5
5150 IF SQ(1)>127 THEN 5150 ELSE LOCATE 5*Z+1,5*T
    +1:PRINT H$:SOUND 1,220+I,5
5160 IF SQ(1)>127 THEN 5160 ELSE LOCATE 5*Z+1,5*T
    +1:PRINT G$:NEXT
5170 FOR I=7 TO 9:IF G(I)=1 THEN 7010 ELSE NEXT
5180 FOR I=1 TO 9:IF G(I)<>2 THEN NEXT:GOTO 7010
5190 C=2:X=X1:Y=Y1:FOR I=4 TO 9:IF G(I)<>2 THEN 525
0
5200 IF G(I-3)=0 THEN 2000
5210 IF I=5 THEN IF G(1)=1 OR G(3)=1 THEN 2000
5220 IF I=8 THEN IF G(4)=1 OR G(6)=1 THEN 2000
5230 IF I=4 OR I=6 THEN IF G(2)=1 THEN 2000
5240 IF I=7 OR I=9 THEN IF G(5)=1 THEN 2000
5250 NEXT:GOTO 7000
5970 '
5980 '      Deplacementt etoile

```

```

5990 '
6000 AS=ASC(A$):IF AS<240 OR AS>243 THEN RETURN
6010 PAPER 1:PEN 1:LOCATE XP,YP:PRINT "*"
6020 IF AS=242 AND XP>8 THEN XP=XP-5:X=X-1
6030 IF AS=243 AND XP<18 THEN XP=XP+5:X=X+1
6040 IF AS=240 AND YP>8 THEN YP=YP-5:Y=Y-1
6050 IF AS=241 AND YP<18 THEN YP=YP+5:Y=Y+1
6060 PEN 3:LOCATE XP,YP:PRINT "*":PAPER 0:RETURN
6970 '
6980 '          Fin du jeu
6990 '
7000 PAPER 2:PEN 0:LOCATE 5,1:PRINT " ABANDONNEZ!"
":FOR J=1 TO 4000:NEXT
7010 PAPER 3:PEN 0:LOCATE 25,14:PRINT " J'AI GAGNE
":I=0:GOTO 7040
7020 PAPER 2:PEN 0:LOCATE 2,1:PRINT"J'ABANDONNE";S
PACE$(11):FOR J=1 TO 4000:NEXT
7030 PAPER 3:PEN 0:LOCATE 25,14:PRINT " J'AI PERDU
":I=1
7040 FOR J=1 TO 3000:NEXT:S(I)=S(I)+1:PAPER 1:PEN 3:
LOCATE 35,16+2*I
7050 PRINT S(I):PAPER 0:LOCATE 1,22:PRINT SPACE$(30)
:LOCATE 25,14
7060 PRINT SPACE$(12):W=W+1:PEN 2:PAPER 1:LOCATE 3
3,11:PRINT W
7070 IF W=7 THEN 7120 ELSE IF I THEN E$(NO,MO)="0
"
7080 IF INKEY$<>"" THEN 7080
7090 IF INKEY$=CHR$(32) THEN 1090 ELSE PAPER 0:PE
N INT(RND*3)+1
7100 LOCATE 1,1:PRINT "APPUYEZ SUR ESPACE POUR L
A SUITE"
7110 FOR I= 1 TO 200:NEXT:LOCATE 1,1:PRINT SPACE$(
32):GOTO 7090
7120 WINDOW 1,40,22,25:PAPER 1:CLS:WINDOW 1,40,1,25:P
EN 0:LOCATE 11,23
7130 IF S(0)>S(1) THEN PRINT "J'EMPORTE "; ELSE PRI
NT "VOUS EMPORTEZ ";
7140 PRINT "LE JEU":FOR I= 1 TO 4000:NEXT
7150 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>""
THEN CLS:END
7160 PEN INT(RND*2)+2:FOR I=1 TO 100:NEXT:LOCATE 5
,25

```



```

7170 PRINT "POUR REFAIRE UNE PARTIE : TOUCHE R";:
GOTO 7150
7970 '
7980 '      Datas de configuration
7990 '
8000 DATA 0,1,1,0,2,0,0,0,2,35,36,0
8010 DATA 1,1,1,2,0,0,0,2,2,25,24,36
8020 DATA 1,0,1,1,2,0,0,0,2,36,35,47
8030 DATA 0,1,1,0,2,0,2,0,0,35,36,0
8040 DATA 1,1,1,0,2,0,2,0,2,14,15,36
8050 DATA 1,0,1,2,2,0,0,2,0,36,35,15
8060 DATA 0,1,0,1,2,2,0,0,0,26,47,0
8070 DATA 0,1,1,2,1,0,0,0,2,36,58,59
8080 DATA 1,0,0,1,2,0,0,0,0,15,47,0
8090 DATA 0,1,1,0,1,2,2,0,0,26,57,58
8100 DATA 0,0,1,1,2,0,0,0,0,35,36,47
8110 DATA 1,1,0,2,0,2,0,0,2,24,25,26
8120 DATA 1,0,1,2,0,0,0,0,2,36,0,0
8130 DATA 1,0,0,2,2,2,0,0,0,15,0,0
8140 DATA 0,1,1,1,2,2,2,0,0,26,35,0
8150 DATA 0,1,0,2,1,0,0,0,0,28,58,0
8160 DATA 1,0,1,1,0,2,0,2,0,47,48,0
8170 DATA 1,0,0,1,1,2,0,0,0,47,58,0
8180 DATA 0,0,1,1,1,2,0,0,0,47,58,0

```

DE VISU

L'ordinateur demande les noms des joueurs, fait choisir le niveau de difficulté et le jeu commence. "Regardez bien" vous conseille-t-on. Ouvrez donc les yeux et surtout essayez de vous souvenir de ce que vous allez voir.

Un couple de lettres, disons FA pour fixer les idées, apparaît sur l'écran puis s'efface au bout d'une demi-seconde. A ce moment-là cinq carrés bleus se dessinent sur le téléviseur et chacun d'eux contient une paire de lettres différente. Choisissez ¹ la case dans laquelle se trouve la syllabe FA et vous aurez gagné un point de la façon la plus simple qui soit.

Les cinq carrés bleus sont alors effacés et le programme vous conseille une seconde fois de bien regarder. Concentrez vous et notez que la syllabe FA est apparue cette fois encore mais qu'elle a été immédiatement suivie par une deuxième paire de lettres, par exemple MI. Quand les cinq carrés bleus reviendront devant vous, vous désignerez celui qui contient le couple FA. Vous verrez aussitôt l'ordinateur modifier les contenus des cinq carrés et vous aurez alors à indiquer celui dans lequel se trouve la syllabe MI. Voilà, ce n'est pas plus difficile que cela de gagner un nouveau point. Mais les choses se compliquent ; on va vous montrer une troisième syllabe, disons RA, à la suite de FA et MI et vous devrez retrouver, en choisissant successivement les carrés bleus qui conviennent, la suite FA MI RA. Puis la suite FA MI RA SO, puis FA MI RA SO DU, etc.

Tôt ou tard vous vous embrouillerez et vous désignerez une case ne faisant pas partie de la série de syllabes choisie par la machine. Vous céderez alors la place au joueur suivant. Celui-ci devra repartir à zéro et aura à retrouver un couple de lettres, puis deux, puis trois... Ces lettres seront, bien entendu, différentes de celles qui vous avaient été proposées.

■ FIN DU JEU

Le clavier passe entre les mains du joueur suivant à chaque fois que l'on se trompe en choisissant une case. Lorsque tous les participants ont pris leur tour, l'ordinateur affiche le nom du gagnant, c'est-à-dire de celui qui a réussi à "tenir le coup" le plus longtemps possible.

1. Il suffit de déplacer, avec les flèches horizontales, le petit repère vert clignotant jusqu'à ce qu'il se trouve sous la case voulue. A ce moment-là, on appuie sur la barre d'espacement.

Quand on a fait porter sa préférence sur le niveau de difficulté 1, les couples de lettres qui apparaissent sont constitués d'une consonne et d'une voyelle. Il est plus facile de se souvenir de ce type de syllabes. La complexité du niveau 2 est liée au fait que ce sont seulement des consonnes qui sont proposées.

■ PHASES ESSENTIELLES

Sous-programme 3000

Affichages successifs des couples qui devront être retrouvés.

3000 à 3030 M\$ est la chaîne constituée des 40 lettres qui ont été déterminées dans le sous-programme 2000. La fonction MID\$ permet de décomposer M\$ en paires de lettres, paires que l'on fera apparaître à l'image les unes après les autres.

| Variables | Fonction | Valeurs possibles |
|-----------|-------------------------------------|-------------------|
| D | Nombre de couples devant apparaître | 1 à 40 |
| M\$ | Chaîne de lettres à retrouver | Chaîne caractères |

Sous-programme 4000

Des cases sont proposées au joueur.

4000 à 4070 Cinq carrés bleus sont disposés sur l'écran et l'un d'eux contient la syllabe à retrouver.

4080 à 4110 On attend qu'une touche soit enfoncée en rappelant au joueur, par un message de couleur bleue, qu'il doit faire son choix.

4120 à 4140 Le couple désigné s'efface du carré bleu et va se recopier un peu plus bas, à la suite des paires que l'on a déjà trouvées.

4150 à 4190 Lorsqu'une erreur a été commise, le programme nous indique quelle case aurait dû être retenue et le tour passe à un autre joueur.

4200 à 4240 Quand on a recomposé une série complète, le score est incrémenté et l'on rejoint le sous-programme 3000, avec une suite de couples à retrouver augmentée d'une unité.

| Variables | Fonction | Valeurs possibles |
|-----------|--------------------------------|---------------------|
| F | Numéro du couple dans M\$ | 1 à 20 |
| K | Numéro du bon carré | 1 à 5 |
| P\$(J) | Couple affiché dans le carré J | Aléatoire |
| C0 et C1 | Numéro d'une consonne | 1 à 21 |
| V | Numéro d'une voyelle | 1 à 5 |
| C\$(C0) | Consonne numéro C0 | Consonne quelconque |
| V\$(V) | Voyelle numéro V | Voyelle quelconque |
| A | Compteur pour un message | 1 à 40 |
| A\$ | Touche enfoncée | Touche quelconque |
| C | Numéro d'une couleur | 1 ou 2 |
| X | Numéro de la case choisie | 1 à 5 |
| S(B) | Score du joueur numéro B | 0 à 20 |

■ MODIFICATIONS POSSIBLES

Durée du temps d'affichage des syllabes

Cette durée pourrait varier en fonction du niveau de difficulté.

Teneur des couples à retrouver

On pourrait ne pas se limiter à l'alphabet et rajouter des chiffres, des signes de ponctuation, etc.

Transformation des couples en triplettes

Au lieu de retrouver deux lettres à chaque fois, il faudrait en retrouver trois ou, pourquoi pas, quatre.

```

50 '
60 '
70 '           DE  VISU
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    0:PAPER 0
110 PEN 3:LOCATE 2,5:PRINT "NOMBRE DE JOUEURS (MA
X 6 ) ?"
120 A$=INKEY$:IF A$="" THEN X=RND:GOTO 120 ELSE
N=VAL(A$)
130 IF N<1 OR N>6 THEN 120 ELSE PEN 2:FOR I=1 TO
    N
140 LOCATE 2,6+2*I:PRINT "NOM DU JOUEUR";I;:LINE I
INPUT N$(I)
150 N$(I)=LEFT$(N$(I),11):NEXT:PEN 3:LOCATE 2,22
160 PRINT "NIVEAU DE DIFFICULTE ( 1 OU 2 ) ?"
170 A$=INKEY$:IF A$="" THEN 170 ELSE NV=VAL(A$):IF
    NV<1 OR NV>2 THEN 170
180 FOR I=1 TO 5:READ V$(I):NEXT:DIM C$(21):FOR I=1
    TO 21
190 READ C$(I):NEXT:SYMBOL 255,32,112,168,32,32,32,32
,32
200 DATA A,E,I,O,U,B,C,D,F,G,H,J,K,L,M,N,P,Q,R,S,T,V,W,X,Y,Z
970 '
980 '           Decor
990 '
1000 CLS:PAPER 1:WINDOW 32,40,1,3:CLS:WINDOW 1,40,19,
    25
1010 CLS:WINDOW 1,40,1,25:X=2:Y=20:PEN 3:FOR I=1 TO
    N
1020 LOCATE X,Y:PRINT N$(I):LOCATE X+12,Y:PRINT "O"
1030 X=X+22:IF X=46 THEN X=2:Y=Y+2
1040 NEXT:PAPER 0:PEN 2:LOCATE 33,2:PRINT "DE VISU
"
1050 MOVE 0,72:DRAW 639,72,0:MOVE 0,40:DRAW 639,40,0
1060 MOVE 320,112:DRAW 320,0,0:X=3:XP=21:C=2:GOTO 6
    000
1970 '
1980 '           Chaine complete
1990 '
2000 PEN 1:LOCATE 1,1:PRINT N$(B):M$=""
2010 FOR I=1 TO 20:J=INT(RND*21)+1:M$=M$+C$(J)

```

```

2020 IF NV=1 THEN V=INT(RND*5)+1:M$=M$+V$(V)
2030 IF NV=2 THEN C=INT(RND*21)+1:M$=M$+C$(C)
2040 NEXT:C=2:D=0:GOTO 4220
2970 '
2980 '      Les couples apparaissent
2990 '
3000 WINDOW 19,22,10,12:PAPER 1:CLS:WINDOW 1,40,1,25
3010 PEN 0:D=D+1:FOR I=1 TO D:LOCATE 20,11
3020 PRINT MID$(M$,2*I-1,2):FOR J=1 TO 800:NEXT
3030 LOCATE 20,11:PRINT "  ":FOR J=1 TO 200:NEXT:N
EXT
3970 '
3980 '      Le joueur choisit un carre
3990 '
4000 FOR I=5 TO 33 STEP 7:WINDOW I,I+3,10,12:CLS:NE
XT:WINDOW 1,40,1,25
4010 IF INKEY$<>"" THEN 4010 ELSE FOR F=1 TO D:PA
PER 1:PEN 0:K=INT(RND*5)+1
4020 FOR J=1 TO 5:LOCATE 7*J-1,11:IF J<>K THEN 40
40
4030 P$(J)=MID$(M$,2*F-1,2):PRINT P$(J):GOTO 4080
4040 CO=INT(RND*21)+1:C1=INT(RND*21)+1
4050 V=INT(RND*5)+1:PRINT C$(CO);
4060 IF NV=1 THEN PRINT V$(V):P$(J)=C$(CO)+V$(V)
4070 IF NV=2 THEN PRINT C$(C1):P$(J)=C$(CO)+C$(C1)
4080 NEXT:PAPER 0:PEN 1:LOCATE 12,7:PRINT "FAITES
VOTRE CHOIX":A=40
4090 A$=INKEY$:IF A$<>"" THEN 4120 ELSE A=A+1:IF
A<40 THEN 4090
4100 C=3-C:PAPER 0:PEN C:LOCATE 2*F-1,16:PRINT ".."
4110 A=0:LOCATE XP,14:PRINT CHR$(255):GOTO 4090
4120 IF A$<>CHR$(32) THEN GOSUB 5000:GOTO 4090
4130 PAPER 1:LOCATE 7*X-1,11:PRINT "  ":PAPER 0
4140 PEN 2:LOCATE 2*F-1,16:PRINT P$(X)
4150 IF P$(X)=MID$(M$,2*F-1,2) THEN 4200 ELSE LOCAT
E 6,5
4160 PAPER 2:PEN 0:PRINT " C'EST FAUX ":FOR I=1 TO
2000
4170 NEXT:PAPER 1:FOR I=0 TO 15:PEN I
4180 LOCATE 7*K-1,11:PRINT P$(K):SOUND 1,250+5*I,10
4190 IF SQ(1)>127 THEN 4190 ELSE NEXT:FOR I=1 TO
3000:NEXT:GOTO 6000
4200 NEXT:S(B)=D:LOCATE 13-22*(B/2=INT(B/2)),20+2*IN

```



```

T((B-1)/2)
4210 PEN 3:PAPER 1:PRINT S(B):IF D=20 THEN 6000
4220 PAPER 0:WINDOW 1,40,4,18:CLS:WINDOW 1,40,1,25
4230 PEN 2:LOCATE 15,11:PRINT "REGARDEZ BIEN":FOR
I=1 TO 800:NEXT
4240 LOCATE 15,11:PRINT SPACE$(13):FOR I=1 TO 400:N
EXT:GOTO 3000
4970 '
4980 '      Deplacement de la fleche
4990 '
5000 AS=ASC(A$):IF AS<242 OR AS>243 THEN RETURN
5010 PAPER 0:LOCATE XP,14:PRINT " "
5020 IF AS=242 AND XP>7 THEN XP=XP-7:X=X-1
5030 IF AS=243 AND XP<35 THEN XP=XP+7:X=X+1
5040 LOCATE XP,14:PRINT CHR$(255):RETURN
5970 '
5980 '      Joueur suivant
5990 '
6000 PAPER 0:WINDOW 1,40,4,18:CLS:WINDOW 1,40,1,25
6010 IF INKEY$<>" " THEN 6010
6020 LOCATE 1,1:PRINT SPACE$(11):B=B+1:IF B=N+1 THE
N 6050
6030 PEN 2:LOCATE 2,11:PRINT N$(B);" , APPUYEZ SUR
ESPACE"
6040 IF INKEY$<>CHR$(32) THEN 6040 ELSE 2000
6050 J=1:FOR I=1 TO N:IF S(I)>S(J) THEN J=I
6060 NEXT:LOCATE 1,8:PEN 2:FOR I=1 TO N
6070 IF S(I)=S(J) THEN PRINT N$(I);" A GAGNE"
6080 NEXT:P$="UNE AUTRE PARTIE : TOUCHE R"
6090 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>" "
THEN CLS:END
6100 PEN INT(RND*3)+1:LOCATE 4,15:PRINT P$:FOR I=1
TO 100:NEXT:GOTO 6090

```

L'INTRUS

Ce jeu constitue un excellent test pour évaluer vos dons d'observation. L'ordinateur va dessiner sous vos yeux une grille contenant 25 motifs. Parmi ceux-ci, un seul n'est pas reproduit en double exemplaire et vous allez, comme vous vous en doutez, devoir indiquer¹ lequel. De deux choses l'une, ou vous vous trompez, ou vous ne vous trompez pas. Si vous choisissez une case qui fait partie d'une paire, on vous donnera gentiment la preuve que vous avez commis une erreur en faisant clignoter la figurine associée à celle que vous avez retenue. Et l'on vous indiquera, tout aussi gentiment, où se trouve la solution du problème. Si vous avez fait le bon choix, on vous félicitera. C'est tout. Enfin, c'est tout pour les grandes lignes. Voyons les détails. Votre Amstrad, pensant sans doute que ce genre de jeu risquerait de s'éterniser, a décidé de limiter votre temps de réflexion.

Gardez donc constamment un œil sur la droite de l'écran, à l'endroit où le chronomètre égrène les secondes. Et tâchez de faire votre choix avant que le temps qui vous a été attribué ne soit écoulé.

Une remarque pour finir : les caractères graphiques utilisés dans ce programme sont les mêmes que ceux du jeu du Mistigri. Si donc vous faites l'effort de taper l'un de ces deux programmes, vous n'aurez ensuite pas beaucoup de difficulté pour obtenir l'autre.

■ FIN DU JEU

Le programme a prévu la possibilité de faire jouer successivement plusieurs participants (6 au maximum). Chacun d'eux aura à retrouver le motif solitaire dans trois grilles différentes et se verra attribuer un nombre de points directement lié à la rapidité avec laquelle la bonne solution aura été donnée. Naturellement, aucun point ne sera ajouté au score du joueur qui laissera le chronomètre arriver à zéro.

Le niveau de difficulté détermine le temps de réflexion accordé aux partenaires. Plus il est élevé, et plus le temps imparti est réduit.

■ PHASES ESSENTIELLES

Sous-programme 2000

Le joueur choisit une des 25 cases.

1. Il suffit de déplacer, avec les quatre flèches, la petite étoile jaune clignotante jusqu'à ce qu'elle se trouve dans la case voulue. A ce moment-là, on appuie sur la barre d'espacement.

- 2000 Le nom du joueur qui va prendre son tour est affiché en jaune, à droite et en bas du téléviseur.
- 2010 à 2050 La préparation d'une grille commence. L'ordinateur choisit aléatoirement un motif et l'adjoint à douze couples pour arriver au total de 25 figures.
- 2060 à 2070 On bat les cartes pour les mélanger...
- 2080 à 2110 ... puis on les fait apparaître sur un tableau de cinq éléments sur cinq.
- 2120 à 2160 Le programme attend que l'on se décide à choisir une case de l'écran. Lorsque c'est une flèche qui est enfoncée, la petite étoile clignotante change de place.

| Variables | Fonction | Valeurs possibles |
|-------------|--|-------------------|
| D | Numéro du joueur | 1 à 6 |
| N | Nombre de participants | 1 à 6 |
| N\$(D) | Nom des joueurs | Chaîne caractères |
| L | Nombre d'essais par joueur | 1 à 3 |
| B | Temps qui s'écoule | 0 à 150 |
| R | Numéro du dessin solitaire | 1 à 13 |
| A(I,J) | Valeur de la case (I,J) | 1 à 13 |
| S,T,U et V | Coordonnées des cases qui sont échangées | 1 à 5 |
| A | Variable intermédiaire pour l'échange | 1 à 13 |
| G\$(A(I,J)) | Dessin de la case (I,J) | Chaîne caractères |
| E | Compteur pour l'affichage d'un message | 0 à 60 |
| A\$ | Touche enfoncée | Touche quelconque |
| C | Numéro d'une couleur | 0 ou 1 |
| XP et YP | Coordonnées de l'étoile clignotante | 5 à 25 |

Sous-programme 3000

La bonne case a été retrouvée.

3000 à 3010 Le message "c'est juste" est affiché dans un rectangle bleu.

3020 à 3050 Le graphique situé dans la bonne case se colore de différentes façons et une petite musique est entendue. Dès que ceci est terminé, le score du joueur est augmenté du nombre de secondes qui n'ont pas été utilisées.

| Variables | Fonction | Valeurs possibles |
|-----------|--------------------------------|-------------------|
| X et Y | Coordonnées de la case choisie | 1 à 5 |
| S(D) | Score du joueur numéro D | 0 à 450 |

Sous-programme 4000

Le choix a porté sur une mauvaise case.

4000 On fait savoir au joueur qu'il s'est trompé.

4010 à 4020 La case associée à celle qui vient d'être retenue est recherchée.

4030 à 4090 Les deux motifs identiques se mettent à clignoter. Il ne reste plus à l'ordinateur qu'à nous donner, en la faisant apparaître en noir, la case solution.

■ MODIFICATIONS POSSIBLES

Augmentation du nombre de cases

Le jeu gagnerait alors en intérêt, mais aussi en difficulté.

Augmentation du nombre de cases non appariées

Au lieu de chercher un seul dessin qui ne soit associée à aucun autre, il faudrait s'efforcer d'en trouver trois, ou plus.

```

50 '
60 '
70 '           L'INTRUS
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    0:PAPER 0:SYMBOL AFTER 203
110 SYMBOL 204,3,3,3,3,1,1,63,63:SYMBOL 205,192,192,1
    92,192,128,128,252,252
120 SYMBOL 206,3,3,3,3,12,12,48,48:SYMBOL 207,192,192
    ,192,192,48,48,12,12
130 SYMBOL 208,4,6,7,7,7,7,7:SYMBOL 209,0,0,0,128,19
    2,224,240,248
140 SYMBOL 210,7,7,4,255,127,63,31,15:SYMBOL 211,252,
    254,0,255,254,252,248,240
150 SYMBOL 212,0,0,0,255,4,4,63,235:SYMBOL 213,0,0,0,2
    24,0,0,0,129
160 SYMBOL 214,127,63,31,0,0,0,0:SYMBOL 215,255,129
    ,0,0,0,0,0
170 SYMBOL 216,24,56,48,48,24,139,103,31:SYMBOL 217,
    24,28,12,12,24,209,230,248
180 SYMBOL 218,7,63,71,135,27,32,64,64:SYMBOL 219,22
    4,252,226,225,216,4,2,2
190 SYMBOL 220,15,15,7,3,1,97,145,9:SYMBOL 221,224,22
    4,192,128,0,3,4,8
200 SYMBOL 222,5,5,3,1,1,1,1:SYMBOL 223,16,32,32,64,
    64,128,0,0
210 SYMBOL 224,32,80,136,136,4,4,2,2:SYMBOL 225,8,20,
    36,36,64,64,128,128
220 SYMBOL 226,3,7,5,7,7,7,3,3:SYMBOL 227,128,192,64,1
    92,192,192,128,128
230 SYMBOL 228,15,31,63,127,255,63,63,38:SYMBOL 229,
    248,248,252,254,255,252,252,100
240 SYMBOL 230,38,38,38,62,62,62,255,255:SYMBOL 231,
    100,100,100,124,124,124,255,255
250 SYMBOL 232,0,0,0,0,0,15,63,255:SYMBOL 233,0,0,0,0,
    0,1,194,254
260 SYMBOL 234,255,63,15,0,0,0,0:SYMBOL 235,254,194
    ,1,0,0,0,0,0
270 SYMBOL 236,60,60,195,195,195,195,3,3:SYMBOL 237,
    0,0,0,12,14,15,12,12
280 SYMBOL 238,12,12,48,48,48,48,15,15:SYMBOL 239,60,
    63,62,60,252,252,252,248

```

```

290 SYMBOL 240,0,0,0,0,0,0,0,0:SYMBOL 241,14,12,8,24,1
6,48,96,96
300 SYMBOL 242,0,32,31,31,22,16,16,16:SYMBOL 243,192,
192,128,128,128,128,128,128
310 SYMBOL 244,0,0,48,240,48,48,63,31:SYMBOL 245,0,0,
0,0,0,2,252,252
320 SYMBOL 246,13,12,12,12,12,12,0,0:SYMBOL 247,236,1
2,12,12,12,12,0,0
330 SYMBOL 248,0,0,0,1,3,6,255,255:SYMBOL 249,48,96,1
92,128,1,2,252,252
340 SYMBOL 250,6,3,1,0,0,0,0,0:SYMBOL 251,2,1,128,192,
96,48,0,0
350 SYMBOL 252,32,32,127,127,231,231,255,255:SYMBOL
253,4,4,254,254,231,231,255,255
360 SYMBOL 254,127,63,56,24,31,15,7,3:SYMBOL 255,254,
252,28,24,248,240,224,192
370 K=204:DIM G$(13):FOR I=1 TO 13
380 G$(I)=CHR$(K)+CHR$(K+1)+CHR$(8)+CHR$(8)+CHR$(10)+
CHR$(K+2)+CHR$(K+3)
390 K=K+4:NEXT:PEN 3:LOCATE 2,5:PRINT "NOMBRE DE J
OUEURS (MAX 6)"
400 A$=INKEY$:IF A$="" THEN X=RND:GOTO 400
410 N=VAL(A$):IF N<1 OR N>6 THEN 400 ELSE PEN 2
420 FOR I=1 TO N:LOCATE 2,6+2*I:PRINT "NOM DU JOU
EUR";I;
430 LINE INPUT N$(I):N$(I)=LEFT$(N$(I),9):NEXT
440 PEN 3:LOCATE 2,20:PRINT "CHOISISSEZ LA DIFFICU
LTE (DE 1 A 3)"
450 A$=INKEY$:IF A$="" THEN 450 ELSE NV=VAL(A$)
460 IF NV<1 OR NV>3 THEN 450
970 '
980 '      Decor
990 '
1000 CLS:WINDOW 31,40,1,3:PAPER 1:CLS:WINDOW 1,40,1,2
5
1010 PAPER 0:PEN 2:LOCATE 32,2:PRINT "L'INTRUS":LOC
ATE 28,10
1020 PRINT "TEMPS :":PEN 1:FOR I=1 TO N:LOCATE 27,
18+I
1030 PRINT N$(I):NEXT:XP=15:YP=15:X=3:Y=3
1970 '
1980 '      Choix d'une case
1990 '

```



```

2000 D=D+1:IF D=N+1 THEN 6000 ELSE PEN 3:LOCATE
27,18+D:PRINT N$(D)
2010 FOR L=1 TO 3:B=200-50*N*V:PAPER 0:PEN 2:WINDO
W 1,25,1,25
2020 CLS:LOCATE 3,12:PRINT "ESSAI NUMERO";L:K=1:R=I
NT(RND*13)+1
2030 F=0:FOR I=1 TO 5:FOR J=1 TO 5:IF K=R AND F T
HEN K=K+1
2040 A(I,J)=K:K=K+1:IF K=14 THEN K=1:F=1
2050 NEXT:NEXT:FOR I=1 TO 30:S=INT(RND*5)+1:T=INT(R
ND*5)+1
2060 U=INT(RND*5)+1:V=INT(RND*5)+1:A=A(S,T)
2070 A(S,T)=A(U,V):A(U,V)=A:NEXT:CLS:PAPER 1:PEN 2
2080 FOR I=1 TO 5:FOR J=1 TO 5:WINDOW 5*I-3,5*I,5*
J-3,5*J
2090 CLS:NEXT:NEXT:WINDOW 1,40,1,25:FOR I=1 TO 5:FO
R J=1 TO 5
2100 LOCATE 5*I-2,5*J-2:PRINT G$(A(I,J)):NEXT:NEXT
2110 IF INKEY$<>" " THEN 2110 ELSE E=60:C=1
2120 A$=INKEY$:IF A$<>" " THEN 2160 ELSE E=E+1:IF
E<60 THEN 2120
2130 PAPER 0:PEN C:LOCATE 2,1:PRINT "DESIGNEZ LA C
ASE INTRUSE"
2140 B=B-1:PEN 2:LOCATE 35,10:PRINT B:IF B<1 THEN
4060
2150 PAPER 1:PEN 3*C:LOCATE XP,YP:PRINT "*":E=0:C=1
-C:GOTO 2120
2160 IF A$<>CHR$(32) THEN GOSUB 5000:GOTO 2120
2970 '
2980 '      Gagne
2990 '
3000 PAPER 1:PEN 0:WINDOW 28,40,6,8:CLS:IF A(X,Y)<>R
THEN 4000
3010 LOCATE 2,2:PRINT "C'EST JUSTE":FOR I=1 TO 200
0:NEXT
3020 PAPER 0:CLS:WINDOW 1,40,1,25:FOR I=1 TO 15:SOU
ND 1,200-2*I,10
3030 IF SQ(1)>127 THEN 3030 ELSE PAPER 1:PEN 1:LOC
ATE 5*X-2,5*Y-2
3040 PRINT G$(R):NEXT:S(D)=S(D)+B:PAPER 0:PEN 1
3050 LOCATE 36,18+D:PRINT S(D):GOTO 4090
3970 '
3980 '      Perdu

```

```

3990 '
4000 LOCATE 2,2:PRINT "C'EST FAUX":FOR I=1 TO 2000
:NEXT
4010 WINDOW 1,40,1,25:FOR I=1 TO 5:FOR J=1 TO 5
4020 IF (I=X AND J=Y) OR A(I,J)<>A(X,Y) THEN NEXT:NE
XT
4030 PAPER 1:LOCATE XP,YP:PRINT " ":FOR K=1 TO 14:
SOUND 1,400+2*K,10
4040 IF SQ(1)>127 THEN 4040 ELSE PEN K:LOCATE 5*I
-2,5*J-2
4050 PRINT G$(A(I,J)):LOCATE 5*X-2,5*Y-2:PRINT G$(A(X
,Y)):NEXT
4060 FOR I=1 TO 1000:NEXT:PAPER 1:PEN 0:FOR I=1 T
O 5:FOR J=1 TO 5
4070 IF A(I,J)<>R THEN NEXT:NEXT ELSE LOCATE 5*I-2,
5*J-2
4080 PRINT G$(R):FOR I=1 TO 4000:NEXT
4090 PAPER 0:WINDOW 28,40,6,8:CLS:WINDOW 1,40,1,25
4100 NEXT L:PEN 1:LOCATE 27,18+D:PRINT N$(D):GOTO 2
000
4970 '
4980 '      Deplacement etoile
4990 '
5000 AS=ASC(A$):IF AS<240 OR AS>243 THEN RETURN
5010 PAPER 1:PEN 1:LOCATE XP,YP:PRINT "*"
5020 IF AS=242 AND XP>5 THEN XP=XP-5:X=X-1
5030 IF AS=243 AND XP<25 THEN XP=XP+5:X=X+1
5040 IF AS=240 AND YP>5 THEN YP=YP-5:Y=Y-1
5050 IF AS=241 AND YP<25 THEN YP=YP+5:Y=Y+1
5060 PEN 3:LOCATE XP,YP:PRINT "*";PAPER 0:RETURN
5970 '
5980 '      Fin du jeu
5990 '
6000 J=1:FOR I=1 TO N:IF S(I)>S(J) THEN J=I
6010 NEXT:WINDOW 2,25,2,10:PAPER 1:PEN 0:CLS:WINDOW
1,40,1,25
6020 LOCATE 5,3:FOR I=1 TO N:IF S(I)=S(J) THEN PRIN
T N$(I);" A GAGNE"
6030 NEXT
6040 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>""
THEN CLS:END
6050 PEN INT(RND*3)+1:PAPER 0:LOCATE 1,16
6060 PRINT "UNE AUTRE PARTIE : TOUCHE R";:FOR I=1
TO 100:NEXT:GOTO 6040

```

PENDU

Le but du jeu est connu de tous : il faut reconstituer un mot que l'ordinateur a choisi. Supposons qu'il s'agisse du mot ECOLE ; la première lettre de ce mot apparaît en haut de l'écran, suivie de trois tirets et de la dernière lettre de ce même mot. Puisqu'un message clignotant vous invite à taper une lettre au clavier, pressez la touche C. Une petite musique indique que vous êtes sur la bonne voie et cela est confirmé par l'affichage du caractère C à la place du premier tiret. Si ensuite vous tapez les lettres O et L, vous aurez retrouvé le mot ECOLE et vous serez déclaré gagnant.

Si vous choisissez une lettre qui ne se trouve pas dans le mot, disons A, vous aurez droit à quelques notes de musique très graves et l'ordinateur commencera à dessiner un pendu. La première fois il ne s'agira que de la base de la potence, mais ce sera ensuite au tour de la potence elle-même, de la corde, de la tête du pendu, de son corps etc. S'il advient que vous vous soyez trompé dix fois, la totalité du pendu apparaîtra devant vos yeux et l'on vous fera remarquer que vous avez perdu. Puis on vous donnera, tout de même, la bonne réponse. Notez que l'ordinateur, bon prince, affiche dans un coin de l'image les lettres que vous lui avez proposées et qu'il n'a pas retenues. Cela devrait vous éviter de taper deux fois de suite sur une touche qui ne correspond pas à une lettre utilisée dans le mot.

Si on le souhaite, il est possible de jouer à plusieurs avec ce programme. L'un des joueurs choisit, à la place de l'ordinateur, un mot. Dès qu'il l'a entré au clavier, l'écran s'efface et les autres participants doivent alors retrouver ce mot. Là encore, le pendu se dessine au fur et à mesure que l'on se trompe.

Une dernière chose ; le temps de réflexion est limité et vous devrez garder un œil sur le chronomètre. Quand vous verrez qu'il arrive aux alentours de zéro, vous vous dépêcherez de taper une lettre ; sinon l'ordinateur ajoutera d'office, un élément supplémentaire au dessin du pendu.

Note : Positionner la touche CAPS LOCK pour que les lettres proposées à l'ordinateur soient en majuscule.

■ FIN DU JEU

Il n'y a naturellement que deux possibilités, on trouve ou on ne trouve pas le mot.

Le niveau de difficulté est lié à la complexité des mots que l'ordinateur choisit. Au niveau 1, par exemple, il ne propose que des mots d'un usage très courant, et, au niveau 4, il nous demande de reconstituer des mots tout à fait invraisemblables.

■ PHASES ESSENTIELLES

Sous-programme 2000

Analyse de la lettre tapée par le joueur.

2000 à 2020 La phrase "Proposez une lettre" clignote sur l'écran à peu près toutes les secondes. A chaque fois, une unité est soustraite au chronomètre.

2030 à 2040 Le nombre d'essais est enregistré et le programme vérifie si la lettre proposée par le joueur n'a pas été citée précédemment.

2050 à 2080 On regarde si la lettre choisie se trouve à l'intérieur du mot à reconstituer. Si c'est le cas, elle est affichée, prenant la place du tiret correspondant.

2090 à 2110 Si ce n'est pas le cas, elle est écrite dans la case des lettres non valables et une partie du pendu se dessine.

| Variables | Fonction | Valeurs possibles |
|-----------|--|----------------------|
| F | Compteur pour l'affichage du message | 0 à 100 |
| Z | Numéro d'une couleur | 0 ou 1 |
| G | Compteur pour le chronomètre | 0 à 30 |
| D | Drapeau indiquant qu'une lettre est contenue dans le mot ou ne l'est pas | 0 ou 1 |
| A | Nombre d'essais | 0 à 10 et plus |
| L | Longueur du mot à retrouver | 0 à 20 |
| A\$ | Lettre tapée au clavier | caractère quelconque |
| N\$ | Chaîne formée par toutes les lettres déjà trouvées | chaîne caractères |

| Variables | Fonction | Valeurs possibles |
|-----------|------------------------------------|----------------------|
| C | Nombre de lettres justes | 0 à 18 |
| B | Nombre de lettres fausses | 0 à 10 |
| M\$(I) | Lettre numéro I du mot à retrouver | caractère quelconque |

Sous-programme 4000

Dessin des différentes parties du pendu.

- 4000 Base de la potence.
- 4010 Élément vertical de la potence.
- 4020 Élément horizontal de la potence.
- 4030 à 4040 Corde.
- 4050 à 4060 Tête du pendu.
- 4070 à 4090 Corps du pendu.
- 4100 à 4130 Bras du pendu.
- 4140 à 4170 Jambes du pendu.

■ MODIFICATIONS POSSIBLES

Augmentation du nombre de mots connus par l'ordinateur

80 de ceux-ci sont contenus dans le programme et il n'y a pas de difficulté majeure à en ajouter de nouveaux.

Augmentation du temps de réflexion

Le laps de temps accordé à un joueur pour qu'il se décide à presser une touche est très facilement modifiable. On peut même imaginer de supprimer totalement la notion de chronomètre dans ce programme.

```

50 '
60 '
70 '          PENDU
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    0:PAPER 0
110 PEN 2:LOCATE 1,8:PRINT "TOUCHE 1: L'ORDINATEUR
    CHOISIRA UN MOT"
120 LOCATE 1,19:PRINT "TOUCHE 2: L'UN DE VOUS CHO
    ISIT UN MOT"
130 PRINT "QUE LES AUTRES DEVRONT RETROUVER":PE
    N 3
140 A$=INKEY$:IF A$="" THEN X=RND:GOTO 140
150 IF A$="2" THEN 200 ELSE IF A$<>"1" THEN 140
160 LOCATE 1,19:PRINT SPACE$(72):LOCATE 1,11
170 PRINT "CHOISISSEZ LA DIFFICULTE ( DE 1 A 4 )"
180 A$=INKEY$:IF A$="" THEN 180 ELSE NV=VAL(A$):IF
    NV<1 OR NV>4 THEN 180
190 X=INT(RND*20)+1:X=X+(NV-1)*20:FOR I=1 TO X:READ
    M$:NEXT:GOTO 1000
200 LOCATE 1,22:LINE INPUT "MOT A RETROUVER ";M$
210 IF LEN(M$)>20 OR LEN(M$)<3 THEN RUN
970 '
980 '          Decor
990 '
1000 CLS:PAPER 1:WINDOW 32,40,3,5:CLS:WINDOW 19,40,1
    2,14:CLS
1010 WINDOW 19,40,16,20:CLS:WINDOW 19,40,22,24:CLS
1020 WINDOW 1,40,1,25:PEN 3:LOCATE 20,17
1030 PRINT "LETTRES NON VALABLES":PEN 0:LOCATE 2
    0,23
1040 PRINT "NOMBRE D'ESSAIS : 0":PEN 2:PAPER 0:LOC
    ATE 33,4
1050 PRINT " PENDU ":L=LEN(M$):DIM M$(20):FOR I=2 T
    O L-1
1060 M$(I)=MID$(M$,I,1):LOCATE 2*I-1,1:PRINT "-":NEXT
1070 LOCATE 1,1:PRINT LEFT$(M$,1):LOCATE 2*L-1,1
1080 PRINT RIGHT$(M$,1):F=100:G=30
1970 '
1980 '          Le joueur propose une lettre
1990 '
2000 A$=INKEY$:IF A$<>" " THEN 2030 ELSE F=F+1:IF

```

```

F<100 THEN 2000
2010 F=0:Z=1-Z:PAPER 1:PEN Z:LOCATE 20,13:PRINT "PROPOSEZ UNE LETTRE":G=G-1
2020 PAPER 0:PEN 3:LOCATE 28,4:PRINT G:IF G THEN 2000 ELSE A$="":GOTO 2090
2030 D=0:A=A+1:PAPER 1:PEN 0:LOCATE 37,23:PRINT A
2040 IF INSTR(N$,A$) THEN SOUND 1,150,30:A$="":GOTO 2090
2050 FOR I=2 TO L-1:IF A$<>M$(I) THEN 2070 ELSE PAPER 0:PEN 2
2060 LOCATE 2*I-1,1:PRINT A$:N$=N$+A$:D=1:C=C+1:IF C=L-2 THEN 3030
2070 NEXT:FOR I=1 TO 10:SOUND 1,350-200*D-10*I,6
2080 IF SQ(1)>127 THEN 2080 ELSE NEXT:IF D THEN G=30:GOTO 2000
2090 SOUND 1,250,20:B=B+1:PAPER 1:PEN 0:LOCATE 18+2*B,19:PRINT A$
2100 ON B GOSUB 4000,4010,4020,4030,4050,4070,4100,4120,4140,4160
2110 WINDOW 1,40,1,25:G=30:IF B<10 THEN 2000
2970 '
2980 '      Fin du jeu
2990 '
3000 WINDOW 15,21,6,8:PAPER 1:CLS:PEN 0:LOCATE 2,2:PRINT "PERDU"
3010 WINDOW 1,40,1,25:PAPER 0:PEN 3:LOCATE 1,1:PRINT SPACE$(40)
3020 LOCATE 1,1:PRINT "REPONSE : ";M$:GOTO 3040
3030 WINDOW 15,21,6,8:PAPER 1:CLS:PEN 0:LOCATE 2,2:PRINT "GAGNE"
3040 WINDOW 1,40,1,25:FOR I=1 TO 3000:NEXT
3050 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>"" THEN CLS:END
3060 PAPER 0:PEN INT(RND*3)+1:LOCATE 15,10
3070 PRINT "UNE AUTRE PARTIE: TOUCHE R":FOR I=1 TO 80:NEXT:GOTO 3050
3970 '
3980 '      Dessin du pendu
3990 '
4000 PAPER 1:WINDOW 2,15,24,25:CLS:RETURN
4010 PAPER 1:WINDOW 3,3,3,23:CLS:RETURN
4020 PAPER 1:WINDOW 3,11,3,3:CLS:RETURN
4030 MOVE 48,243:DRAW 143,352,3:DRAWR 0,-48,3:MOVER

```



```

2,0
4040 DRAWR 0,48,3:DRAW 48,240,3:RETURN
4050 PAPER 2:WINDOW 9,10,7,8:CLS:PLOT 136,292,0
4060 PLOT 152,292,0:MOVE 136,280:DRAWR 10,0,0:RETUR
N
4070 MOVE 143,272:DRAWR 0,-8,2:MOVER 2,0
4080 PAPER 3:WINDOW 9,10,10,15:CLS
4090 MOVE 128,192:DRAWR 32,0,1:RETURN
4100 MOVE 128,256:DRAWR -32,-80,3:DRAWR -8,-2,3:DRA
WR 8,-1,3
4110 DRAWR -6,-10,3:DRAWR 4,0,3:DRAW 128,248,3:RETU
RN
4120 MOVE 160,256:DRAWR 32,-80,3:DRAWR 8,-2,3:DRAWR
-8,-1,3
4130 DRAWR 6,-10,3:DRAWR -4,0,3:DRAW 160,248,3:RETU
RN
4140 MOVE 128,160:DRAWR 0,-96,2:DRAWR -8,-16,2:DRAW
R 4,0,2
4150 DRAWR 8,16,2:DRAWR 0,94,2:RETURN
4160 MOVE 159,160:DRAWR 0,-96,2:DRAWR 8,-16,2:DRAW
R -4,0,2
4170 DRAWR -8,16,2:DRAWR 0,94,2:RETURN
4970 '
4980 '      Mots de l'ordinateur
4990 '
5000 DATA SUCRE,VOITURE,BALANCE,ECOLE,RENARD,ENFA
NT,GLACE,ARDOISE
5010 DATA LIVRE,CERCLE,POMME,BRIQUE,ORANGE,FOUDRE,
ARMOIRE,MUSIQUE
5020 DATA CHAISE,NAVIRE,CABINE,TIGRE,APPETIT,CAFETI
ERE,FAUVETTE
5030 DATA DECIMETRE,GOUVERNEUR,OMBRAGE,DEGRADATI
ON,OSSATURE
5040 DATA ARCHIDUC,FEUILLETON,BANDEROLLE,FAINEANT
,ESCALIER
5050 DATA ARRIMAGE,DECAGONE,CHAMPAGNE,GAZOMETRE,A
RISTOCRATE
5060 DATA HARMONICA,ECHANTILLON,CARICATURE,ELECTR
OSCOPE,COAGULATION
5070 DATA APPRECIATION,DEMINERALISATION,BARYTON,H
IPPOCAMPE
5080 DATA BRANDEBOURG,DESSICCATION,HEMISPHERIQUE,
CAHOTEMEMT

```

5090 DATA DEHARNACHEMENT,GRAPHOLOGUE,HOMONYME,CL
 AVECINISTE
 5100 DATA GALACTOMETRE,BELLIGERANT,ECCLESIASTIQU
 E,DYNAMITAGE
 5110 DATA ASSUJETTISSEMENT,ZIRCONIUM,CASTANEACEES
 ,FORMALDEHIDE
 5120 DATA BUTYROMETRE,SYNALLAGMATIQUE,TRANSGANG
 ETIQUE,EMPHYTEOTIQUE
 5130 DATA ZINGLBERACEES,BALENOPTERE,XEROPHTALMIE,
 TETRASYLLABIQUE
 5140 DATA CEPHALOTHORAX,ZINCOGRAVURE,ECHINODERME
 ,ASPARAGINEES
 5150 DATA FLUORESCINE,ELLIPSOGRAPHE,HYPOCONDIAQ
 UE
 5160 DATA CARBORUNDUM,HIPPOCASTANEES

LA DERNIÈRE ÉTOILE

Pressez la touche 3 quand l'ordinateur vous demande le nombre maximal d'étoiles que l'on peut enlever. Puis désignez ¹ l'une des 18 étoiles qui sont alignées dans un rectangle bleu. Il y a de fortes chances pour que l'ordinateur vous rappelle à l'ordre en faisant clignoter un message en bas de l'écran. "3 étoiles au maximum" vous dit-il dans diverses couleurs. Vous avez effectivement le droit de désigner l'étoile de votre choix, mais à condition que ce soit l'une des trois dernières, la seizième, la dix-septième ou la dix-huitième donc. Choisissez, par exemple, l'avant-dernière. Pas de problème, cette fois vous avez respecté la logique du jeu et les deux dernières étoiles s'effacent. Si vous aviez retenu la seizième, vous l'auriez fait disparaître, ainsi que ses deux suivantes. Et si vous aviez choisi la dernière, elle seule aurait été effacée. D'accord ?

C'est maintenant au tour de l'ordinateur de jouer. Il fait semblant de réfléchir un peu pour la forme, puis enlève lui-même, après les avoir fait clignoter, une, deux ou trois étoiles. Naturellement, la règle du jeu est aussi valable pour lui et la ou les étoiles qu'il fait disparaître doivent obligatoirement être les dernières de la série.

C'est à vous de jouer à nouveau. Indiquez l'une des trois dernières étoiles de la suite et elle s'effacera, ainsi que ses suivantes éventuelles. L'ordinateur prendra alors le relais et enlèvera, selon son humeur, un nombre d'étoiles variant entre un et trois.

Reste maintenant à savoir combien de temps durera ce manège. Pas longtemps, car le nombre d'étoiles décroît rapidement et tôt ou tard l'instant fatidique, pour vous ou pour l'ordinateur, arrivera. Il faudra bien que l'un de vous n'ait pas d'autre choix que de prendre la dernière étoile. Et il aura perdu.

Si vous voulez gagner vous avez donc à résoudre le problème suivant : comment faire en sorte qu'au dernier moment, l'ordinateur n'ait rien d'autre à faire que de prendre la dernière étoile.

En début de jeu, lorsque le programme vous demande le maximum d'étoiles qu'il sera possible d'enlever, vous pourrez choisir une autre valeur que 3, pourvu qu'elle soit comprise entre 2 et 9. Bien entendu, le principe du jeu restera le même, et le perdant sera toujours celui à qui reviendra la dernière étoile.

1. Il suffit de déplacer, avec les flèches horizontales, le petit repère clignotant en vert et bleu jusqu'à ce qu'il se trouve sous l'étoile voulue. A ce moment-là, on appuie sur la barre d'espacement.

■ FIN DU JEU

Une partie est terminée dès que le joueur ou l'ordinateur ne peut plus faire autrement que de prendre la dernière étoile. Un point est alors ajouté au score du vainqueur et le jeu reprend. Le nombre maximal d'étoiles que l'on peut supprimer en un coup est modifiable à ce moment-là.

En fonction du niveau de difficulté choisi, on a devant soi un adversaire redoutable (niveau 3) ou un adversaire très déconcentré (niveau 1).

■ PHASES ESSENTIELLES

Sous-programme 2000

Le joueur choisit une étoile.

2000 à 2020 Le programme attend que le joueur se décide à désigner une étoile. Toutes les secondes, la petite flèche qui sert de repère change de couleur.

2030 à 2040 Lorsque la barre d'espacement est enfoncée, le programme prend connaissance de l'étoile choisie.

2050 à 2060 Si la position de cette étoile ne respecte pas la règle du jeu, l'ordinateur prend soin de l'indiquer au joueur.

2070 à 2090 Mais si le coup est valable, les étoiles correspondantes sont effacées. Lorsque le joueur a pris la dernière étoile, le jeu s'arrête.

| Variables | Fonction | Valeurs possibles |
|-----------|--------------------------------------|-------------------|
| E | Compteur pour l'affichage du message | 0 à 100 |
| C | Numéro d'une couleur | 1 ou 2 |
| XP | Abscisse du repère clignotant | 4 à 38 |
| A\$ | Touche enfoncée | Touche quelconque |
| X | Numéro de l'étoile désignée | 1 à 18 |

| Variables | Fonction | Valeurs possibles |
|-----------|------------------------------------|-------------------|
| M | Nombre d'étoiles qui s'effacent | 2 à 9 |
| N | Nombre d'étoiles qui restent | 0 à 18 |
| A | Nombre maximum d'étoiles à effacer | 2 à 9 |

Sous-programme 3000

C'est à l'ordinateur de jouer.

3000 à 3010 L'ordinateur perd un peu de temps et, s'il s'aperçoit qu'il a perdu, arrête le jeu.

3020 à 3070 Le programme effectue les calculs qui lui permettront de gagner ou de perdre (en fonction du niveau de difficulté et de l'habileté du joueur). Il fait ensuite clignoter les étoiles qu'il compte supprimer et, une fois son coup réalisé, regarde s'il ne doit pas se déclarer vainqueur de la partie.

| Variables | Fonction | Valeurs possibles |
|-----------|-----------------------------|----------------------|
| NV | Niveau de difficulté | 1 à 3 |
| E\$ | Chaîne formée de 18 étoiles | chaîne de caractères |

■ MODIFICATIONS POSSIBLES

Modification du nombre d'étoiles en début de partie

Ce nombre a été fixé à 18 mais pourrait sans problème être diminué.

Changement de la règle du jeu

Le vainqueur pourrait être, par exemple, non plus celui qui laisserait la dernière étoile à l'autre, mais celui qui la prendrait.

Jeu pour deux participants

A tour de rôle, chacun choisirait une étoile. L'ordinateur ne serait là que pour vérifier que l'on ne prend pas n'importe quelle étoile et pour tenir les scores à jour.

```
50 '
60 '
70 '          LA  DERNIERE ETOILE
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    0:PAPER 0
110 PEN 2:LOCATE 1,5:PRINT "EN CHOISSANT UNE ET
OILE , ON LA FAIT"
120 PRINT "DISPARAITRE , AINSI QUE SES SUIVANTES
."
130 PRINT:PRINT "LE PERDANT EST CELUI QUI PREND
LA ...."
140 PRINT "... DERNIERE ETOILE ."
150 A$=INKEY$:IF A$<>"" THEN GOTO 180 ELSE X=RND
160 FOR I=1 TO 50:NEXT:PEN INT(RND*3)+1:LOCATE 1,2
0
170 PRINT "CHOISISSEZ LA DIFFICULTE ( DE 1 A 3 )":
GOTO 150
180 NV=VAL(A$):IF NV<1 OR NV>3 THEN 150
190 SYMBOL 253,0,0,32,112,168,32,32,32:SYMBOL 254,1,0
,0,3,0,0,1,0
200 SYMBOL 255,36,168,112,254,112,168,36,32
970 '
980 '      Decor
990 '
1000 CLS:PAPER 1:WINDOW 24,40,1,3:CLS:WINDOW 21,36,1
6,18
1010 CLS:WINDOW 23,34,20,22:CLS:WINDOW 2,39,9,11:CLS
1020 WINDOW 1,40,1,25:PEN 3:LOCATE 22,17:PRINT "ORD
INATEUR  0"
1030 PEN 2:LOCATE 24,21:PRINT "JOUEUR  0":PAPER 0
1040 LOCATE 25,2:PRINT "DERNIERE ETOILE":LOCATE 2,
24
1050 PEN 1:PRINT "ON PEUT ENLEVER  ETOILES AU M
AXIMUM"
```

```

1060 FOR I=1 TO 18:E$=E$+CHR$(254)+CHR$(255):NEXT
1070 C=1:PAPER 1:PEN 2:LOCATE 3,10:PRINT E$
1080 A$=INKEY$:IF A$<>" " THEN 1110 ELSE PAPER 0:P
EN C
1090 LOCATE 4,6:PRINT "NOMBRE MAXIMUM D'ETOILES A
ENLEVER"
1100 FOR I=1 TO 400:NEXT:C=1-C:GOTO 1080
1110 A=VAL(A$):IF A<2 THEN 1080 ELSE N=18:E=100:XP
=38:X=18:C=1
1120 LOCATE 4,6:PRINT SPACE$(34):PEN 1:LOCATE 17,24
:PRINT A
1970 '
1980 '      Le joueur joue
1990 '
2000 PAPER 0:PEN 1:LOCATE 7,6:PRINT "CHOISISSEZ UN
E ETOILE"
2010 A$=INKEY$:IF A$<>" " THEN 2030 ELSE E=E+1:IF
E<100 THEN 2010
2020 PEN C:LOCATE XP,12:PRINT CHR$(253):C=3-C:E=0:G
OTO 2010
2030 IF A$<>CHR$(32) THEN GOSUB 4000:GOTO 2010
2040 M=N-X+1:IF M<=A THEN 2070 ELSE FOR I=2 TO 1
3:SOUND 1,400+10*I,10
2050 IF SQ(1)>127 THEN 2050 ELSE PEN I:LOCATE 17,2
4
2060 PRINT A;"ETOILES AU MAXIMUM":NEXT:GOTO 2010
2070 IF M<1 THEN 2010 ELSE PAPER 1:LOCATE 2*X+1,1
0
2080 PRINT SPACE$(2*M):PAPER 0:LOCATE XP,12:PRINT
" "
2090 SOUND 1,200,5:N=X-1:IF N=0 THEN 5000
2970 '
2980 '      L'ordinateur joue
2990 '
3000 PAPER 0:PEN 2:LOCATE 7,6:PRINT "CHUT ... JE RE
FLECHIS"
3010 FOR I=1 TO 2000:NEXT:IF N=1 THEN 5020
3020 M=(N-1) MOD (A+1):IF N<A+2 THEN 3040
3030 IF M=0 OR RND>0.5*N-0.5 THEN M=INT(RND*A)+1
3040 N=N-M:PAPER 1:PEN 2:FOR I=1 TO 5:LOCATE 2*N+
3,10
3050 PRINT LEFT$(E$,2*M):FOR J=1 TO 50:NEXT
3060 LOCATE 2*N+3,10:PRINT SPACE$(2*M):FOR J=1 TO

```



```

50:NEXT
3070 NEXT:X=N:XP=2*X+2:IF N>1 THEN 2000 ELSE 5000
3970 '
3980 '   Deplacement repere
3990 '
4000 AS=ASC(A$):IF AS<242 OR AS>243 THEN RETURN
4010 PAPER 0:LOCATE XP,12:PRINT " "
4020 IF AS=242 AND XP>4 THEN XP=XP-2:X=X-1
4030 IF AS=243 AND XP<38 THEN XP=XP+2:X=X+1
4040 PEN 2:LOCATE XP,12:PRINT CHR$(253):RETURN
4970 '
4980 '   Fin du jeu
4990 '
5000 WINDOW 7,27,5,7:PAPER 1:CLS:PEN 0:LOCATE 4,2
5010 PRINT "VOUS AVEZ PERDU":T=T+1:GOTO 5040
5020 WINDOW 7,27,5,7:PAPER 1:CLS:PEN 0:LOCATE 4,2
5030 PRINT "VOUS AVEZ GAGNE":U=U+1
5040 FOR I=1 TO 2000:NEXT:SOUND 1,200,5
5050 WINDOW 1,40,1,25:PEN 3:LOCATE 33,17:PRINT T
5060 PEN 2:LOCATE 31,21:PRINT U:IF INKEY$<>"" THEN
5060
5070 A$=INKEY$:IF A$<>"" THEN 5100 ELSE E=E+1:IF
E<60 THEN 5070
5080 PAPER 0:PEN INT(RND*3)+1:LOCATE 2,14
5090 PRINT "APPUYEZ N'IMPORTE OU POUR LA SUITE":E
=0:GOTO 5070
5100 PAPER 0:LOCATE 2,14:PRINT SPACE$(34)
5110 WINDOW 7,27,5,7:CLS:WINDOW 1,40,1,25:GOTO 1070

```

POUSSE - POUSSE

Les vingt-quatre premières lettres de l'alphabet apparaissent dans une grille de cinq cases sur cinq. Ces lettres sont disposées de façon absolument quelconque et tout votre travail va consister à les remettre dans l'ordre habituel. Voyons comment le clavier devra être utilisé.

Supposons que la configuration initiale des deux premières lignes soit la suivante :

| | | | | |
|---|---|---|---|---|
| L | B | D | N | C |
| A | F | M | J | |

Vous désignez ¹ le caractère C et celui-ci prend automatiquement la place de la case vide.

| | | | | |
|---|---|---|---|---|
| L | B | D | N | |
| A | F | M | J | C |

Puis vous choisissez la lettre L et la première ligne se déplace d'un cran sur la droite.

| | | | | |
|---|---|---|---|---|
| | L | B | D | N |
| A | F | M | J | C |

Il ne vous restera plus qu'à indiquer la lettre A et celle-ci rejoindra sa place définitive, en haut et à gauche de l'écran.

| | | | | |
|---|---|---|---|---|
| A | L | B | D | N |
| | F | M | J | C |

Voyons le B maintenant : désignez la lettre F et elle rejoindra la case vide.

1. Il suffit de déplacer, avec les quatre flèches, le petit pavé jaune clignotant jusqu'à ce qu'il se trouve dans la case voulue. A ce moment-là, on appuie sur la barre d'espacement.

| | | | | |
|---|---|---|---|---|
| A | L | B | D | N |
| F | | M | J | C |

Portez alors votre choix sur le caractère L. Il se retrouvera immédiatement dans le carré du dessous.

| | | | | |
|---|---|---|---|---|
| A | | B | D | N |
| F | L | M | J | C |

Choisissez la lettre B et vous aurez alors réussi à écrire les lettres A et B à leurs emplacements définitifs.

| | | | | |
|---|---|---|---|---|
| A | B | | D | N |
| F | L | M | J | C |

Voilà, il ne vous reste plus qu'à continuer avec les autres lettres de l'alphabet pour que chacune retrouve sa place normale.

■ FIN DU JEU

Une partie arrive à son terme quand toutes les lettres ont trouvé leur place et que la case libre est située en bas et à droite de la grille.

Le temps que prend l'ordinateur pour mélanger les lettres est variable. Il est directement proportionnel au niveau de difficulté choisi par le joueur.

■ PHASES ESSENTIELLES

Sous-programme 1000

Mélange des lettres de la grille.

1000 à 1050 Le décor est planté et l'ordinateur place les lettres de l'alphabet dans le tableau A\$, dans l'ordre normal.

- 1060 à 1090 La case libre est déplacée d'un cran dans l'une des quatre directions possibles. C'est ce mouvement, répété un certain nombre de fois, qui va engendrer le désordre dans le tableau.
- 1100 à 1130 Le tableau A\$ est maintenant constitué par des lettres rangées de façon aléatoire. Il est affiché sur le téléviseur, en même temps que le message d'attente est effacé.

| Variables | Fonction | Valeurs possibles |
|-----------|--|----------------------|
| K | Code ASCII d'une lettre | 65 à 90 |
| U et V | Coordonnées de la case libre | 1 à 5 |
| N | Niveau de difficulté | 1 à 3 |
| X et Y | Coordonnées de la prochaine case libre | 1 à 5 |
| A\$(X,Y) | Lettre contenue dans la case X,Y | lettre de l'alphabet |

Sous-programme 2000

Le joueur choisit une lettre.

- 2000 à 2030 Un message dont la couleur change de temps en temps, indique au joueur qu'il doit se décider à choisir une lettre.
- 2040 à 2050 L'ordinateur regarde si l'on enfonce une flèche ou si l'on choisit la case vide.
- 2060 à 2070 Si la lettre retenue ne se trouve pas sur la ligne ou sur la colonne de la case vide, le coup est ignoré.
- 2080 à 2100 La lettre désignée subit une translation vers le carré libre, entraînant, lorsque cela doit se faire, d'autres lettres avec elle.
- 2110 à 2130 C'est la même chose que précédemment, mais il s'agit d'un décalage dans le sens vertical.

2140 à 2150 L'ordinateur visualise les cases une par une et cela lui permet de savoir si la fin du jeu est atteinte.

| Variables | Fonction | Valeurs possibles |
|-----------|--|-------------------|
| E | Compteur pour l'affichage du message | 0 à 60 |
| XP et YP | Coordonnées du point lumineux | 6 à 22 et 7 à 23 |
| A\$ | Touche enfoncée | Touche quelconque |
| X , Y | Coordonnées de la case désignée | 1 à 5 |
| F | Signe de la différence entre les abscisses puis les ordonnées de la case pointée et de la case libre | 1 ou - 1 |

■ MODIFICATIONS POSSIBLES

Changement de la configuration de la grille

Le jeu se compliquerait un peu si l'on avait devant les yeux une grille de 6 carrés sur 4.

Introduction d'un compteur qui totaliserait le nombre d'essais

On ne serait alors déclaré vainqueur qu'à condition de ne pas dépasser un certain seuil.

```

50 '
60 '
70 '          POUSSE - POUSSE
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    0:PAPER 0
110 PEN 2:LOCATE 3,10:PRINT "VOUS DEVEZ RECONSTIT
UER L'ALPHABET EN"
120 PRINT:PRINT "POUSSANT DES LETTRES DANS LA CA
SE VIDE."
130 A$=INKEY$:IF A$<>"" THEN 160 ELSE E=E+1:X=RND
140 IF E<50 THEN 130 ELSE E=0:PEN INT(RND*3)+1:LOC

```

```

ATE 1,20
150 PRINT "CHOISISSEZ LA DIFFICULTE ( DE 1 A 3
)":GOTO 130
160 N=VAL(A$):IF N<1 OR N>3 THEN 130 ELSE SYMBOL
255,0,0,0,28,28,28,0,0
970 '
980 ' Decor
990 '
1000 CLS:WINDOW 26,40,3,5:PAPER 1:CLS:WINDOW 2,24,3,2
5
1010 CLS:WINDOW 3,23,4,24:PAPER 0:CLS:WINDOW 1,40,1,2
5
1020 PEN 2:LOCATE 27,4:PRINT "POUSSE-POUSSE":PEN
3
1030 LOCATE 6,10:PRINT "UNE MINUTE ....":LOCATE 6,1
4
1040 PRINT "JE MELANGE":LOCATE 9,16:PRINT "LES LE
TTRES."
1050 K=64:FOR J=1 TO 5:FOR I=1 TO 5:K=K+1:A$(I,J)=C
HR$(K)
1060 NEXT:NEXT:U=5:V=5:J=2:FOR K=1 TO 10+30*(N-1)
1070 I=INT(RND*4):IF ABS(J-I)=2 THEN 1070 ELSE X=U
+(I=3)-(I=1)
1080 Y=V+(I=0)-(I=2):IF A$(X,Y)="" THEN 1070
1090 A$(U,V)=A$(X,Y):A$(X,Y)="Y":U=X:V=Y:J=I:NEXT
1100 WINDOW 3,23,4,24:CLS:PAPER 1:PEN 2:FOR J=1 TO
5
1110 FOR I=1 TO 5:WINDOW 4*I,4*I+2,4*J+1,4*J+3:CLS
1120 LOCATE 2,2:PRINT A$(I,J):NEXT:NEXT
1130 WINDOW 1,40,1,25:LOCATE 4*U+1,4*V+2:PRINT " "
970 '
980 ' Une lettre est designee
990 '
2000 E=60:XP=14:YP=15:X=3:Y=3:C=2
2010 A$=INKEY$:IF A$<>"" THEN 2040 ELSE E=E+1:IF
E<60 THEN 2010
2020 PAPER 0:PEN C:LOCATE 3,1:PRINT "CHOISISSEZ UN
E LETTRE":C=5-C
2030 PAPER 1:PEN (C-2)*3:LOCATE XP,YP:PRINT CHR$(25
5):E=0:GOTO 2010
2040 IF A$<>CHR$(32) THEN GOSUB 3000:GOTO 2010
2050 IF A$(X,Y)="Y" THEN 2010 ELSE FOR I=1 TO 5
2060 IF A$(X,I)="Y" THEN 2110 ELSE NEXT:FOR I=1 TO

```



```

5
2070 IF A$(I,Y)<>"Y" THEN NEXT:GOTO 2010
2080 PEN 2:PAPER 1:LOCATE 4*X+1,4*Y+2:PRINT " ":F=
SGN(U-X)
2090 FOR I=U TO X+F STEP -F:LOCATE 4*I+1,4*Y+2:PR
INT A$(I-F,Y)
2100 A$(I,Y)=A$(I-F,Y):NEXT:A$(X,Y)="Y":GOTO 2140
2110 PEN 2:PAPER 1:LOCATE 4*X+1,4*Y+2:PRINT " ":F=
SGN(V-Y)
2120 FOR I=V TO Y+F STEP -F:LOCATE 4*X+1,4*I+2:PR
INT A$(X,I-F)
2130 A$(X,I)=A$(X,I-F):NEXT:A$(X,Y)="Y"
2140 U=X:V=Y:K=65:FOR J=1 TO 5:FOR I=1 TO 5
2150 IF A$(I,J)<>CHR$(K) THEN 2010 ELSE K=K+1:NEXT:
NEXT:GOTO 4000
2970 '
2980 '      Deplacement point lumineux
2990 '
3000 AS=ASC(A$):IF AS<240 OR AS>243 THEN RETURN
3010 PAPER 1:LOCATE XP,YP:PRINT " "
3020 IF AS=242 AND X>1 THEN XP=XP-4:X=X-1
3030 IF AS=243 AND X<5 THEN XP=XP+4:X=X+1
3040 IF AS=240 AND Y>1 THEN YP=YP-4:Y=Y-1
3050 IF AS=241 AND Y<5 THEN YP=YP+4:Y=Y+1
3060 PEN 3:LOCATE XP,YP:PRINT CHR$(255):RETURN
3970 '
3980 '      Fin du jeu
3990 '
4000 WINDOW 25,40,11,13:PAPER 1:CLS:SOUND 1,200,20:P
EN 0
4010 LOCATE 1,2:PRINT "VOUS AVEZ GAGNE":FOR I=1 T
O 3000:NEXT
4020 IF INKEY$<>" " THEN 4020 ELSE WINDOW 1,40,1,25
4030 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>" "
THEN CLS:END
4040 PAPER 0:LOCATE 1,1:PRINT "UNE AUTRE PARTIE:
TOUCHE R"
4050 PEN INT(RND*3)+1:FOR I=1 TO 300:NEXT:GOTO 403
0

```

LA MARE AUX DIABLES

Donnez à l'ordinateur qui vous le demande les noms des deux joueurs et choisissez pour commencer le niveau de difficulté 1. Puis regardez bien la grille qui s'est dessinée sur le téléviseur. Ses cases sont vides dans leur grande majorité, mais certaines contiennent des poissons, il y en a cinq ; et deux autres contiennent des têtes de diables. Repérez les unes et les autres car elles vont s'effacer au bout de quelques secondes.

Vous voyez alors quatre points d'interrogation qui clignotent sur la grille. Désignez ¹ l'un d'eux et, suivant votre perspicacité ou votre chance (votre malchance ?), vous verrez... ce que vous verrez. En effet, derrière les points d'interrogation, peuvent se cacher :

- *un poisson* : vous avez gagné un point et il vous faudra retrouver les autres ;
- *une case vide* : vous n'avez rien gagné du tout, mais vous pouvez continuer à chercher les poissons ;
- *un diable* : non seulement vous n'avez pas marqué de point, mais le tour passe illico à l'autre joueur.

Il faut savoir tout de même que le poisson et le diable n'ont pas été placés n'importe où par l'ordinateur. Le poisson est l'un de ceux qui se trouvaient sur la grille initiale et vous avez eu le temps d'apercevoir le diable un court instant avant que tout ne soit remplacé par les quatre points d'interrogation. Ainsi, si votre mémoire visuelle est bonne, vous ne vous laisserez pas tenter par le diable et vous ne choisirez que les cases qui contiennent des poissons. D'accord ?

■ FIN DU JEU

Le jeu se termine quand l'un des joueurs a atteint le score de 15 points. Chaque participant, à moins qu'il ne tombe sur une tête de diable, a cinq essais pour retrouver la totalité des poissons.

En choisissant le niveau de difficulté 1, les quatre carrés proposés par l'ordinateur contiennent un poisson, deux cases vides et un diable. Au niveau 2, vous aurez affaire à un poisson, une case vide et

1. Il suffit de déplacer, avec les quatre flèches, le petit carré clignotant jusqu'à ce qu'il se trouve dans la case voulue. A ce moment-là, on appuie sur la barre d'espacement.

deux diables. Et au niveau 3, vous n'aurez le choix qu'entre un poisson et trois diables.

■ PHASES ESSENTIELLES

Sous-programme 1000

Dessin de la grille et répartition des cases.

1000 à 1040 Une grille de 11 carrés sur 7 est dessinée et les noms des joueurs sont affichés en bas de l'écran.

1050 Le contenu de toutes les cases est annulé.

1060 à 1080 Cinq poissons apparaissent à des emplacements différents.

1090 à 1110 Puis c'est au tour de 2, 4 ou 6 diables de se montrer.

1120 à 1160 Une attente de quelques secondes est programmée et les figurines disparaissent de la grille.

| Variables | Fonction | Valeurs possibles |
|--------------|---------------------------------|---|
| N\$(0) | Nom du joueur numéro 0 | Chaîne caractères |
| N\$(1) | Nom du joueur numéro 1 | Chaîne caractères |
| A(I,J) | Contenu de la case I,J | 0 : Case libre 1 : Poisson 2 : Diable |
| X(I) et Y(I) | Coordonnées d'un poisson | 1 à 11 et 1 à 7 |
| E\$ | Dessin d'un poisson | Chaîne caractères |
| NV | Niveau de difficulté | 1 à 3 |
| F\$ | Dessin d'un diable | Chaîne caractères |
| H\$ | Effacement d'un dessin | Chaîne caractères |
| XP et YP | Coordonnées du pavé clignotant | 64 à 544 et 48 à 336 |
| U et V | Coordonnées de la case désignée | 1 à 11 et 1 à 7 |
| Z(I) et T(I) | Coordonnées d'un diable | 1 à 11 et 1 à 7 |
| C | Nombre de coups restant à jouer | 1 à 5 |

Sous-programme 2000

Détermination des cases à proposer au joueur.

- 2000 En fonction du niveau de difficulté, le programme effectue le branchement adéquat.
- 2010 à 2030 Deux cases vides sont choisies et leur contenu est porté à 3 pour que l'ordinateur les repère.
- 2040 à 2050 Une seule case vide est recherchée et on lui affecte la valeur 3. Une deuxième case est tirée au sort parmi celles qui contiennent un diable et son contenu est porté à 5.
- 2060 à 2070 Au niveau de difficulté le plus élevé, on n'a plus besoin de cases vides et c'est pourquoi les coordonnées de deux diables sont nécessaires. Les cases correspondantes se voient attribuer la valeur 5.
- 2080 à 2110 Il ne reste qu'à compléter le tableau avec un poisson (case portée au niveau 4) et par un diable (case égale à 5).

| Variables | Fonction | Valeurs possibles |
|-----------|--|-------------------|
| M et N | Coordonnées de la première case | 1 à 11 et 1 à 7 |
| P et Q | Coordonnées de la deuxième case | 1 à 11 et 1 à 7 |
| X et Y | Coordonnées du poisson | 1 à 11 et 1 à 7 |
| Z et T | Coordonnées du diable | 1 à 11 et 1 à 7 |
| E | Compteur pour l'affichage d'un message | 0 à 80 |
| G | Numéro d'une couleur | 2 ou 3 |

Sous-programme 3000

Le joueur choisit une case.

- 3000 à 3040 Un message rappelle au joueur qu'il doit choisir une case et, en même temps, quatre points d'interrogation clignotent sur le téléviseur.

- 3050 à 3060 L'ordinateur vérifie que la case retenue est bien l'une de celles qu'il a proposées.
- 3070 Si la case désignée ne contient ni un poisson ni un diable, le chiffre 0 est affiché à l'endroit désigné par le joueur.
- 3080 à 3100 Un poisson est attrapé et son graphisme clignote dans différentes couleurs.
- 3110 à 3120 C'est sur le diable que l'on a jeté son dévolu. La tête de celui-ci se met à clignoter en noir et une musique grave résonne.
- 3130 à 3180 Le score du joueur et le nombre de coups qu'il lui reste à tenter sont affichés. Puis les contenus des quatre cases qui ont été concernées dans la phase de jeu reprennent leurs valeurs initiales. Pour finir, les points d'interrogation disparaissent de la grille.

| Variables | Fonction | Valeurs possibles |
|-----------|-----------------------------------|-------------------|
| A\$ | Touche enfoncée | Touche quelconque |
| F | Numéro du joueur | 0 ou 1 |
| I\$ | Dessin d'un point d'interrogation | Chaîne caractères |
| S(F) | Score du joueur numéro F | 0 à 15 |

■ MODIFICATIONS POSSIBLES

Augmentation du nombre de cases proposées au joueur

Le nombre de points d'interrogation pourrait être porté à un niveau supérieur et les cases supplémentaires pourraient contenir d'autres poissons ou... d'autres diables. Et pourquoi ne pas en prévoir une qui donnerait plusieurs points au joueur qui la désignerait, une sorte de case bonus en quelque sorte.

50 '
60 '
70 '

LA MARE AUX DIABLES

```

80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24: BORDER
    0:PAPER 0
110 SYMBOL 244,32,32,127,127,231,231,255,255:SYMBOL
    245,4,4,254,254,231,231,255,255
120 SYMBOL 246,127,63,56,24,31,15,7,3:SYMBOL 247,254,
    252,28,24,248,240,224,192
130 SYMBOL 248,0,0,0,0,15,63,255:SYMBOL 249,0,0,0,0,
    0,1,194,254
140 SYMBOL 250,255,63,15,0,0,0,0:SYMBOL 251,254,194
    ,1,0,0,0,0,0
150 SYMBOL 252,3,7,14,24,24,0,0,0:SYMBOL 253,128,192,
    96,48,24,24,48,96
160 SYMBOL 254,0,1,3,3,3,0,0,3:SYMBOL 255,192,128,0,0,
    0,0,0,0
170 E$=CHR$(248)+CHR$(249)+CHR$(8)+CHR$(8)+CHR$(10)+
    CHR$(250)+CHR$(251)
180 F$=CHR$(244)+CHR$(245)+CHR$(8)+CHR$(8)+CHR$(10)+
    CHR$(246)+CHR$(247)
190 G$=CHR$(252)+CHR$(253)+CHR$(8)+CHR$(8)+CHR$(10)
    +CHR$(254)+CHR$(255)
200 I$=CHR$(150)+CHR$(156)+CHR$(8)+CHR$(8)+CHR$(10)+
    CHR$(147)+CHR$(153)
210 H$=" " +CHR$(8)+CHR$(8)+CHR$(10)+" ":PEN 2
220 LOCATE 2,3:PRINT "LA MARE AUX DIABLES":PEN 3:LOCATE 1,10
230 LINE INPUT "NOM DU PREMIER JOUEUR ";N$(0):N$(0)=LEFT$(N$(0),11)
240 LOCATE 1,12:LINE INPUT "NOM DU DEUXIEME JOUEUR ";N$(1)
250 N$(1)=LEFT$(N$(1),11):E=50
260 A$=INKEY$:IF A$<>" " THEN 290 ELSE E=E+1:X=RND:IF E<50 THEN 260
270 E=0:PEN INT(RND*3)+1:LOCATE 1,20
280 PRINT "CHOISISSEZ LA DIFFICULTE ( DE 1 A 3 )" :GOTO 260
290 NV=VAL(A$):IF NV<1 OR NV>3 THEN 260 ELSE DIM A(11,7)
970 '
980 ' Decor
990 '
1000 CLS:WINDOW 2,35,2,23:PAPER 1:CLS:WINDOW 1,40,1,2

```



```

5
1010 FOR I=23 TO 551 STEP 48:MOVE I,32:DRAW I,384,
0
1020 MOVE I+2,32:DRAW I+2,384,0:NEXT:FOR I=39 TO 37
5 STEP 48
1030 MOVE 16,I:DRAW 560,I,0:MOVE 16,I+2:DRAW 560,I+2
:NEXT:PAPER 0
1040 PEN 2:LOCATE 2,25:PRINT N$(0);:PEN 3:LOCATE 21,
25:PRINT N$(1);
1050 FOR J=1 TO 7:FOR I=1 TO 11:A(I,J)=0:NEXT:NEXT
1060 PAPER 1:PEN 2:FOR I=1 TO 5:X(I)=INT(RND*11)+1
1070 Y(I)=INT(RND*7)+1:IF A(X(I),Y(I)) THEN 1070
1080 LOCATE 3*X(I),3*Y(I):PRINT E$:A(X(I),Y(I))=1:NEXT
1090 PEN 0:FOR I=1 TO 2*Nv:Z(I)=INT(RND*11)+1
1100 T(I)=INT(RND*7)+1:IF A(Z(I),T(I)) THEN 1100
1110 LOCATE 3*Z(I),3*T(I):PRINT F$:A(Z(I),T(I))=2:NEXT
1120 PAPER 0:PEN 2:LOCATE 3,1:PRINT N$(F);" REGARDE
Z BIEN ...."
1130 FOR I=1 TO 7000:NEXT:LOCATE 3,1:PRINT SPACE$(
30)
1140 PAPER 1:FOR I=1 TO 5:LOCATE 3*X(I),3*Y(I):PRINT
H$:NEXT
1150 FOR I=1 TO 2*Nv:LOCATE 3*Z(I),3*T(I):PRINT H$:N
EXT:PAPER 0
1160 PEN 1:C=5:LOCATE 32,1:PRINT C:XP=304:YP=192:U=
6:V=4
1970 '
1980 '      Preparation des cases
1990 '
2000 ON NV GOTO 2010,2040,2060
2010 M=INT(RND*11)+1:N=INT(RND*7)+1:IF A(M,N) THEN
2010
2020 A(M,N)=3:P=INT(RND*11)+1:Q=INT(RND*7)+1
2030 IF A(P,Q) THEN 2020 ELSE A(P,Q)=3:GOTO 2080
2040 M=INT(RND*11)+1:N=INT(RND*7)+1:IF A(M,N) THEN
2040
2050 A(M,N)=3:I=INT(RND*4)+1:P=Z(I):Q=T(I):A(P,Q)=5:GOTO
2080
2060 I=INT(RND*6)+1:M=Z(I):N=T(I):A(M,N)=5
2070 J=INT(RND*6)+1:IF J=I THEN 2070 ELSE P=Z(J):Q=
T(J):A(P,Q)=5
2080 I=INT(RND*5)+1:IF A(X(I),Y(I))<>1 THEN 2080
2090 X=X(I):Y=Y(I):A(X,Y)=4

```

```

2100 I=INT(RND*2*NV)+1:Z=Z(I):T=T(I)
2110 IF A(Z,T)>2 THEN 2100 ELSE A(Z,T)=5:E=80:G=2
2970 '
2980 '      Le joueur joue
2990 '
3000 A$=INKEY$:IF A$<>" " THEN 3050 ELSE E=E+1:IF
E<80 THEN 3000
3010 PAPER 1:PEN G:LOCATE 3*M,3*N:PRINT G$:LOCATE
3*P,3*Q
3020 PRINT G$:LOCATE 3*X,3*Y:PRINT G$:LOCATE 3*Z,3
*T:PRINT G$
3030 PAPER 0:PEN G-2:LOCATE 4,1:PRINT "CHOISISSEZ
UNE CASE"
3040 LOCATE 2+19*F,25:PRINT N$(F);G=5-G:E=0:GOSUB
4060:GOTO 3000
3050 IF A$<>CHR$(32) THEN GOSUB 4000:GOTO 3000
3060 IF A(U,V)<3 THEN 3000 ELSE PAPER 1
3070 IF A(U,V)=3 THEN PEN 0:LOCATE 3*U,3*V:PRINT I
$:GOTO 3130
3080 IF A(U,V)=5 THEN 3110 ELSE FOR I=0 TO 14:SOU
ND 1,300-10*I,5
3090 IF SQ(1)>127 THEN 3090 ELSE PEN I:LOCATE 3*U,
3*V
3100 PRINT E$:NEXT:S(F)=S(F)+1:GOTO 3130
3110 J=0:FOR I=1 TO 20:PEN J:LOCATE 3*U,3*V:SOUND
1,350+5*I,6
3120 IF SQ(1)>127 THEN 3120 ELSE PRINT F$:J=1-J:NE
XT:C=1
3130 PAPER 0:PEN 2+F:LOCATE 2+19*F,25:PRINT N$(F);
3140 LOCATE 13+19*F,25:PRINT S(F);C=C-1:PEN 1:LOCA
TE 32,1
3150 PRINT C:A(M,N)=A(M,N)-3:A(P,Q)=A(P,Q)-3:A(X,Y)=0
3160 A(Z,T)=2:FOR I=1 TO 1000:NEXT:PAPER 1:LOCATE 3
*M,3*N
3170 PRINT H$:LOCATE 3*P,3*Q:PRINT H$:LOCATE 3*X,3
*Y:PRINT H$
3180 LOCATE 3*Z,3*T:PRINT H$:IF C THEN 2000 ELSE
5000
3970 '
3980 '      Déplacement point lumineux
3990 '
4000 AS=ASC(A$):IF AS<240 OR AS>243 THEN RETURN E
LSE G=2:GOSUB 4060

```

```

4010 IF AS=242 AND U>1 THEN XP=XP-48:U=U-1
4020 IF AS=243 AND U<11 THEN XP=XP+48:U=U+1
4030 IF AS=240 AND V>1 THEN YP=YP+48:V=V-1
4040 IF AS=241 AND V<7 THEN YP=YP-48:V=V+1
4050 G=3:GOSUB 4060:RETURN
4060 MOVE XP,YP:DRAW 3,3,G-1:MOVER -3,0:DRAW 3,-3
,G-1:RETURN
4970 '
4980 '      Joueur suivant
4990 '
5000 FOR I=1 TO 900:NEXT:F=1-F:IF F=0 THEN IF S(0)
>14 OR S(1)>14 THEN 5040
5010 PAPER 0:LOCATE 1,1:PRINT SPACE$(35):FOR I=1 TO
12
5020 LOCATE 1,1:PEN I:PRINT "A VOUS , ";N$(F):SOUND
1,50-3*I,1
5030 FOR J=1 TO 60:NEXT:NEXT:G=2:GOSUB 4060:GOTO
1050
5040 WINDOW 1,23,6,9:PAPER 1:CLS:PEN 0:LOCATE 2,2
5050 IF S(0)>=S(1) THEN PRINT N$(0);" A GAGNE"
5060 IF S(1)>=S(0) THEN PRINT N$(1);" A GAGNE"
5070 IF INKEY$<>" " THEN 5070 ELSE WINDOW 1,40,1,25
:PAPER 0
5080 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>" "
THEN CLS:END
5090 FOR I=1 TO 100:NEXT:PEN INT(RND*3)+1:LOCATE 6
,14
5100 PRINT "UNE AUTRE PARTIE : TOUCHE R":GOTO 50
80

```

LABYRINTHE

Après avoir choisi le niveau de difficulté 1, vous verrez l'ordinateur dessiner à gauche de l'écran une grille de 22 carrés sur 23. C'est le labyrinthe que vous allez devoir traverser. Pour l'instant, aucune porte n'apparaît mais un petit carré vert clignote en bas de la grille : il indiquera votre position tout au long du jeu. Si vous portez votre attention sur la ligne supérieure de la grille, vous y verrez un point lumineux ; il se trouve dans la case qu'il va falloir rejoindre, une fois le labyrinthe traversé. Voyons maintenant comment se font les déplacements à l'intérieur de la grille.

Un message coloré vous invite à appuyer sur l'une des flèches du clavier. Choisissez par exemple celle qui pointe vers le haut et vous réaliserez votre premier pas dans le labyrinthe. Votre petit carré vert clignotant montera d'un cran et le passage entre sa position initiale et la case qu'il viendra d'atteindre sera ouvert.

Si vous pressez à ce moment-là la flèche qui va sur la gauche, votre carré se retrouvera dans la case située immédiatement à gauche de celle qu'il occupait et la porte qu'il aura empruntée sera ouverte. Voilà, ce n'est pas plus compliqué que cela de se déplacer dans la grille et il vous suffira de réitérer ce type d'action aussi souvent que nécessaire pour atteindre le point d'arrivée.

Naturellement, tout ceci se complique par le fait que tout labyrinthe qui se respecte contient des portes donnant sur des couloirs qui ne mènent nulle part. Si ce genre de mésaventure vous arrivait, sachez qu'il est toujours possible de revenir en arrière et de repasser dans les cases que l'on a déjà visitées.

Un dernier détail pour finir : ne vous étonnez pas que l'ordinateur ne vous laisse pas toujours le choix entre les quatre flèches. Une, deux ou même trois d'entre elles peuvent très bien ne pas être proposées. Cela correspond tout simplement au fait que dans un labyrinthe, on est souvent obligé de suivre des chemins pré-établis.

■ FIN DU JEU

L'ordinateur proclame la fin d'une partie et propose un nouveau labyrinthe quand on a atteint la sortie.

Au niveau de difficulté 1, la case d'arrivée est représentée par un point lumineux. Au niveau 2, ce point n'apparaît pas mais on sait quand même que l'arrivée se trouve quelque part sur la ligne supé-

rieure de la grille. Le niveau 3, pour sa part, se caractérise par le fait que la grille elle-même n'est pas dessinée. Difficile alors de savoir si l'on est déjà passé par ici ou par là...

■ PHASES ESSENTIELLES

Sous-programme 1000

Construction d'un labyrinthe.

- 1000 Le contenu de toutes les cases de la grille est annulé et les coordonnées du point de départ sont communiquées au programme.
- 1010 Un branchement adéquat est réalisé en fonction du résultat d'un tirage au sort.
- 1020 à 1030 La case située en haut de la précédente est sélectionnée. L'ordinateur vérifie que ce n'est pas déjà une case du labyrinthe.
- 1040 à 1050 Idem pour une case située à gauche.
- 1060 à 1070 Idem pour une case située à droite.
- 1080 à 1090 Idem pour une case située en dessous.
- 1100 Lorsque le programme se rebranche 25 fois de suite à cette ligne, la probabilité d'avoir affaire à un labyrinthe qui a tourné sur lui-même est très grande. On reprend alors tout à zéro.
- 1110 à 1140 Des portes sont ouvertes de façon aléatoire dans toutes les cases du labyrinthe, ce qui crée des chemins qui ne mènent nulle part.

| Variables | Fonction | Valeurs possibles |
|-----------|--------------------------------------|-------------------|
| T(I,J) | Contenu de la case (I,J) | 0 puis 10 à 13 |
| X et Y | Coordonnées d'une case du labyrinthe | 1 à 22 et 1 à 23 |
| A | Variable aléatoire | 1 à 5 |

| Variables | Fonction | Valeurs possibles |
|-----------|---|-------------------|
| B | Compteur pour les labyrinthes impossibles | 0 à 25 |
| F | Couleur du carré clignotant | 0 ou 2 |
| G | Abscisse de la case d'arrivée | 1 à 22 |

Sous-programme 2000

Affichage de la grille et des flèches.

2000 à 2020 Le labyrinthe, sans aucune porte ouverte, est dessiné.

2030 à 2080 Lorsqu'il est possible de se déplacer dans l'une des quatre directions, la flèche correspondante est affichée.

| Variables | Fonction | Valeurs possibles |
|-----------|--|---|
| N | Niveau de difficulté | 1 à 3 |
| X et Y | Coordonnées de la case dans laquelle est le joueur | 1 à 22 et 1 à 23 |
| H | Tableau indiquant si les flèches doivent apparaître ou non | 0 : pas de flèche 1 : flèche haute 2 : flèche basse 3 : flèche gauche 4 : flèche droite |

Sous-programme 3000

Déplacement dans le labyrinthe.

3000 à 3040 Un message dont les couleurs changent, indique que l'on doit faire porter son choix sur une flèche. En même temps, le carré donnant la position du joueur clignote dans la grille. Au niveau 1, la case d'arrivée clignote elle aussi.

3050 Une flèche vient d'être enfoncée et l'ordinateur vérifie si le coup est valide.

3060 à 3140 Le petit pavé clignotant du joueur est déplacé dans le labyrinthe et la porte par laquelle il est passé est laissée ouverte. Dans le cas où l'on aurait rejoint le point d'arrivée, le jeu s'arrête.

| Variables | Fonction | Valeurs possibles |
|-----------|--|-------------------|
| A\$ | Touche enfoncée | Touche quelconque |
| E | Compteur pour l'affichage d'un message | 0 à 60 |
| U | Numéro de la flèche pressée | 0 à 3 |

■ MODIFICATIONS POSSIBLES

Augmentation de la taille du labyrinthe

L'ordinateur est tout à fait capable de générer un labyrinthe contenant un nombre plus élevé de cases. Simplement, cela lui prendrait un peu plus de temps.

Modification de l'emplacement de la case d'arrivée

Celle-ci serait placée n'importe où dans la grille par le programme et le jeu consisterait à explorer le labyrinthe avec méthode pour la retrouver.

```

50 '
60 '
70 '          LABYRINTHE
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:DIM T(2
3,24)
110 PAPER 1:BORDER 0:WINDOW 27,40,2,4:CLS:PEN 2:LOC
ATE 2,2
120 PAPER 0:PRINT " LABYRINTHE ":WINDOW 1,40,1,25:E
=60
130 A$=INKEY$:IF A$<>" " THEN 160 ELSE E=E+1:X=RND
140 IF E<60 THEN 130 ELSE E=0:PEN INT(RND*3)+1:LOC
ATE 2,15

```

```

150 PRINT "CHOISISSEZ LA DIFFICULTE (DE 1 A 3)":GO
TO 130
160 N=VAL(A$):IF N<1 OR N>3 THEN 130 ELSE LOCATE
2,15
170 PEN 1:PRINT "UNE MINUTE , JE FABRIQUE UN LAB
YRINTHE"
1970 '
1980 '      Construction du labyrinthe
1990 '
1000 FOR I=1 TO 23:FOR J=1 TO 22:T(J,I)=0:NEXT J,I:X
=11:Y=23:T(11,23)=10
1010 A=INT(RND*5)+1:ON A GOTO 1020,1020,1040,1060,1
080
1020 IF T(X,Y-1)>9 THEN 1100 ELSE T(X,Y)=T(X,Y)+1
1030 Y=Y-1:T(X,Y)=10:B=0:IF Y=1 THEN 1110 ELSE 1010
1040 IF X=1 THEN 1010 ELSE IF T(X-1,Y)>9 THEN 110
0
1050 T(X,Y)=T(X,Y)+2:X=X-1:T(X,Y)=10:B=0:GOTO 1010
1060 IF X=22 THEN 1010 ELSE IF T(X+1,Y)>9 THEN 11
00
1070 X=X+1:T(X,Y)=12:B=0:GOTO 1010
1080 IF Y=23 THEN 1010 ELSE IF T(X,Y+1)>9 THEN 11
00
1090 Y=Y+1:T(X,Y)=11:B=0:GOTO 1010
1100 B=B+1:IF B<25 THEN 1010 ELSE SOUND 1,200,1:G
OTO 1000
1110 FOR I=2 TO 22:FOR J=2 TO 22:IF T(J,I)<9 THEN
1130
1120 IF RND<0.3 THEN T(J,I)=13:GOTO 1140 ELSE 1140
1130 IF RND<0.4 THEN T(J,I)=11 ELSE T(J,I)=12
1140 NEXT:G=X:LOCATE 2,15:PRINT SPACE$(38)
1970 '
1980 '      Decor
1990 '
2000 X=11:Y=23:IF N=3 THEN 2030 ELSE FOR I=32 TO
384 STEP 16
2010 MOVE I,16:DRAWR 0,368,1:NEXT:FOR I=16 TO 384
STEP 16
2020 MOVE 32,I:DRAWR 352,0,1:NEXT
2030 PAPER 1:WINDOW 32,34,13,23:CLS:WINDOW 28,38,17,
19
2040 CLS:WINDOW 1,40,1,25:PEN 3
2050 IF T(X,Y)=11 OR T(X,Y)=13 THEN LOCATE 33,14:PRI
NT CHR$(240):H(0)=1

```

```

2060 IF T(X,Y)=12 OR T(X,Y)=13 THEN LOCATE 29,18:PRINT
CHR$(242):H(2)=3
2070 IF T(X+1,Y)=11 THEN LOCATE 37,18:PRINT CHR$(24
3):H(3)=4
2080 IF T(X,Y+1)=11 OR T(X,Y+1)=13 THEN LOCATE 33,2
2:PRINT CHR$(241):H(1)=2
2970 '
2980 '      Le joueur choisit une fleche
2990 '
3000 A$=INKEY$:IF A$<>"" THEN 3050 ELSE E=E+1:IF
E<60 THEN 3000
3010 PAPER 0:PEN INT(RND*3)+1:LOCATE 27,7:PRINT "A
PPUYEZ SUR"
3020 LOCATE 27,9:PRINT "UNE FLECHE":E=0:F=2-F:GOSU
B 3040
3030 IF N=1 THEN PLOT 16*G+24,376,F:GOTO 3000 ELS
E 3000
3040 MOVE 16*X+22,390-16*Y:DRAWR 3,3,F:MOVER -3,0:D
RAWR 3,-3,F:RETURN
3050 U=ASC(A$)-240:IF U<0 OR U>3 THEN 3000 ELSE F
=0:GOSUB 3040
3060 IF H(U)<>1 THEN 3080 ELSE Y=Y-1
3070 MOVE 16*X+18,384-16*Y:DRAWR 13,0,0:GOTO 3140
3080 IF H(U)<>2 THEN 3100 ELSE Y=Y+1
3090 MOVE 16*X+18,400-16*Y:DRAWR 13,0,0:GOTO 3140
3100 IF H(U)<>3 THEN 3120 ELSE X=X-1
3110 MOVE 16*X+32,386-16*Y:DRAWR 0,13,0:GOTO 3140
3120 IF H(U)<>4 THEN 3140 ELSE X=X+1
3130 MOVE 16*X+16,386-16*Y:DRAWR 0,13,0
3140 F=2:GOSUB 3040:FOR U=0 TO 3:H(U)=0:NEXT:IF X<
G OR Y<>1 THEN 2030
3970 '
3980 '      Fin du jeu
3990 '
4000 SOUND 1,200,5:WINDOW 26,40,7,9:PAPER 1:CLS
4010 LOCATE 1,2:PEN 0:PRINT "VOUS AVEZ GAGNE":WIND
OW 1,40,1,24
4020 IF INKEY$<>"" THEN 4020 ELSE PAPER 0
4030 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>""
THEN CLS:END
4040 PEN INT(RND*3)+1:LOCATE 2,1:FOR I=1 TO 100:NE
XT
4050 PRINT "UNE AUTRE PARTIE : TOUCHE R":GOTO 40
30

```

NIM

Quatre lignes d'étoiles sont dessinées devant vos yeux. La première contient sept étoiles, la seconde cinq, la troisième trois et la dernière une seule. Vous êtes libre de choisir ¹ n'importe laquelle de ces seize étoiles. Cela aura pour effet non seulement de la faire disparaître, mais aussi d'effacer toutes les étoiles qui se trouvent sur sa droite.

Ce sera alors au tour de l'ordinateur de jouer. Il jettera son dévolu sur l'une des étoiles qui restent, la fera clignoter quelques instants et la supprimera. Cela sera accompagné là encore de la disparition de toutes les étoiles placées à droite de celle qui a été choisie.

Un message clignotant vous indiquera à ce moment-là que c'est de nouveau à vous de désigner une étoile. Quand vous l'aurez fait, l'ordinateur prendra son tour et supprimera un certain nombre d'étoiles d'une même ligne.

Il faudra bien que tôt ou tard, la fin de la partie soit proclamée. Le perdant sera, entre vous et l'ordinateur, celui qui ne pourra pas faire autrement que de prendre la dernière étoile du tableau.

■ FIN DU JEU

L'ordinateur s'avoue vaincu lorsqu'il se rend compte qu'il va être obligé de prendre la dernière étoile. En contrepartie, il ne manque pas de se déclarer vainqueur quand la dernière étoile en question revient au joueur. Un point est accordé au gagnant de chacune des parties et les scores restent affichés sur l'écran. Notons qu'afin d'équilibrer les chances, c'est tantôt au joueur de commencer une partie, tantôt à la machine.

L'ordinateur est un redoutable adversaire au niveau de difficulté 3 ; il ne laisse aucune chance au joueur qui viendrait à commettre une erreur. Par contre, au niveau 2 et surtout au niveau 1, il donne souvent l'impression de bien peu réfléchir avant de se décider à supprimer des étoiles.

■ PHASES ESSENTIELLES

Sous-programme 2000

Le joueur désigne une étoile.

1. Il suffit de déplacer, avec les quatre flèches, le petit repère jaune clignotant jusqu'à ce qu'il se trouve sous l'étoile voulue. A ce moment-là, on appuie sur la barre d'espacement.

- 2000 à 2020 L'ordinateur attend que le joueur porte son choix sur une étoile et le rappelle à l'ordre de temps en temps par un message clignotant.
- 2030 à 2040 La barre d'espacement vient d'être enfoncée. Il faut vérifier que c'est bien une étoile qui a été désignée et non pas une case vide.
- 2050 L'étoile retenue ainsi que ses suivantes sont effacées.
- 2060 Si l'on a pris la dernière étoile, le jeu est terminé.
- 2070 à 2080 Si l'on a fait en sorte de ne laisser à l'ordinateur qu'une étoile (ou 3 étoiles situées sur des lignes différentes), il en est réduit à se déclarer perdant.

| Variables | Fonction | Valeurs possibles |
|-----------|---|-------------------|
| A\$ | Touche enfoncée | Touche quelconque |
| E | Compteur pour l'affichage d'un message | 0 à 60 |
| C | Numéro d'une couleur | 2 ou 3 |
| XP et YP | Coordonnées du repère clignotant | 5 à 29 et 6 à 18 |
| X et Y | Coordonnées d'une étoile | 1 à 7 et 1 à 4 |
| M(Y) | Nombre d'étoiles de la ligne Y | 1 à 7 |
| YO | Ordonnée de la case contenant le repère | 1 à 4 |
| N(Y) | Largeur de la ligne Y | 6 à 30 |

Sous-programme 3000

C'est au tour de l'ordinateur de jouer.

- 3000 à 3010 L'ordinateur fait semblant de réfléchir et, en fonction du niveau de difficulté, se lance ou pas dans les calculs qui lui donneront une bonne chance de gagner.
- 3020 à 3170 Le nombre d'étoiles contenues dans chaque rangée est traduit en binaire et l'ordinateur en déduit la ligne

dans laquelle il va choisir une étoile. Mais il se peut que le joueur ait très bien joué et que la méthode de l'ordinateur ne donne rien ; une étoile absolument quelconque est alors tirée au sort.

3180 à 3220 Le programme fait clignoter les étoiles qui vont disparaître et arrête la partie si l'ordinateur a pris la dernière de ces étoiles.

| Variables | Fonction | Valeurs possibles |
|-----------|---|-------------------|
| P(I,J) | Chiffre binaire | 0 ou 1 |
| L | Quotients successifs par 2 du nombre d'étoiles de chaque rangée | 0 à 3.5 |
| R ou U | Abscisse de l'étoile qui va être effacée | 1 à 7 |
| Y | Ordonnée de l'étoile qui va être effacée | 1 à 4 |

■ MODIFICATIONS POSSIBLES

Nouvelle répartition des étoiles

Rien ne nous oblige à avoir toujours le même tableau d'étoiles en début de partie.

Modification de la règle du jeu

Le gagnant serait alors justement celui qui prendrait la dernière étoile.

```

50 '
60 '
70 '          N I M
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    0:PAPER 1
110 WINDOW 34,40,1,3:CLS:WINDOW 1,40,1,25:PAPER 0:PEN
    2
120 LOCATE 35,2:PRINT " NIM ":PEN 3:LOCATE 1,10

```



```

130 PRINT "VOUS POUVEZ ENLEVER AUTANT D'ETOILES
"
140 PRINT "QUE VOUS VOULEZ DANS UNE MEME RANGEE
."
150 PRINT:PRINT"EN CHOISSANT UNE ETOILE , ON LA
FAIT"
160 PRINT "DISPARAITRE AINSI QUE CELLES QUI SONT
"
170 PRINT "SUR SA DROITE.":PEN 1:LOCATE 1,24
180 PRINT "LE PERDANT EST CELUI QUI PREND ...."
190 PRINT ".... LA DERNIERE ETOILE .":E=60
200 A$=INKEY$:IF A$<>" " THEN 230 ELSE E=E+1:X=RND
210 IF E<60 THEN 200 ELSE PEN INT(RND*3)+1:LOCATE
1,19
220 PRINT "CHOISISSEZ LA DIFFICULTE (DE 1 A 3)"
230 NV=VAL(A$):IF NV<1 OR NV>3 THEN 200 ELSE WIND
OW 1,40,4,23
240 SYMBOL 253,0,0,0,32,112,168,32,32:SYMBOL 254,1,0,
0,3,0,0,1,0
250 SYMBOL 255,36,168,112,254,112,168,36,32:FOR I=1
TO 7
260 E$=E$+CHR$(254)+CHR$(255)+" ":NEXT:CLS:PAPER 1
:PEN 2
970 '
980 ' Decor
990 '
1000 WINDOW 25,40,20,22:CLS:LOCATE 2,2:PRINT "JOUEU
R : 0"
1010 WINDOW 25,40,16,18:CLS:PEN 3:LOCATE 2,2:PRINT "
ORDINATEUR: 0"
1020 PAPER 1:PEN 2:J=4:FOR I=7 TO 1 STEP -2:WINDOW
2,4*I+3,J,J+2
1030 CLS:LOCATE 3,2:PRINT LEFT$(E$,4*I-2):J=J+4:NEXT
1040 WINDOW 1,40,1,25:XP=29:YP=6:X=7:Y=1:Y0=1
1050 M(1)=7:M(2)=5:M(3)=3:M(4)=1:N(1)=30:N(2)=22:N(3)=14
1060 N(4)=6:E=60:C=2:D=D+1:IF D/2=INT(D/2) THEN 3000
1970 '
1980 ' Le joueur joue
1990 '
2000 A$=INKEY$:IF A$<>" " THEN 2030 ELSE E=E+1:IF
E<60 THEN 2000
2010 C=5-C:PAPER 0:PEN C:LOCATE 5,2:PRINT "CHOISIS
SEZ UNE ETOILE"

```

```

2020 PAPER 1:PEN 3*(C-2):LOCATE XP,YP:PRINT CHR$(25
3):E=0:GOTO 2000
2030 IF A$<>CHR$(32) THEN GOSUB 4000:GOTO 2000
2040 IF X>M(Y) THEN 2000 ELSE Y0=Y:M(Y)=X-1:SOUND
1,200,1
2050 PAPER 1:LOCATE 4*X,4*Y+1:PRINT SPACE$(N(Y)-4*
X)
2060 FOR I=1 TO 4:IF M(I)=0 THEN NEXT:GOTO 5000
2070 K=0:FOR I=1 TO 4:IF M(I)>1 THEN 3000 ELSE IF
M(I) THEN K=K+1
2080 NEXT:IF K=1 OR K=3 THEN 5020
2970 '
2980 '      L'ordinateur joue
2990 '
3000 PAPER 1:PEN 2:LOCATE 5,2:PRINT "CHUT, JE REFL
ECHIS .."
3010 FOR I=1 TO 1000:NEXT:IF RND>0.1+0.3*NVT THEN
3080
3020 FOR I=1 TO 4:K=M(I):FOR J=0 TO 2:L=K/2
3030 IF L=INT(L) THEN P(I,J)=0 ELSE P(I,J)=1
3040 K=INT(L):NEXT:NEXT:FOR J=2 TO 0 STEP -1
3050 K=0:R=0:FOR I=1 TO 4:IF P(I,J)=0 THEN 3070
3060 K=K+1:IF M(I)>R THEN R=M(I):Y=I
3070 NEXT:IF K/2 <> INT(K/2) THEN 3110 ELSE NEXT
3080 Y=INT(RND*4)+1:IF M(Y)=0 THEN 3080
3090 U=M(Y):I=INT(RND*M(Y))+1
3100 M(Y)=M(Y)-I:GOTO 3180
3110 U=M(Y):M(Y)=0:P(Y,0)=0:P(Y,1)=0
3120 P(Y,2)=0:FOR J=2 TO 0 STEP -1:K=0
3130 FOR I=1 TO 4:IF P(I,J)=1 THEN K=K+1
3140 NEXT:IF K/2 <> INT(K/2) THEN M(Y)=M(Y)+2^J
3150 NEXT:FOR I=1 TO 4:IF M(I)>1 THEN 3180
3160 NEXT:K=0:FOR I=1 TO 4:IF M(I)=1 THEN K=K+1
3170 NEXT:IF K/2=INT(K/2) THEN M(Y)=1-M(Y)
3180 PEN 2:FOR I=1 TO 5:LOCATE 4*M(Y)+4,4*Y+1
3190 PRINT RIGHT$(E$,4*U-4*M(Y)):FOR J=1 TO 150:NEX
T
3200 LOCATE 4*M(Y)+4,4*Y+1:PRINT SPACE$(4*U-4*M(Y))
3210 FOR J=1 TO 100:NEXT:NEXT:FOR I=1 TO 4
3220 IF M(I) THEN Y=Y0:GOTO 2000 ELSE NEXT:GOTO 5
020
3970 '
3980 '      Deplacement repere

```

```

3990 '
4000 AS=ASC(A$):IF AS<240 OR AS>243 THEN RETURN
4010 PAPER 1:LOCATE XP,YP:PRINT " "
4020 IF AS=242 AND X>1 THEN XP=XP-4:X=X-1
4030 IF AS=243 AND X<7 AND TEST(16*XP+48,16*(25-Y
P)) THEN XP=XP+4:X=X+1
4040 IF AS=240 AND Y>1 THEN YP=YP-4:Y=Y-1
4050 IF AS=241 AND Y<4 AND TEST(16*XP,16*(23-YP))
THEN YP=YP+4:Y=Y+1
4060 PEN 3:LOCATE XP,YP:PRINT CHR$(253):RETURN
4970 '
4980 '      Fin du jeu
4990 '
5000 WINDOW 20,31,12,14:PAPER 1:CLS:PEN 0:LOCATE 2,
2
5010 PRINT "J'AI GAGNE":S=S+1:GOTO 5040
5020 WINDOW 20,31,12,14:PAPER 1:CLS:PEN 0:LOCATE 2,
2
5030 PRINT "J'AI PERDU":T=T+1
5040 WINDOW 1,40,1,25:FOR I=1 TO 2000:NEXT
5050 PAPER 1:PEN 3:LOCATE 37,17:PRINT S:PEN 2
5060 IF INKEY$<>"" THEN 5060 ELSE LOCATE 37,21:PR
INT T
5070 A$=INKEY$:IF A$<>"" THEN 5100 ELSE E=E+1:IF
E<60 THEN 5070
5080 E=0:PAPER 0:PEN INT(RND*3)+1:LOCATE 5,2
5090 PRINT "APPUYEZ N'IMPORTE OU ":GOTO 5070
5100 LOCATE 5,2:PRINT SPACE$(21):WINDOW 20,31,12,14
5110 PAPER 0:CLS:WINDOW 1,40,1,25:GOTO 1020

```

LE MOT

Une comparaison entre le listing de ce programme et celui du pendu fait apparaître que les seules parties différentes concernent le décor et la phase de jeu proprement dite. Alors, si vous avez déjà tapé l'un des deux programmes, contentez-vous simplement de le modifier pour qu'il s'adapte à l'autre jeu.

Votre travail va consister, exactement comme dans le jeu habituel du pendu, à retrouver un mot que l'ordinateur a choisi. Afin de rendre les explications plus compréhensibles, nous supposerons qu'il s'agit du mot RENARD. Pour l'instant, on ne connaît aucune lettre et, en haut de l'écran, apparaissent six tirets ; chacun d'eux correspond à une lettre du mot RENARD.

Un message clignotant nous invite à taper au clavier, non pas une, mais six lettres. Allons-y, essayons par exemple A B C D E F. Ceci n'a pas l'air de plaire beaucoup à l'ordinateur car il se met à dessiner derechef un pendu. Pour l'instant ce n'est pas bien grave et seule la base de la potence est visible mais, si l'on se trompe encore, la machine dessinera la potence elle-même, la corde, la tête du pendu, son corps, etc. Si par malheur nous sommes arrivés à nous tromper dix fois, le pendu complet aura été reconstitué et nous aurons perdu.

Mais nous n'en sommes pas encore là, aussi regardons les indications que l'ordinateur a porté dans différentes cases de l'écran et essayons d'en tirer profit. Il est d'abord indiqué que parmi les six lettres que l'on a proposées, certaines font partie du mot à retrouver, ce sont le A, le D et le E, et que les autres, le B, le C et le F, n'ont pas de rapport avec le mot en question. Le fait de connaître d'un seul coup trois bonnes lettres est intéressant mais cela ne nous dit pas à quels endroits sont placées ces lettres dans le mot que l'on cherche. On le saura certainement un peu plus tard ; pour l'instant tapons, comme le programme nous le demande, six autres lettres au clavier, pourquoi pas G E H I J K ? L'ordinateur ajoute au rayon des lettres non valables la suite G H I J K mais n'additionne pas un élément nouveau au dessin du pendu. Vous avez deviné que c'est la lettre E qui nous a évité des ennuis : elle fait partie en effet du mot à retrouver, ce que nous savions, mais on l'a cette fois proposée en deuxième position, à sa place réelle dans le mot RENARD donc. Nous en avons d'ailleurs la confirmation par le programme qui a remplacé le deuxième des six tirets en haut de l'image par la lettre E.

Vous savez maintenant ce qu'il vous reste à faire pour gagner. Vous devez retrouver toutes les lettres d'un mot, et ensuite les proposer à l'ordinateur dans le bon ordre. Un crayon et un bout de papier vous seront peut-être nécessaires...

Si on le souhaite, il est possible de jouer à plusieurs avec ce programme. L'un des joueurs choisit un mot à la place de l'ordinateur. Dès qu'il l'a entré au clavier, l'écran s'efface et les autres participants doivent alors retrouver ce mot. Là encore, le pendu se dessine au fur et à mesure que l'on se trompe.

Note : Positionner la touche CAPS LOCK pour que les lettres proposées à l'ordinateur soient en majuscules.

■ FIN DU JEU

Il n'y a naturellement que deux possibilités, on trouve ou on ne trouve pas le mot.

Le niveau de difficulté est lié à la complexité des mots que l'ordinateur choisit. Au niveau 1, par exemple, il ne propose que des mots d'un usage très courant, et, au niveau 4, il nous demande de reconstituer des mots tout à fait invraisemblables.

■ PHASES ESSENTIELLES

Sous-programme 2000

Analyse du mot entré par le joueur.

2000 à 2040 Le programme attend que le joueur ait tapé un nombre de lettres égal à la longueur du mot à retrouver. Il ignore tous les caractères qui ne sont pas alphabétiques.

2050 Le programme va comparer chacune des lettres du mot à retrouver avec chacune des lettres du mot que l'on a proposé.

2060 à 2100 Lorsqu'une lettre a été tapée à la bonne place, elle est affichée, prenant de ce fait l'emplacement du tiret correspondant en haut de l'écran.

2110 à 2130 Si l'une des lettres que l'on a donnée à l'ordinateur est contenue dans le mot cherché, mais à une place autre que celle que l'on a indiquée, le programme nous le fait savoir.

- 2140 à 2160 Les lettres qui ne se trouvent pas dans le mot de l'ordinateur sont affichées dans le cadre voulu.
- 2170 à 2190 Lorsque, parmi toutes les lettres que l'on a choisies, il ne s'en trouve aucune à la bonne place, le pendu se dessine, élément par élément.

| Variables | Fonction | Valeurs possibles |
|-----------|---|------------------------|
| A\$ | Touche enfoncée | Touche quelconque |
| T\$(I) | Lettre que l'on a tapée en position I | Caractère alphabétique |
| E | Compteur pour l'affichage d'un message | 0 à 30 |
| B\$ | Chaîne formée de tirets | Chaîne caractères |
| M\$(J) | Lettre numéro J du mot de l'ordinateur | Caractère alphabétique |
| A(I) | Drapeau indiquant que la lettre numéro I a déjà été trouvée | 0 ou 1 |
| G | Drapeau indiquant qu'une lettre a été trouvée à sa bonne place | 0 ou 1 |
| U | Drapeau indiquant qu'une lettre se trouve dans le mot de l'ordinateur | 0 ou 1 |
| D | Drapeau indiquant qu'une lettre a été trouvée dans le mot de l'ordinateur | 0 ou 1 |
| H | Nombre de lettres justes | 0 à L |
| L | Longueur du mot à trouver | 0 à 20 |
| N\$ | Chaîne formée par des lettres valables mais en mauvaise position | Chaîne caractères |
| P\$ | Chaîne formée par des lettres non valables | Chaîne caractères |
| B | Compteur pour le dessin du pendu | 0 à 10 |

Sous-programme 4000

Dessin des différentes parties du pendu.

4000 Base de la potence.

4010 Élément vertical de la potence.

4020 Élément horizontal de la potence.
 4030 à 4040 Corde.
 4050 à 4060 Tête du pendu.
 4070 à 4090 Corps du pendu.
 4100 à 4130 Bras du pendu.
 4140 à 4170 Jambes du pendu.

■ MODIFICATIONS POSSIBLES

Augmentation du nombre de mots connus par l'ordinateur

80 de ceux-ci sont contenus dans le programme et il n'y a pas de difficulté majeure à en ajouter de nouveaux.

Possibilité pour l'ordinateur de fabriquer lui-même des mots

Il tirerait alors au sort une série quelconque de lettres de l'alphabet et nous aurions à les retrouver, sans doute avec quelque peine.

```

50 '
60 '
70 '          LE  MOT
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    0:PAPER 0
110 PEN 2:LOCATE 1,8:PRINT "TOUCHE 1: L'ORDINATEUR
    CHOISIRA UN MOT"
120 LOCATE 1,19:PRINT "TOUCHE 2: L'UN DE VOUS CHO
    ISIT UN MOT"
130 PRINT "QUE LES AUTRES DEVRONT RETROUVER":PE
    N 3
140 A$=INKEY$:IF A$="" THEN X=RND:GOTO 140
150 IF A$="2" THEN 200 ELSE IF A$<>"1" THEN 140
160 LOCATE 1,19:PRINT SPACE$(72):LOCATE 1,11
170 PRINT "CHOISISSEZ LA DIFFICULTE ( DE 1 A 4 )"
180 A$=INKEY$:IF A$="" THEN 180 ELSE NV=VAL(A$):IF
    NV<1 OR NV>4 THEN 180
  
```

```

190 X=INT(RND*20)+1:X=X+(NV-1)*20:FOR I=1 TO X:READ
M$:NEXT:GOTO 1000
200 LOCATE 1,22:LINE INPUT "MOT A RETROUVER ";M$
210 IF LEN(M$)>20 OR LEN(M$)<3 THEN RUN
970 '
980 '      Decor
990 '
1000 CLS:L=LEN(M$):DIM M$(20),T$(20),A(20):FOR I=1 TO
L:B$=B$+"-"
1010 PEN 2:LOCATE 2*I-1,1:PRINT "-":M$(I)=MID$(M$,I,1
)
1020 NEXT:PAPER 1:WINDOW 31,40,3,5:CLS:PAPER 0:LOCA
TE 2,2
1030 PRINT " LE MOT ":WINDOW 19,40,9,13:PAPER 1:CLS
:PEN 3
1040 LOCATE 3,2:PRINT "PROPOSEZ      LETTRES":WINDO
W 19,40,15,19
1050 CLS:PEN 0:LOCATE 4,2:PRINT "LETTRES DU MOT"
1060 WINDOW 19,40,21,25:CLS:LOCATE 2,2
1070 PRINT "LETTRES NON VALABLES":WINDOW 1,40,1,25
1970 '
1980 '      Le joueur propose un mot
1990 '
2000 PAPER 1:PEN 3:LOCATE 20,12:PRINT B$:FOR I=1 T
O L
2010 A$=INKEY$:IF A$<>"" THEN 2030 ELSE E=E+1:IF
E<30 THEN 2010
2020 PEN 3*INT(RND*2):LOCATE 29,10:PRINT L:E=0:GOTO
2010
2030 IF A$<"A" OR A$>"Z" THEN 2010 ELSE PEN 3:LOC
ATE 19+I,12
2040 PRINT A$:T$(I)=A$:NEXT:G=0:FOR J=1 TO L:U=0:D=
0
2050 FOR I=1 TO L:IF T$(J)<>M$(I) OR I<>J THEN 2110
2060 IF A(I) THEN 2140 ELSE U=1:G=1:PAPER 0:PEN 2:
SOUND 1,200,3
2070 LOCATE 2*I-1,1:PRINT M$(I):A(I)=1:H=H+1:IF H=L
THEN 3030
2080 V=INSTR(N$,M$(I)):IF V=0 THEN 2110
2090 N$=LEFT$(N$,V-1)+RIGHT$(N$,LEN(N$)-V)
2100 PAPER 1:PEN 0:LOCATE 20,18:PRINT N$;" "
2110 IF I=J OR T$(J)<>M$(I) OR A(I) THEN 2140
2120 U=1:IF INSTR(N$,T$(J)) THEN IF D=0 THEN 2140

```

```

2130 N$=N$+T$(J):PAPER 1:PEN 0:LOCATE 20,18:PRINT N
$:D=1
2140 NEXT:IF U OR INSTR(P$,T$(J)) THEN 2170
2150 P$=P$+T$(J):PAPER 1:PEN 0:LOCATE 20,24:PRINT P
$;
2160 IF LEN(P$)=20 THEN 3000
2170 NEXT:IF G=1 THEN 2190 ELSE B=B+1
2180 ON B GOSUB 4000,4010,4020,4030,4050,4070,4100,
4120,4140,4160
2190 IF INKEY$<>" " THEN 2190 ELSE WINDOW 1,40,1,25
:IF B<10 THEN 2000
2970 '
2980 '      Fin du jeu
2990 '
3000 WINDOW 15,21,3,5:PAPER 1:CLS:PEN 0:LOCATE 2,2:P
RINT "PERDU"
3010 WINDOW 1,40,1,25:PAPER 0:PEN 3:LOCATE 1,1:PRIN
T SPACE$(40)
3020 LOCATE 1,1:PRINT "REPONSE : ";M$:GOTO 3040
3030 WINDOW 15,21,3,5:PAPER 1:CLS:PEN 0:LOCATE 2,2:P
RINT "GAGNE"
3040 IF INKEY$<>" " THEN 3040 ELSE WINDOW 1,40,1,25
:FOR I=1 TO 3000:NEXT
3050 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>" "
THEN CLS:END
3060 PAPER 0:PEN INT(RND*3)+1:LOCATE 15,7
3070 PRINT "UNE AUTRE PARTIE: TOUCHE R":FOR I=1
TO 80:NEXT:GOTO 3050
3970 '
3980 '      Dessin du pendu
3990 '
4000 PAPER 1:WINDOW 2,15,24,25:CLS:RETURN
4010 PAPER 1:WINDOW 3,3,3,23:CLS:RETURN
4020 PAPER 1:WINDOW 3,11,3,3:CLS:RETURN
4030 MOVE 48,243:DRAW 143,352,3:DRAWR 0,-48,3:MOVER
2,0
4040 DRAWR 0,48,3:DRAW 48,240,3:RETURN
4050 PAPER 2:WINDOW 9,10,7,8:CLS:PLOT 136,292,0
4060 PLOT 152,292,0:MOVE 136,280:DRAWR 10,0,0:RETUR
N
4070 MOVE 143,272:DRAWR 0,-8,2:MOVER 2,0
4080 PAPER 3:WINDOW 9,10,10,15:CLS
4090 MOVE 128,192:DRAWR 32,0,1:RETURN

```

4100 MOVE 128,256:DRAW -32,-80,3:DRAW -8,-2,3:DRAW
 8,-1,3
 4110 DRAW -6,-10,3:DRAW 4,0,3:DRAW 128,248,3:RETU
 RN
 4120 MOVE 160,256:DRAW 32,-80,3:DRAW 8,-2,3:DRAW
 -8,-1,3
 4130 DRAW 6,-10,3:DRAW -4,0,3:DRAW 160,248,3:RETU
 RN
 4140 MOVE 128,160:DRAW 0,-96,2:DRAW -8,-16,2:DRAW
 4,0,2
 4150 DRAW 8,16,2:DRAW 0,94,2:RETURN
 4160 MOVE 159,160:DRAW 0,-96,2:DRAW 8,-16,2:DRAW
 -4,0,2
 4170 DRAW -8,16,2:DRAW 0,94,2:RETURN
 4970 '
 4980 ' Mots de l'ordinateur
 4990 '
 5000 DATA SUCRE,VOITURE,BALANCE,ECOLE,RENARD,ENFA
 NT,GLACE,ARDOISE
 5010 DATA LIVRE,CERCLE,POMME,BRIQUE,ORANGE,FOUDRE,
 ARMOIRE,MUSIQUE
 5020 DATA CHAISE,NAVIRE,CABINE,TIGRE,APPETIT,CAFETI
 ERE,FAUVETTE
 5030 DATA DECIMETRE,GOUVERNEUR,OMBRAGE,DEGRADATI
 ON,OSSATURE
 5040 DATA ARCHIDUC,FEUILLETON,BANDEROLLE,FAINEANT
 ,ESCALIER
 5050 DATA ARRIMAGE,DECAGONE,CHAMPAGNE,GAZOMETRE,A
 RISTOCRATE
 5060 DATA HARMONICA,ECHANTILLON,CARICATURE,ELECTR
 OSCOPE,COAGULATION
 5070 DATA APPRECIATION,DEMINERALISATION,BARYTON,H
 IPPOCAMPE
 5080 DATA BRANDEBOURG,DESSICCATION,HEMISPHERIQUE,
 CAHOTEMENT
 5090 DATA DEHARNACHEMENT,GRAPHOLOGUE,HOMONYME,CL
 AVECINISTE
 5100 DATA GALACTOMETRE,BELLIGERANT,ECCLESIASTIQU
 E,DYNAMITAGE
 5110 DATA ASSUJETTISSEMENT,ZIRCONIUM,CASTANEACEES
 ,FORMALDEHIDE
 5120 DATA BUTYROMETRE,SYNALLAGMATIQUE,TRANSGANG
 ETIQUE,EMPHYTEOTIQUE

5130 DATA ZINGLBERACEES,BALENOPTERE,XEROPHTALMIE,
TETRASYLLABIQUE
5140 DATA CEPHALOTHORAX,ZINCOGRAVURE,ECHINODERME
,ASPARAGINEES
5150 DATA FLUORESCINE,ELLIPSOGRAPHE,HYPOCONDIAQ
UE
5160 DATA CARBORUNDUM,HIPPOCASTANEES

LES FRUITS

| | | | | | |
|---|---|---|---|---|---|
| M | A | N | | | |
| | | D | A | | E |
| | | | R | I | N |

Regardez bien où se trouvent les différents noms de fruits et vous aurez ensuite moins de mal à les rechercher car, tôt ou tard, l'ordinateur réalisera sa triste besogne ; il va en effet compléter la totalité de la grille avec des lettres qu'il tire au sort et qui n'ont rien à voir, naturellement, avec des noms de fruits. Voici, toujours en parlant du mot *mandarine*, ce que l'on risque d'avoir devant les yeux :

| | | | | | |
|---|---|---|---|---|---|
| M | A | N | T | U | L |
| S | E | D | A | M | E |
| G | R | I | R | I | N |

Vous avez quand même repéré le mot *mandarine* ? Bravo, faites-le savoir à l'ordinateur en désignant ¹ le premier caractère de ce mot, c'est-à-dire le M. Vous verrez alors d'un seul coup toutes les lettres du mot *mandarine* changer de couleur puis s'effacer de la grille. Il

- 136 -

ne vous restera plus qu'à trouver les neufs autres noms de fruits pour gagner.

Notez qu'au niveau 1, si l'on pointe une lettre qui n'est pas la première d'un nom de fruit, elle ne fait rien d'autre que de se mettre à clignoter ; puis le jeu reprend son cours normal.

■ FIN DU JEU

La seule façon de gagner la partie est de retrouver la totalité des dix mots cachés par l'ordinateur.

Voici ce qui différencie les niveaux de difficulté :

- Au niveau 1, les mots que l'on doit retrouver apparaissent d'abord seuls sur l'écran avant d'être noyés dans l'ensemble des autres lettres. De plus, afin de guider le joueur, on voit, au milieu des autres lettres qui sont vertes, dix lettres colorées en noir. Ces lettres-là sont les dernières de chacun des noms de fruits. Inutile donc de les désigner, elles ne feraient que perdre leur couleur noire, mais servez-vous en pour reconstituer les mots en partant de leur fin.
- Au niveau 2, la grille est directement proposée au joueur sans que les emplacements des mots à retrouver aient été montrés auparavant. D'autre part, les dernières lettres des noms de fruits n'ont pas de couleur spéciale.
- Au niveau 3, la dernière lettre de chacun des mots à retrouver est visible en noir mais la durée d'une partie est limitée à deux minutes trente secondes. De plus, lorsque l'on choisit une mauvaise case, le temps est diminué d'office de dix secondes.
- Au niveau 4, la durée d'une partie est aussi limitée à deux minutes trente, mais toutes les lettres de la grille ont la même couleur. Le fait de se tromper de case enlève, cette fois encore, dix secondes au chronomètre.

■ PHASES ESSENTIELLES

Sous-programme 1000

Construction de la grille.

- 1000 à 1030 Les dix noms de fruits sont donnés à l'ordinateur.
- 1040 à 1050 Les coordonnées d'une case de la grille sont tirées au sort. On va placer dans cette case, si elle est libre, la première lettre d'un des mots à retrouver.
- 1060 à 1140 Toutes les lettres qui constituent un nom de fruits sont placées dans des cases contiguës déterminées de façon aléatoire. Mais il peut arriver qu'un mot ne puisse pas s'écrire dans un endroit de la grille ; on lui cherche alors un nouvel emplacement.
- 1150 à 1170 Des lettres quelconques sont placées dans les cases qui n'ont pas été occupées.

| Variables | Fonction | Valeurs possibles |
|------------------|--|--|
| L | Numéro d'un des dix noms de fruits | 1 à 10 |
| M\$(L) | Nom du fruit numéro L | Chaîne caractères |
| X et Y | Coordonnées d'une case | 1 à 15 et 1 à 12 |
| A\$(X,Y) | Caractère contenu dans la case (X,Y) | Lettre de l'alphabet |
| A(X,Y) | Valeur de la case (X,Y) | 0 : case ordinaire 1 à 10 : première lettre 11 : case déjà trouvée |
| C | Compteur pour les cas impossibles | 0 à 20 |
| M | Numéro d'une lettre dans un mot | 1 à 9 |
| B | Nombre aléatoire | 0 à 3 |
| X(L,M) et Y(L,M) | Coordonnées de la lettre numéro M dans le mot numéro L | 1 à 15 et 1 à 12 |

Sous-programme 3000

Le joueur désigne une lettre de la grille.

- 3000 à 3010 La dernière lettre de chacun des dix noms de fruits est colorée en noir (niveaux 1 et 3).
- 3020 à 3060 Le joueur est invité, par un message clignotant, à désigner une case de la grille. Le programme en pro-

fite pour retrancher une unité au chronomètre (niveaux 3 et 4). Le jeu est arrêté si l'on a complètement utilisé son capital de temps.

3070 à 3080 Les caractéristiques de la touche enfoncée sont analysées. L'ordinateur fait comme s'il n'avait rien vu si l'on choisit une case effacée.

3090 à 3100 La lettre que l'on a désignée n'est pas la première d'un nom de fruits. Elle se met à clignoter bruyamment et dix secondes sont soustraites au chronomètre.

3110 à 3140 On a trouvé une bonne lettre. Toutes celles du mot correspondant changent de couleur sur la grille.

3150 à 3160 Puis elles s'effacent.

3170 Le programme boucle jusqu'à ce que l'on ait trouvé tous les noms de fruits.

| Variables | Fonction | Valeurs possibles |
|-----------|--|----------------------|
| A\$ | Touche enfoncée | Touche quelconque |
| E | Compteur pour l'affichage d'un message | 0 à 60 |
| C | Numéro d'une couleur | 0 ou 3 |
| XP et YP | Coordonnées du pavé clignotant | 34 à 482 et 10 à 362 |
| T | Variable du chronomètre | 150 à 0 |
| NV | Niveau de difficulté | 1 à 4 |
| U et V | Coordonnées de la case choisie | 1 à 15 et 1 à 12 |
| S | Nombre de mots trouvés | 0 à 10 |

■ MODIFICATIONS POSSIBLES

Augmentation de la taille de la grille

On a la possibilité, avec notre ordinateur, de faire entrer jusqu'à ... 1000 caractères dans l'écran.

Non affichage de la liste des mots à retrouver

Le jeu se compliquerait beaucoup si l'on ne savait pas exactement quels sont les mots à rechercher. Naturellement, il faudrait écrire en DATA toute une série de mots mais seuls dix d'entre eux seraient tirés au sort pour être utilisés dans la grille.

```
50 '
60 '
70 '          LES    FRUITS
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    O:PAPER 0
110 A$=INKEY$:IF A$<>"" THEN 140 ELSE E=E+1:X=RND

120 IF E<60 THEN 110 ELSE PEN INT(RND*3)+1:LOCATE
    1,10
130 PRINT "CHOISISSEZ LA DIFFICULTE  ( DE 1 A 4 )"
    :E=0:GOTO 110
140 NV=VAL(A$):IF NV<1 OR NV>4 THEN 110 ELSE PEN
    2
150 LOCATE 1,10:PRINT "UNE MINUTE , JE MELANGE LE
    S LETTRES ...."
970 '
980 '      Fabrication de la grille
990 '
1000 DATA MANDARINE,MYRTILLE,ABRICOT,RAISIN,BANAN
    E
1010 DATA ORANGE,FRAISE,FRAMBOISE,NOISETTE,AMANDE
1020 DIM A(16,13),A$(16,13):FOR L=1 TO 10
1030 READ M$(L):NEXT:FOR L=1 TO 10
1040 X=INT(RND*15)+1:Y=INT(RND*12)+1:IF A$(X,Y)<>"" T
    HEN 1040
1050 X(L,1)=X:Y(L,1)=Y:A$(X,Y)=LEFT$(M$(L),1)
1060 A(X,Y)=L:C=0:FOR M=2 TO LEN(M$(L))
1070 B=INT(RND*4):ON B GOTO 1090,1100,1110
1080 IF Y=12 OR A$(X,Y+1)<>"" THEN 1120 ELSE Y=Y+1
    :GOTO 1140
1090 IF X=1 OR A$(X-1,Y)<>"" THEN 1120 ELSE X=X-1:
    GOTO 1140
1100 IF Y=1 OR A$(X,Y-1)<>"" THEN 1120 ELSE Y=Y-1:
```

```

GOTO 1140
1110 IF X=15 OR A$(X+1,Y)<>"" THEN 1120 ELSE X=X+1
:GOTO 1140
1120 C=C+1:IF C<20 THEN 1070 ELSE SOUND 1,90,1:FOR
R M=1 TO LEN(M$(L))
1130 A$(X(L,M),Y(L,M))="":NEXT:A(X(L,1),Y(L,1))=0:GOTO 104
0
1140 X(L,M)=X:Y(L,M)=Y:A$(X,Y)=MID$(M$(L),M,1)
1150 C=0:NEXT:NEXT:FOR I=1 TO 12:FOR J=1 TO 15
1160 IF A$(J,I)="" THEN A$(J,I)=CHR$(65+RND*25)
1170 NEXT:NEXT:CLS:T=150:XP=258:YP=202:U=8:V=6
1970 '
1980 '      Decor
1990 '
2000 PAPER 1:WINDOW 1,31,1,25:CLS:WINDOW 33,40,4,6
2010 CLS:PAPER 0:LOCATE 2,2:PRINT "FRUITS":WINDOW
1,40,1,25
2020 FOR I=8 TO 488 STEP 32:MOVE I,0:DRAWR 0,400,0:
NEXT
2030 FOR I=8 TO 392 STEP 32:MOVE 0,I:DRAWR 496,0,0:
NEXT
2040 PEN 1:FOR I=1 TO 10:LOCATE 32,I+9:PRINT M$(I):
NEXT
2050 IF NV>1 THEN 2100 ELSE J=0:PAPER 1:FOR L=1
TO 10
2060 SOUND 1,200,2:FOR M=1 TO LEN(M$(L)):PEN J
2070 LOCATE 2*X(L,M),2*Y(L,M):PRINT A$(X(L,M),Y(L,M))
2080 FOR I=1 TO 250:NEXT:IF M=1 THEN FOR I=1 TO 1
200:NEXT
2090 NEXT:FOR I=1 TO 2000:NEXT:J=3-J:NEXT:FOR I=1
TO 3000:NEXT
2100 PAPER 1:PEN 2:FOR I=1 TO 12:FOR J=1 TO 15:LO
CATE 2*J,2*I
2110 PRINT A$(J,I):NEXT:NEXT:IF NV=2 OR NV=4 THEN
3020
2970 '
2980 '      Le joueur choisit une lettre
2990 '
3000 FOR L=1 TO 10:I=LEN(M$(L)):X=X(L,I):Y=Y(L,I)
3010 PEN 0:LOCATE 2*X,2*Y:PRINT A$(X,Y):NEXT
3020 IF INKEY$<>"" THEN 3020 ELSE E=60
3030 A$=INKEY$:IF A$<>"" THEN 3070 ELSE E=E+1:IF
E<60 THEN 3030

```

```

3040 PAPER 0:PEN C:LOCATE 33,1:PRINT "DESIGNEZ":LO
CATE 33,2
3050 PRINT "UNE CASE":GOSUB 4060:E=0:C=3-C:IF NV<3
THEN 3030
3060 T=T-1:PEN 1:LOCATE 33,25:PRINT T;:IF T>0 THEN
3030 ELSE 5000
3070 IF A$<>CHR$(32) THEN GOSUB 4000:GOTO 3030
3080 IF A(U,V)=11 THEN 3020 ELSE IF A(U,V) THEN 31
10
3090 FOR I=0 TO 14:PAPER 1:PEN I:LOCATE 2*U,2*V:PR
INT A$(U,V)
3100 SOUND 1,300+10*I,3:FOR J=1 TO 90:NEXT:NEXT:T=
T-10:GOTO 3020
3110 L=A(U,V):FOR C=0 TO 3 STEP 3:PAPER 1:PEN C
3120 FOR M=1 TO LEN(M$(L)):X=X(L,M):Y=Y(L,M)
3130 LOCATE 2*X,2*Y:PRINT A$(X,Y):SOUND 1,120-3*M,2
3140 FOR K=1 TO 200:NEXT:NEXT:PAPER 0:LOCATE 32,L+
9
3150 PRINT M$(L):NEXT:PAPER 1:FOR M=1 TO LEN(M$(L)):
X=X(L,M)
3160 Y=Y(L,M):LOCATE 2*X,2*Y:PRINT " ":A(X,Y)=11:NEXT
3170 S=S+1:IF S<10 THEN C=3:GOTO 3020 ELSE 5000
3970 '
3980 '      Deplacement point lumineux
3990 '
4000 AS=ASC(A$):IF AS<240 OR AS> 243 THEN RETURN
ELSE C=1:GOSUB 4060
4010 IF AS=242 AND U>1 THEN U=U-1:XP=XP-32
4020 IF AS=243 AND U<15 THEN U=U+1:XP=XP+32
4030 IF AS=240 AND V>1 THEN V=V-1:YP=YP+32
4040 IF AS=241 AND V<12 THEN V=V+1:YP=YP-32
4050 C=3:GOSUB 4060:RETURN
4060 MOVE XP,YP:DRAW 3,3,C:MOVER -3,0:DRAW 3,-3,C:
RETURN
4970 '
4980 '      Fin du jeu
4990 '
5000 WINDOW 10,20,1,3:PAPER 1:CLS:LOCATE 2,2
5010 PEN 0:IF S=10 THEN PRINT "G A G N E" ELSE P
RINT "P E R D U"
5020 WINDOW 1,40,1,25:FOR I=1 TO 2000:NEXT
5030 IF INKEY$<>" " THEN 5030 ELSE PAPER 0
5040 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>" "

```

```
THEN CLS:END
5050 FOR I=1 TO 100:NEXT:PEN INT(RND*3)+1:LOCATE 1
,13
5060 PRINT "UNE AUTRE PARTIE : TOUCHE R":GOTO 50
40
```

LE CARRÉ CHINOIS

L'ordinateur indique que c'est lui qui commence, réfléchit deux ou trois secondes et place l'un de ses pions, une tête de diable noire, dans l'une des neuf cases que contient la grille ; puis il vous précise que c'est maintenant à vous de jouer. Choisissez ¹ n'importe quelle case libre et vous verrez aussitôt un petit bonhomme vert s'y placer. L'ordinateur choisira alors une des sept cases qui restent et vous demandera de jouer dans l'une des six cases qu'il aura laissées.

Méfiez-vous quand même et réfléchissez avant de poser vos pions sur la grille, sinon l'ordinateur aura tôt fait de gagner la manche ; il ne manquera pas alors de vous faire remarquer que vous êtes le perdant, s'attribuera un point, et, narquois, vous proposera de recommencer une partie avec la ferme intention de vous faire subir un sort identique.

Hein quoi, la règle du jeu ? Quelle règle du jeu ? Ah oui, c'est vrai, il faut bien que je vous la donne pour que vous puissiez lutter à armes égales contre un ordinateur qui la connaît parfaitement. La voici, elle est toute simple : est déclaré gagnant celui qui a réussi à aligner trois de ses pions, horizontalement, verticalement ou suivant l'une des deux diagonales. Il se peut d'ailleurs que ni vous ni l'ordinateur n'arriviez à aligner trois pions ; la partie sera déclarée nulle dans ce cas.

■ FIN DU JEU

L'ordinateur arrête le jeu dès que trois pions, les siens ou ceux du joueur sont alignés sur la grille. Un point est alors ajouté au score du vainqueur. Pour décréter qu'une partie est nulle, le programme est obligé d'attendre que les neuf cases de la grille soient occupées.

Le niveau de difficulté influence la façon dont joue l'ordinateur. Au niveau 1, il choisit ses cases sans respecter une méthode précise. Mais au niveau 2, il suit un cheminement qui doit normalement, à moins qu'on ne se montre plus malin que lui, le conduire à la victoire.

Naturellement, ce n'est pas toujours à l'ordinateur de jouer en premier dans des parties successives. Une fois sur deux, il laisse la première place au joueur.

1. Il suffit de déplacer, avec les quatre flèches, la petite étoile jaune clignotante jusqu'à ce qu'elle se trouve dans la case voulue. A ce moment-là, on appuie sur la barre d'espacement.

■ PHASES ESSENTIELLES

Sous-programme 2000

L'ordinateur fait tout pour gagner.

2000 à 2020 L'ordinateur indique qu'il va jouer en premier. Au niveau 1, il ne cherche pas à suivre la méthode qui doit, théoriquement, l'amener à remporter la manche.

2030 à 2230 L'ordinateur prend bien soin de récupérer la case centrale puis suit rigoureusement le chemin qui lui a été tracé. C'est en analysant ces lignes en détail que vous comprendrez comment on peut ne pas être entraîné irrémédiablement vers la défaite. Voici, pour vous aider, la façon dont sont numérotées les cases de la grille :

| | | |
|---|---|---|
| 1 | 2 | 3 |
| 4 | 5 | 6 |
| 7 | 8 | 9 |

| Variables | Fonction | Valeurs possibles |
|-----------|--------------------------------------|--|
| NV | Niveau de difficulté | 1 ou 2 |
| C | Numéro d'une case | 1 à 9 |
| B(C) | Valeur contenue par la case numéro C | 0 : case libre 1 : ordinateur 2 : joueur |

Sous-programme 3000

L'ordinateur joue au hasard, mais se méfie du joueur.

3000 Le sous-programme est parcouru deux fois ; la première permet de faire le point sur la situation de l'ordinateur et la seconde sur celle du joueur.

- 3010 à 3040 L'ordinateur aurait-il déjà deux pions sur une même horizontale ?
- 3050 à 3080 Ou deux pions sur une même verticale ?
- 3090 à 3120 Ou encore deux pions sur une même diagonale ?
- 3130 à 3160 Si aucune des trois configurations précédentes n'a été trouvée, une place quelconque est tirée au sort.

| Variables | Fonction | Valeurs possibles |
|-----------|---|------------------------------|
| F | Drapeau indiquant qui est concerné | 1 : ordinateur 2 : joueur |
| U0 | Drapeau indiquant à quel endroit doit se rebrancher le sous-programme 4000 une fois exécuté | 0 ou 1 |

Sous-programme 4000

L'un des participants a-t-il gagné ?

- 4000 Le sous-programme va être parcouru deux fois ; une première fois afin de savoir si l'ordinateur a gagné, et une seconde pour que l'on sache si ce ne serait pas le joueur.
- 4010 Trois pions occuperaient-ils une même ligne ?
- 4020 Ou une même colonne ?
- 4030 à 4040 Une même diagonale, alors ?
- 4050 à 4070 Si personne n'a gagné, on se doit de vérifier qu'il reste au moins une case libre. Sinon, on déclare que la partie est nulle.

■ MODIFICATIONS POSSIBLES

Programme conçu pour deux joueurs

Chacun d'eux désignerait, à tour de rôle, une case de la grille et l'ordinateur se bornerait à vérifier que l'on choisit bien une case qui

n'est pas déjà occupée. Bien entendu, ce serait à lui de dire qui a gagné.

```
50 '
60 '
70 '          LE CARRE CHINOIS
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    0:PAPER 1
110 SYMBOL 248,32,32,127,127,231,231,255,255:SYMBOL
    249,4,4,254,254,231,231,255,255
120 SYMBOL 250,127,63,56,24,31,15,7,3:SYMBOL 251,254,
    252,28,24,248,240,224,192
130 SYMBOL 252,3,3,3,3,1,1,63,63:SYMBOL 253,192,192,1
    92,192,128,128,252,252
140 SYMBOL 254,3,3,3,3,12,12,48,48:SYMBOL 255,192,192
    ,192,192,48,48,12,12
150 F$=CHR$(248)+CHR$(249)+CHR$(8)+CHR$(8)+CHR$(10)+
    CHR$(250)+CHR$(251)
160 G$=CHR$(252)+CHR$(253)+CHR$(8)+CHR$(8)+CHR$(10)
    +CHR$(254)+CHR$(255)
170 WINDOW 26,40,1,3:CLS:PAPER 0:PEN 2:LOCATE 2,2
180 PRINT "CARRE CHINOIS":PEN 1:WINDOW 1,40,1,25:LO
    CATE 1,25
190 PRINT "LE GAGNANT EST CELUI QUI ALIGNE 3 PIO
    NS";
200 A$=INKEY$:IF A$<>"" THEN 230 ELSE E=E+1:X=RND

210 IF E<60 THEN 200 ELSE PEN INT(RND*3)+1:LOCATE
    1,10
220 PRINT "CHOISISSEZ LA DIFFICULTE ( 1 OU 2 )":E
    =0:GOTO 200
230 NV=VAL(A$):IF NV<1 OR NV>2 THEN 200 ELSE LOCA
    TE 1,10
240 PRINT SPACE$(36):PAPER 1:WINDOW 28,40,13,15
970 '
980 '      Decor
990 '
1000 CLS:PEN 2:LOCATE 2,2:PRINT "JOUEUR : 0":PEN 0
1010 WINDOW 24,40,17,19:CLS:LOCATE 2,2:PRINT "ORDIN
```

```

ATEUR : 0"
1020 PEN 3:WINDOW 20,40,21,23:CLS:LOCATE 2,2
1030 PRINT "PARTIES NULLES : 0":XP=12:YP=13:U=2:V=
2
1040 PAPER 1:FOR I=4 TO 14 STEP 5:FOR J=5 TO 15
STEP 5
1050 WINDOW I,I+3,J,J+3:CLS:NEXT:NEXT:WINDOW 1,40,1,2
5:PEN 0
1060 FOR C=1 TO 9:B(C)=0:NEXT:H=H+1:IF H/2<>INT(H/2)
THEN 2000
1070 PAPER 0:PEN 3:LOCATE 2,3:PRINT "VOUS ETES LE
PREMIER"
1080 FOR I=1 TO 2000:NEXT:LOCATE 2,3:PRINT SPACE$(
20):GOSUB 5000
1090 IF C=5 THEN 3000 ELSE C=5:GOSUB 6000:GOSUB
5000:GOTO 3000
1970 '
1980 '   L'ordinateur cherche ou jouer
1990 '
2000 PAPER 0:PEN 3:LOCATE 2,3:PRINT "JE JOUE LE P
REMIER"
2010 FOR I=1 TO 2000:NEXT:LOCATE 2,3:PRINT SPACE$(
18)
2020 IF NV=1 THEN 3000
2030 C=5:GOSUB 6000:GOSUB 5000
2040 IF C/2<>INT(C/2) THEN 2090 ELSE IF C=6 OR C=
8 THEN 2080
2050 C=1:GOSUB 6000:GOSUB 5000:IF C<>9 THEN 3000
2060 IF B(2) OR B(6) THEN C=7:GOSUB 6000:GOSUB 500
0:GOTO 3000
2070 IF B(4) OR B(8) THEN C=3:GOSUB 6000:GOSUB 500
0:GOTO 3000
2080 C=9:GOSUB 6000:GOSUB 5000:IF C<>1 THEN 3000
ELSE 2060
2090 IF C<>1 THEN 2130 ELSE C=9:GOSUB 6000
2100 GOSUB 5000:IF C/2<>INT(C/2) THEN 3000
2110 IF B(4) OR B(6) THEN C=7 ELSE C=3
2120 GOSUB 6000:GOSUB 5000:GOTO 3000
2130 IF C<>9 THEN 2170 ELSE C=1:GOSUB 6000
2140 GOSUB 5000:IF C/2<>INT(C/2) THEN 3000
2150 IF B(4) OR B(6) THEN C=3 ELSE C=7
2160 GOSUB 6000:GOSUB 5000:GOTO 3000
2170 IF C<>3 THEN 2210 ELSE C=7:GOSUB 6000

```

```

2180 GOSUB 5000:IF C/2<>INT(C/2) THEN 3000
2190 IF B(4) OR B(6) THEN C=9 ELSE C=1
2200 GOSUB 6000:GOSUB 5000:GOTO 3000
2210 C=3:GOSUB 6000:GOSUB 5000:IF C/2<>INT(C/2) TH
EN 3000
2220 IF B(4) OR B(6) THEN C=1 ELSE C=9
2230 GOSUB 6000:GOSUB 5000
2970 '
2980 '   L'ordinateur est sans methode
2990 '
3000 F=1
3010 FOR C=1 TO 7 STEP 3
3020 IF B(C)=F AND B(C+1)=F AND B(C+2)=0 THEN C=C+
2:GOTO 3150
3030 IF B(C)=F AND B(C+1)=0 AND B(C+2)=F THEN C=C+
1:GOTO 3150
3040 IF B(C)=0 AND B(C+1)=F AND B(C+2)=F THEN 3150
3050 NEXT:FOR C=1 TO 3
3060 IF B(C)=F AND B(C+3)=F AND B(C+6)=0 THEN C=C+
6:GOTO 3150
3070 IF B(C)=F AND B(C+3)=0 AND B(C+6)=F THEN C=C+
3:GOTO 3150
3080 IF B(C)=0 AND B(C+3)=F AND B(C+6)=F THEN 3150
3090 NEXT:IF B(1)=F AND B(5)=F AND B(9)=0 THEN C=9:
GOTO 3150
3100 IF B(9)=F AND B(5)=F AND B(1)=0 THEN C=1:GOTO
3150
3110 IF B(3)=F AND B(5)=F AND B(7)=0 THEN C=7:GOTO
3150
3120 IF B(7)=F AND B(5)=F AND B(3)=0 THEN C=3:GOTO
3150
3130 F=F+1:IF F=2 THEN 3010
3140 C=INT(RND*9)+1:IF B(C) THEN 3140
3150 GOSUB 6000:U0=1:GOTO 4000
3160 GOSUB 5000:U0=0
3970 '
3980 '   3 pions sont-ils alignes ?
3990 '
4000 F=1
4010 FOR C=2 TO 8 STEP 3:IF B(C-1)=F AND B(C)=F AN
D B(C+1)=F THEN 8020
4020 NEXT:FOR C=4 TO 6:IF B(C-3)=F AND B(C)=F AND
B(C+3)=F THEN 8020

```

```

4030 NEXT:IF B(1)=F AND B(5)=F AND B(9)=F THEN 8020
4040 IF B(3)=F AND B(5)=F AND B(7)=F THEN 8020
4050 F=F+1:IF F=2 THEN 4010
4060 FOR C=1 TO 9:IF B(C) THEN NEXT:GOTO 8000
4070 IF U0 THEN 3160 ELSE 3000
4970 '
4980 '   Le joueur choisit une case
4990 '
5000 A$=INKEY$:IF A$<>" " THEN 5040 ELSE E=E+1:IF
E<60 THEN 5000
5010 Z=3-Z:E=0:PAPER 0:PEN Z/3+2:LOCATE 2,2:
5020 PRINT "CHOISISSEZ UNE CASE":PAPER 1
5030 PEN Z:LOCATE XP,YP:PRINT "*":GOTO 5000
5040 IF A$<>CHR$(32) THEN GOSUB 7000:GOTO 5000
5050 C=U+3*(V-1):IF B(C) THEN 5000 ELSE PAPER 1:PE
N 2
5060 LOCATE 5*U,5*V+1:PRINT G$:LOCATE XP,YP:PRINT
" "
5070 B(C)=2:PAPER 0:LOCATE 2,2:PRINT SPACE$(19):RETU
RN
5970 '
5980 '   L'ordinateur pose un pion
5990 '
6000 B(C)=1:PAPER 1:PEN 0:LOCATE 1,2:PRINT " CHUT ,
JE REFLECHIS! "
6010 FOR I=1 TO 1500:NEXT:PAPER 0:LOCATE 1,2:PRINT
SPACE$(24)
6020 X=(C-1) MOD 3 +1:Y=INT((C-1)/3)+1:PAPER 1:FOR I
=0 TO 12
6030 PEN I:LOCATE 5*X,5*Y+1:PRINT F$:SOUND 1,300+1
0*I,5
6040 IF SQ(1)>127 THEN 6040 ELSE NEXT:RETURN
6970 '
6980 '   Deplacement etoile
6990 '
7000 AS=ASC(A$):IF AS<240 OR AS>243 THEN RETURN
7010 PAPER 1:LOCATE XP,YP:PRINT " "
7020 IF AS=242 AND U>1 THEN XP=XP-5:U=U-1
7030 IF AS=243 AND U<3 THEN XP=XP+5:U=U+1
7040 IF AS=240 AND V>1 THEN YP=YP-5:V=V-1
7050 IF AS=241 AND V<3 THEN YP=YP+5:V=V+1
7060 LOCATE XP,YP:PEN 3:PRINT "*":RETURN
7970 '

```



```

7980 '    Fin du jeu
7990 '
8000 N=N+1:PAPER 1:WINDOW 21,38,7,9:CLS:PEN 0
8010 LOCATE 2,2:PRINT "PARTIE ... NULLE":GOTO 8040
8020 SOUND 1,90,5:PAPER 1:WINDOW 21,38,7,9:CLS:PEN 0
:LOCATE 2,2
8030 IF F=2 THEN PRINT "J'AI PERDU":L=L+1 ELSE PR
INT "J'AI GAGNE":M=M+1
8040 WINDOW 1,40,1,25:PEN 2:LOCATE 37,14:PRINT L
8050 PEN 0:LOCATE 37,18:PRINT M:PEN 3:LOCATE 37,22:
PRINT N
8060 IF INKEY$<>"" THEN 8060 ELSE PAPER 0
8070 IF INKEY$<>"" THEN 8100 ELSE E=E+1:IF E<60 T
HEN 8070
8080 PEN INT(RND*3)+1:E=0:LOCATE 1,1
8090 PRINT "APPUYEZ N'IMPORTE OU":GOTO 8070
8100 LOCATE 1,1:PRINT SPACE$(20):WINDOW 21,38,7,9:CL
S:GOTO 1040

```

SYMÉTRIE

Un carré de 5 cases sur 5 contient les 25 lettres qui proviennent des mots ROUGE, JAUNE, BLANC, BEIGE et MAUVE. Ces 25 lettres sont disposées sans aucun ordre sur la grille et le but du jeu est de leur faire reprendre la place qui est la leur. Les cinq caractères qui constituent le mot ROUGE devront se retrouver sur la première ligne ; ceux qui constituent le mot JAUNE sur la deuxième, etc.

Avant de voir comment se déplacent les cases dans la grille, appuyez deux ou trois fois sur la touche COPY et regardez le rectangle bleu qui est dessiné à droite de l'écran. Vous verrez qu'à chaque essai ledit rectangle change de sens ; s'il était horizontal, il devient vertical et réciproquement. D'accord ?

Venons-en maintenant à notre grille et supposons que vous ayez devant les yeux les trois premières lignes suivantes :

| | | | | |
|---|---|---|---|---|
| R | G | A | O | E |
| E | N | M | I | C |
| B | L | U | J | V |

Supposons aussi que le rectangle de droite soit horizontal. Désignez ¹ la lettre A et l'ordinateur échangera les cases G et O, leur faisant réaliser une symétrie par rapport à A.

G A O va donc devenir O A G

Voici ce qu'il est advenu de la grille :

| | | | | |
|---|---|---|---|---|
| R | O | A | G | E |
| E | N | M | I | C |
| B | L | U | J | V |

Que reste-t-il à faire pour en finir avec la première ligne ? Remplacer le A par un U, bien sûr ! Enfoncez la touche COPY pour que le rectangle bleu prenne la position verticale et faites porter votre choix

1. Il suffit de déplacer, avec les quatre flèches, le petit repère jaune clignotant jusqu'à ce qu'il se trouve dans la case voulue. A ce moment-là, on appuie sur la barre d'espacement.

sur la case M. Immédiatement, les lettres A et U échangeront leurs emplacements, subissant une symétrie verticale par rapport à M.

| | | |
|---|-----------------|---|
| A | | U |
| M | va donc devenir | M |
| U | | A |

Et la grille, dans tout cela ?

| | | | | |
|---|---|---|---|---|
| R | O | U | G | E |
| E | N | M | I | C |
| B | L | A | J | V |

Vous êtes plutôt bien parti, le premier mot a été reconstitué. Il ne vous reste plus maintenant qu'à continuer pour remettre de l'ordre dans la totalité de l'image.

Deux petites choses pour finir. La première, vous le remarquerez bien vite, est que l'ordinateur écrit les noms des couleurs dans l'ordre qu'il veut ; je veux dire qu'il ne place pas forcément le mot ROUGE sur la première ligne ou le mot MAUVE sur la dernière, non, il répartit les mots selon son humeur du moment. Et la deuxième est qu'il se croit autorisé à cacher l'une des 25 lettres derrière un petit carré vert. Il est vrai qu'au niveau 1, ceci ne nous trompe pas longtemps et que la lettre qui manque dans la grille est bien vite repérée. Mais au niveau 3...

■ FIN DU JEU

C'est simple, si l'on arrive à mettre chaque lettre à sa place, on a gagné. Sinon, il faut se résoudre à abandonner. L'ordinateur a affiché dans un coin de l'écran ce qu'il fallait faire pour jeter l'éponge. Appuyez, a-t-il écrit, sur la touche A quand vous êtes dépassé par les événements. Les lettres qui étaient cachées sous les carrés multicolores vous seront alors dévoilées et, à droite du téléviseur, la grille qu'il aurait fallu retrouver vous sera donnée.

Au niveau de difficulté 1, il n'y a qu'une seule lettre cachée par un carré. Au niveau 2, il y en a trois et, au dernier niveau, il y en a ... cinq. Dur, dur !

■ PHASES ESSENTIELLES

Sous-programme 1000

L'ordinateur constitue la grille.

- 1000 Les cinq noms de couleurs sont communiqués à l'ordinateur.
- 1010 à 1030 Ils sont ensuite placés de façon aléatoire dans la grille. Les tableaux A\$ et B\$, identiques pour l'instant, mémorisent les caractères correspondant aux 25 cases de la grille.
- 1040 à 1070 Le programme échange une cinquantaine de couples de lettres en procédant à des symétries horizontales ou verticales.
- 1080 à 1100 Suivant le niveau de difficulté, un, trois ou cinq carrés de couleur ont la charge de cacher certaines lettres de l'écran.

| Variables | Fonction | Valeurs possibles |
|-------------|---|------------------------|
| L | Numéro d'un nom de couleur | 1 à 5 |
| M\$(L) | Nom de la couleur numéro L | Chaîne alphabétique |
| U et V | Coordonnées d'une case de la grille | 1 à 5 |
| A\$(U,V) | Lettre contenue dans la case (U,V) | Caractère alphabétique |
| B\$(U,V) | Lettre contenue dans la case (U,V) mais correctement placée | Caractère alphabétique |
| D(L) | Drapeau permettant de ne pas prendre deux fois le même nom de couleur | 0 ou 1 |
| A\$ | Variable intermédiaire pour l'échange | Caractère alphabétique |
| G et A(U,V) | Couleurs des carrés qui cachent certaines lettres | 0, 2 ou 3 |

Sous-programme 3000

Le joueur choisit une case dans la grille.

- 3000 à 3040 Il est demandé au joueur, à intervalles réguliers, de désigner une lettre. Le clavier est scruté sans arrêt et, si la touche A est enfoncée, l'ordinateur donne la solution du jeu.
- 3050 à 3100 Il s'agit de l'échange entre deux cases situées sur une même horizontale. Les petits carrés de couleur, lorsqu'ils sont concernés, subissent eux aussi la symétrie.
- 3110 à 3160 Cette fois, les cases échangées sont sur la même verticale et positionnées de part et d'autre de la lettre choisie par le joueur.
- 3170 à 3180 Il faut faire le décompte de toutes les lettres qui ne sont pas à la bonne place et le faire savoir au joueur. Bien entendu, le jeu est arrêté quand le décompte est égal à zéro.

| Variables | Fonction | Valeurs possibles |
|-----------|--|-------------------|
| A\$ | Touche enfoncée | Touche quelconque |
| E | Compteur pour l'affichage d'un message | 0 à 60 |
| C | Numéro d'une couleur | 0 ou 3 |
| XP et YP | Coordonnées du repère clignotant | 10 à 26 et 7 à 23 |
| X et Y | Coordonnées de la case choisie | 1 à 5 |
| F | Drapeau indiquant le sens du rectangle | 0 ou 1 |
| A | Variable intermédiaire pour l'échange | 0, 2 ou 3 |
| B\$ | Dessin d'un petit carré | CHR\$ (143) |
| Z | Nombre de lettres mal placées | 0 à 25 |

■ MODIFICATIONS POSSIBLES

Augmentation de la taille de la grille

Il faudrait naturellement que les mots à retrouver correspondent à cette nouvelle taille.

Ajout de nouveaux mots à retrouver

Seulement cinq noms de couleur sont utilisés dans ce programme ; il n'y aurait pas de difficulté majeure à faire en sorte que l'ordinateur choisisse lui-même cinq mots parmi une liste qu'on lui aurait donnée.

Limitation du nombre de tentatives

On pourrait fixer le nombre maximal d'essais que l'on peut faire ou encore fixer le temps au bout duquel le jeu serait arrêté et le joueur déclaré perdant.

```
50 '
60 '
70 ' SYMETRIE
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24: BORDER
    0
110 PAPER 1: WINDOW 31,40,3,5:CLS:PAPER 1:PEN 2:LOCATE 2,2
120 PRINT "SYMETRIE":WINDOW 1,40,1,25:PAPER 0
130 A$=INKEY$:IF A$<>"" THEN 160 ELSE E=E+1:X=RND
140 IF E<60 THEN 130 ELSE PEN INT(RND*3)+1:LOCATE 1,12
150 PRINT "CHOISISSEZ LA DIFFICULTE ( DE 1 A 3 )"
    :E=0:GOTO 130
160 NV=VAL(A$):IF NV<1 OR NV>3 THEN 130 ELSE LOCATE 1,12
170 PRINT "UNE MINUTE , JE MELANGE LES LETTRES
    .."
970 '
980 ' Fabrication de la grille
990 '
1000 FOR L=1 TO 5:READ M$(L):NEXT:FOR V=1 TO 5
1010 L=INT(RND*5)+1:IF D(L) THEN 1010 ELSE FOR U=1
    TO 5
1020 A$(U,V)=MID$(M$(L),U,1):NEXT:D(L)=1:NEXT:FOR V=1 TO 5
1030 FOR U=1 TO 5:B$(U,V)=A$(U,V):NEXT:NEXT:FOR I=1
    TO 50
1040 U=INT(RND*5)+1:V=INT(RND*5)+1:IF U=1 OR U=5 O
```



```

R RND<0.5 THEN 1060
1050 A$=A$(U-1,V):A$(U-1,V)=A$(U+1,V):A$(U+1,V)=A$:GOTO
1080
1060 IF V=1 OR V=5 OR RND<0.5 THEN 1080 ELSE A$=
A$(U,V-1)
1070 A$(U,V-1)=A$(U,V+1):A$(U,V+1)=A$
1080 NEXT:G=1:FOR I=1 TO 2*NV-1:G=G+1-(G=4)
1090 U=INT(RND*5)+1:V=INT(RND*5)+1:IF A(U,V) THEN 10
90
1100 A(U,V)=G::NEXT:DATA ROUGE,JAUNE,BLANC,BEIGE,MAU
VE
1970 '
1980 '      Decor
1990 '
2000 B$=CHR$(143):LOCATE 1,12:PRINT SPACE$(39):PEN
3:LOCATE 3,1
2010 PRINT "IL RESTE      LETTRES A TROUVER":LOCAT
E 1,25
2020 PEN 1:PRINT "ABANDON : TOUCHE A":FOR I=1 TO
5
2030 LOCATE 31,19+I:PRINT M$(I):NEXT:PAPER 1:FOR I=
8 TO 24 STEP 4
2040 FOR J=5 TO 21 STEP 4:WINDOW I,I+2,J,J+2:CLS:NE
XT:NEXT
2050 WINDOW 1,40,1,25:FOR V=1 TO 5:FOR U=1 TO 5:LO
CATE 4*U+5,4*V+2
2060 IF A(U,V) THEN PEN A(U,V):PRINT B$ ELSE PEN 0:
PRINT A$(U,V)
2070 NEXT:NEXT:XP=18:YP=15:X=3:Y=3:F=1
2080 F=1-F:IF F THEN 2100 ELSE WINDOW 33,36,7,17:P
APER 0:CLS:PAPER 1
2090 WINDOW 29,40,11,13:CLS:LOCATE 5,2:PEN 0:PRINT "
COPY":GOTO 2120
2100 WINDOW 29,40,11,13:PAPER 0:CLS:WINDOW 33,36,7,1
7
2110 PAPER 1:CLS:PEN 0:LOCATE 1,6:PRINT "COPY"
2120 WINDOW 1,40,1,25:IF S THEN RETURN ELSE S=1:GO
TO 3170
2970 '
2980 '      Le joueur choisit une case
2990 '
3000 A$=INKEY$:IF A$<>"" THEN 3030 ELSE E=E+1:IF
E<60 THEN 3000

```

```

3010 PAPER 0:PEN C/3+2:LOCATE 9,3:PRINT "DESIGNEZ
UNE CASE"
3020 PAPER 1:LOCATE XP,YP:PRINT "+":C=3-C:E=0:GOTO
3000
3030 IF A$="A" THEN 5000 ELSE IF A$=CHR$(224) THE
N GOSUB 2080:GOTO 3000
3040 IF A$<>CHR$(32) THEN GOSUB 4000:GOTO 3000
3050 PAPER 1:IF F OR X=1 OR X=5 THEN 3110 ELSE A
$=A$(X-1,Y)
3060 A=A(X-1,Y):A$(X-1,Y)=A$(X+1,Y):A(X-1,Y)=A(X+1,Y)
3070 A$(X+1,Y)=A$:A(X+1,Y)=A:LOCATE 4*X+1,4*Y+2
3080 IF A(X-1,Y) THEN PEN A(X-1,Y):PRINT B$ ELSE PE
N 0:PRINT A$(X-1,Y)
3090 LOCATE 4*X+9,4*Y+2:IF A(X+1,Y) THEN PEN A(X+1,
Y):PRINT B$:GOTO 3170
3100 PEN 0:PRINT A$(X+1,Y):GOTO 3170
3110 IF F=0 OR Y=1 OR Y=5 THEN 3170 ELSE A$=A$(X
,Y-1)
3120 A=A(X,Y-1):A$(X,Y-1)=A$(X,Y+1):A(X,Y-1)=A(X,Y+1)
3130 A$(X,Y+1)=A$:A(X,Y+1)=A:LOCATE 4*X+5,4*Y-2
3140 IF A(X,Y-1) THEN PEN A(X,Y-1):PRINT B$ ELSE PE
N 0:PRINT A$(X,Y-1)
3150 LOCATE 4*X+5,4*Y+6:IF A(X,Y+1) THEN PEN A(X,Y+
1):PRINT B$:GOTO 3170
3160 PEN 0:PRINT A$(X,Y+1)
3170 Z=0:FOR V=1 TO 5:FOR U=1 TO 5:IF A$(U,V)<>B$(U,
V) THEN Z=Z+1
3180 NEXT:NEXT:PEN 3:PAPER 0:LOCATE 11,1:PRINT Z:IF
Z THEN 3000 ELSE 6000
3970 '
3980 '      Deplacement repere
3990 '
4000 AS=ASC(A$):IF AS<240 OR AS>243 THEN RETURN
4010 PAPER 1:LOCATE XP,YP:PRINT " "
4020 IF AS=242 AND X>1 THEN XP=XP-4:X=X-1
4030 IF AS=243 AND X<5 THEN XP=XP+4:X=X+1
4040 IF AS=240 AND Y>1 THEN YP=YP-4:Y=Y-1
4050 IF AS=241 AND Y<5 THEN YP=YP+4:Y=Y+1
4060 PEN 3:LOCATE XP,YP:PRINT "+":RETURN
4970 '
4980 '      Le joueur a abandonne
4990 '
5000 FOR V=1 TO 5:FOR U=1 TO 5:IF A(U,V)=0 THEN 50

```

```

20
5010 PEN 1:PAPER A(U,V):LOCATE 4*U+5,4*V+2:PRINT A$(U,V)
5020 NEXT:NEXT:PEN 0:PAPER 1:WINDOW 29,36,19,24:CLS
5030 PRINT "SOLUTION":FOR V=1 TO 5:LOCATE 2,V+1:FOR U=1 TO 5
5040 PRINT B$(U,V);:SOUND 1,300+3*U+3*V,4:NEXT:NEXT
5050 WINDOW 1,5,9,11:CLS:PEN 3:LOCATE 1,2:PRINT "PERDU":GOTO 6020
5970 '
5980 '      Fin du jeu
5990 '
6000 WINDOW 1,5,9,11:PAPER 1:PEN 0:CLS:LOCATE 1,2:PRINT "GAGNE"
6010 FOR I=1 TO 40:SOUND 1,150-3*I,4:NEXT
6020 IF INKEY$<>"" THEN 6020 ELSE WINDOW 1,40,1,25
6030 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>"" THEN CLS:END
6040 FOR I=1 TO 100:NEXT:PEN INT(RND*3)+1:LOCATE 1,20
6050 PAPER 0:PRINT "UNE AUTRE PARTIE : TOUCHE R":GOTO 6030

```

LES BRIGANDS

Vous voyez les trois personnages jaunes sur la gauche de l'écran ? Ce sont des brigands. Et les trois personnages verts qui sont à proximité et qui ont un fusil sur l'épaule ? Eh bien, ce sont les gendarmes qui les ont fait prisonniers. Les seconds ont la lourde tâche d'emmener les premiers sur l'île où se trouve le pénitencier, c'est-à-dire sur la droite de l'écran. Le problème qui est le leur, et qui va devenir le vôtre, est le suivant : on ne dispose pour rejoindre la prison que d'une barque qui ne peut contenir que deux personnes au maximum. Il vous faudra donc exécuter plusieurs allers-retours pour que notre petit monde soit arrivé à bon port.

Voyons comment il faut s'y prendre ; commencez par faire porter votre choix ¹ sur l'un des gardiens et vous le verrez immédiatement sauter dans la barque. Faites la même chose avec un deuxième gardien et il ira s'asseoir à côté de son camarade dans l'embarcation. Aussitôt celle-ci se mettra en marche et se dirigera vers la partie droite du téléviseur. Mais elle n'atteindra jamais sa destination car le jeu sera arrêté brusquement par l'ordinateur et l'on vous fera remarquer que vous avez perdu. Qu'est-ce que vous avez fait de mal ? Voyons, faisons les comptes, vous avez laissé sur la berge un gendarme et trois bandits, et ces derniers, avec l'avantage du nombre, ont fait un mauvais sort à notre malheureux gardien. Vous venez de découvrir la règle qu'il va vous falloir respecter. Si vous voulez gagner, vous devrez faire en sorte de ne jamais laisser sur l'une des deux berges des bandits qui soient supérieurs en nombre aux gendarmes. OK ?

On recommence une nouvelle partie. Faites monter dans la barque un gendarme et un voleur et, cette fois, l'embarcation naviguera sans ennui jusqu'à la rive opposée. Et là, que ferez-vous ? Ramener la barque à vide pour reprendre deux nouveaux passagers ? Pas question, l'ordinateur refuse de déplacer le bateau s'il n'y a pas au moins une personne à bord. Remettez donc le gendarme dans la barque et appuyez sur la touche B. Votre gardien sera reconduit seul sur le premier rivage.

Voilà, c'est à vous maintenant de découvrir comment enchaîner les trajets suivants pour que les six personnages se retrouvent ensemble sur la rive droite.

1. Il suffit de déplacer, avec les quatre flèches, le petit repère coloré jusqu'à ce qu'il se trouve sous le bonhomme voulu. A ce moment-là, on appuie sur la barre d'espacement.

■ FIN DU JEU

Une partie est gagnée lorsque l'on a réussi à faire passer de gauche à droite les six personnages. Si l'on a laissé sur une berge plus de brigands que de gardiens, l'ordinateur décrète que l'on a perdu.

En choisissant le niveau de difficulté numéro 2, on a affaire à des bandits récalcitrants. Ils ne tiennent pas à rester sur la rive droite et, dès que l'on hésite un peu, ils en profitent pour remonter dans la barque. Rien d'autre à faire alors que de les ramener sur la première berge.

■ PHASES ESSENTIELLES

Sous-programme 2000

L'embarquement des passagers est effectué sur la rive gauche.

2000 à 2040 En attendant que le joueur choisisse un personnage, un message clignote dans diverses couleurs. Lorsque l'un des bonshommes est déjà installé sur la barque, un second message apparaît, indiquant que l'on peut, si on le souhaite, faire partir le bateau.

2050 à 2060 Le programme analyse la touche que l'on a enfoncée. Un branchement vers le sous-programme de déplacement du bateau est réalisé si nécessaire.

2070 à 2090 C'est un gendarme que l'on veut faire embarquer. Son graphisme est effacé de la berge et sa partie supérieure est dessinée sur le bateau.

2100 à 2120 Par contre, là, c'est un brigand qui saute dans le bateau.

| Variables | Fonction | Valeurs possibles |
|-----------|--|-------------------|
| XP et YP | Coordonnées du repère clignotant | 3 à 37 et 18 à 24 |
| A\$ | Touche enfoncée | Touche quelconque |
| E | Compteur pour l'affichage d'un message | 0 à 60 |
| H | Nombre de personnages sur la barque | 0 à 2 |

| Variables | Fonction | Valeurs possibles |
|-----------|--|--|
| X et Y | Coordonnées du personnage désigné par le joueur | 1 à 3 et 1 à 2 |
| A(X) | Drapeau indiquant si la case X est occupée par un gendarme | 0 ou 1 |
| E\$ | Effacement d'un personnage | Chaîne caractères |
| E(H) | Nature d'un personnage embarqué | 0 : place libre 1 : gendarme 2 : brigand |
| A | Nombre de gendarmes restants | 0 à 3 |
| B(X) | Drapeau indiquant si la case X est occupée par un brigand | 0 ou 1 |
| B | Nombre de brigands restants | 0 à 3 |

Sous-programme 3000

La barque se dirige vers la rive droite.

- 3000 Les messages qui demandaient d'indiquer un bon-homme sont effacés.
- 3010 à 3070 Les dessins des personnages embarqués et du bateau sont décalés d'une unité sur la droite pour donner l'illusion du mouvement. Lorsque l'embarcation arrive au milieu du bras de mer, l'ordinateur vérifie qu'il ne reste pas plus de bandits que de gendarmes sur une berge.
- 3080 à 3110 Le débarquement d'un gardien, s'il y en a un qui a fait la traversée, est réalisé.
- 3120 à 3150 C'est au tour des bandits d'être débarqués.
- 3160 Le jeu se poursuit sauf si les six personnages sont arrivés à bon port.

| Variables | Fonction | Valeurs possibles |
|-----------|--|-------------------|
| U | Abscisses du bateau et des personnages | 13 à 24 |

| Variables | Fonction | Valeurs possibles |
|-----------|--|-------------------|
| B\$ | Graphisme du bateau | Chaîne caractères |
| C\$ | Graphisme d'un gendarme | Chaîne caractères |
| C(J) | Drapeau indiquant si la case J est occupée par un gendarme | 0 ou 1 |
| C | Nombre de gendarmes sur la rive droite | 0 à 3 |
| D\$ | Graphisme d'un brigand | Chaîne caractères |
| D(J) | Drapeau indiquant si la case J est occupée par un brigand | 0 ou 1 |
| D | Nombre de brigands sur la rive droite | 0 à 3 |

■ MODIFICATIONS POSSIBLES

Instauration d'un compteur de traversées

On ne serait alors déclaré gagnant qu'en réussissant à amener les six personnages sur la rive droite en un nombre précis de coups.

Augmentation du nombre de gendarmes et de voleurs

Au lieu de n'avoir que six bonshommes dans le jeu, on en aurait huit, ou, pourquoi pas ? dix.

```

50 '
60 '
70 '          LES   BRIGANDS
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    0:PAPER 1
110 WINDOW 24,35,1,3:CLS:PAPER 0:PEN 2:LOCATE 2,2
120 PRINT " BRIGANDS ":WINDOW 1,40,1,25:PEN 3
130 SYMBOL 252,128,70,38,20,12,60,62,62:SYMBOL 253,2
    8,28,28,28,28,8,8,12
140 SYMBOL 254,0,24,24,16,56,124,186,121:SYMBOL 255,
    56,56,16,16,16,16,16,24
150 B$=CHR$(213)+CHR$(143)+CHR$(143)+CHR$(143)+CHR$
    (212)

```

```

160 C$=CHR$(252)+CHR$(8)+CHR$(10)+CHR$(253)
170 D$=CHR$(254)+CHR$(8)+CHR$(10)+CHR$(255)
180 A$=INKEY$:IF A$<>"" THEN 210 ELSE E=E+1:X=RND
190 IF E<60 THEN 180 ELSE PEN INT(RND*3)+1:LOCATE
  1,10
200 PRINT "CHOISISSEZ LA DIFFICULTE (1 OU 2)":E=0:
GOTO 180
210 NV=VAL(A$):IF NV<1 OR NV>2 THEN 180 ELSE LOCA
TE 1,10
220 PRINT SPACE$(35):A=3:B=3:SYMBOL 251,0,0,0,8,28,42
  ,8,8
970 '
980 '      Decor
990 '
1000 MOVE 0,302:DRAWR 639,0,1:MOVE 0,304:DRAWR 639,
  0,1
1010 MOVE 160,0:DRAWR 0,302,1:MOVE 162,0:DRAWR 0,30
  2,1
1020 MOVE 460,0:DRAWR 0,302,1:MOVE 458,0:DRAWR 0,30
  2,1
1030 PEN 2:FOR I=3 TO 9 STEP 3:LOCATE I,16:PRINT C
  $:A(I/3)=1:NEXT
1040 PEN 3:FOR I=3 TO 9 STEP 3:LOCATE I,22:PRINT D
  $:B(I/3)=1:NEXT
1050 PEN 1:LOCATE 12,20:PRINT B$:E$=" "+CHR$(8)+CH
  R$(10)+" "
1970 '
1980 '      Embarquement a gauche
1990 '
2000 IF INKEY$<>"" THEN 2000 ELSE XP=9:YP=18:X=3:Y
  =1
2010 A$=INKEY$:IF A$<>"" THEN 2050 ELSE E=E+1:IF
  E<60 THEN 2010
2020 PEN INT(RND*3)+1:LOCATE 1,8:PRINT "PERSONNAGE
  "
2030 E=0:LOCATE XP,YP:PRINT CHR$(251):IF H=0 THEN
  2010
2040 LOCATE 12,8:PRINT "BATEAU":GOTO 2010
2050 IF A$="B" AND H THEN 3000
2060 IF A$<>CHR$(32) THEN GOSUB 6000:GOTO 2010
2070 IF Y=2 THEN 2100 ELSE IF A(X)=0 THEN 2010
2080 A(X)=0:LOCATE 3*X,16:PRINT E$:PEN 2:LOCATE 13+
  2*H,19

```

```

2090 PRINT CHR$(252):H=H+1:E(H)=1:A=A-1:IF H=1 THEN
  2010 ELSE 3000
2100 IF B(X)=0 THEN 2010 ELSE B(X)=0:LOCATE 3*X,22
2110 PRINT E$:PEN 3:LOCATE 13+2*H,19:PRINT CHR$(25
4)
2120 H=H+1:E(H)=2:B=B-1:B(X)=0:IF H<>2 THEN 2010
2970 '
2980 '   Trajet gauche-droite
2990 '
3000 LOCATE 1,8:PRINT SPACE$(10):LOCATE 12,8:PRINT
SPACE$(6)
3010 LOCATE XP,YP:PRINT " ":FOR U=13 TO 24:LOCATE
U,19
3020 IF E(1)=1 THEN PEN 2:PRINT " ";CHR$(252);
3030 IF E(1)=2 THEN PEN 3:PRINT " ";CHR$(254);
3040 IF E(2)=1 THEN PEN 2:PRINT " ";CHR$(252)
3050 IF E(2)=2 THEN PEN 3:PRINT " ";CHR$(254)
3060 PEN 1:LOCATE U-1,20:PRINT " ";B$:SOUND 1,300+
10*U,9
3070 IF SQ(1)>127 THEN 3070 ELSE IF U=18 THEN GOS
UB 7000
3080 NEXT:FOR I=1 TO 2:IF E(I)<>1 THEN 3120
3090 FOR J=1 TO 3:IF C(J) THEN 3110 ELSE LOCATE 2
3+2*I,19
3100 PRINT " ":PEN 2:LOCATE 28+3*J,16:PRINT C$:C(J)
=1:J=3:C=C+1
3110 NEXT:GOTO 3160
3120 IF E(I)<>2 THEN 3160
3130 FOR J=1 TO 3:IF D(J) THEN 3150 ELSE LOCATE 2
3+2*I,19
3140 PRINT " ":PEN 3:LOCATE 28+3*J,22:PRINT D$:D(J)
=1:J=3:D=D+1
3150 NEXT
3160 NEXT:H=0:E(1)=0:E(2)=0:IF C+D=6 THEN 7020
3970 '
3980 '   Embarquement a droite
3990 '
4000 IF INKEY$<>"" THEN 4000 ELSE XP=31:YP=18:X=1:
Y=1
4010 A$=INKEY$:IF A$<>"" THEN 4050 ELSE E=E+1:IF
E<60 THEN 4010
4020 PEN INT(RND*3)+1:LOCATE 31,8:PRINT "PERSONNAG
E"

```

```

4030 E=0:LOCATE XP,YP:PRINT CHR$(251):IF NV=2 AND
RND<0.2 THEN 8000
4040 IF H=0 THEN 4010 ELSE LOCATE 23,8:PRINT "BAT
EAU":GOTO 4010
4050 IF A$="B" AND H THEN 5000
4060 IF A$<>CHR$(32) THEN GOSUB 6000:GOTO 4010
4070 IF Y=2 THEN 4100 ELSE IF C(X)=0 THEN 4010
4080 C(X)=0:LOCATE 3*X+28,16:PRINT E$:PEN 2:LOCATE
25+2*H,19
4090 PRINT CHR$(252):H=H+1:E(H)=1:C=C-1:IF H=1 THEN
4010 ELSE 5000
4100 IF D(X)=0 THEN 4010 ELSE D(X)=0:LOCATE 3*X+28,
22
4110 PRINT E$:PEN 3:LOCATE 25+2*H,19:PRINT CHR$(25
4)
4120 H=H+1:E(H)=2:D=D-1:D(X)=0:IF H<>2 THEN 4010
4970 '
4980 '      Trajet droite-gauche
4990 '
5000 LOCATE 31,8:PRINT SPACE$(10):LOCATE 23,8:PRINT
SPACE$(6)
5010 LOCATE XP,YP:PRINT " ":FOR U=24 TO 13 STEP -
1:LOCATE U,19
5020 IF E(1)=1 THEN PEN 2:PRINT CHR$(252);" ";
5030 IF E(1)=2 THEN PEN 3:PRINT CHR$(254);" ";
5040 IF E(2)=1 THEN PEN 2:PRINT CHR$(252);" "
5050 IF E(2)=2 THEN PEN 3:PRINT CHR$(254);" "
5060 PEN 1:LOCATE U-1,20:PRINT B$;" ":SOUND 1,500-
10*U,9
5070 IF SQ(1)>127 THEN 5070 ELSE IF U=18 THEN GOS
UB 7000
5080 NEXT:FOR I=1 TO 2:IF E(I)<>1 THEN 5120
5090 FOR J=1 TO 3:IF A(J) THEN 5110 ELSE LOCATE 1
1+2*I,19
5100 PRINT " ":PEN 2:LOCATE 3*J,16:PRINT C$:A(J)=1:J
=3:A=A+1
5110 NEXT:GOTO 5160
5120 IF E(I)<>2 THEN 5160
5130 FOR J=1 TO 3:IF B(J) THEN 5150 ELSE LOCATE 1
1+2*I,19
5140 PRINT " ":PEN 3:LOCATE 3*J,22:PRINT D$:B(J)=1:J
=3:B=B+1
5150 NEXT

```

```

5160 NEXT:H=0:E(1)=0:E(2)=0:GOTO 2000
5970 '
5980 '   Deplacement repere
5990 '
6000 AS=ASC(A$):IF AS<240 OR AS>243 THEN RETURN E
LSE LOCATE XP,YP:PRINT " "
6010 IF AS=242 AND X>1 THEN XP=XP-3:X=X-1
6020 IF AS=243 AND X<3 THEN XP=XP+3:X=X+1
6030 IF AS=240 AND Y>1 THEN YP=YP-6:Y=Y-1
6040 IF AS=241 AND Y<2 THEN YP=YP+6:Y=Y+1
6050 PEN 2:LOCATE XP,YP:PRINT CHR$(251):RETURN
6970 '
6980 '   A-t-on perdu ?
6990 '
7000 IF (A=0 OR A>=B) AND (C=0 OR C>=D) THEN RETU
RN ELSE PEN 0
7010 WINDOW 14,24,10,12:PAPER 1:CLS:LOCATE 2,2:PRINT
"P E R D U":GOTO 7030
7020 WINDOW 14,24,10,12:PAPER 1:CLS:PEN 0:LOCATE 2,
2:PRINT "G A G N E"
7030 FOR I=1 TO 20:SOUND 1,200-5*I,10:NEXT
7040 IF INKEY$<>"" THEN 7040 ELSE WINDOW 1,40,1,25
:PAPER 0
7050 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>""
THEN CLS:END
7060 FOR I=1 TO 100:NEXT:PEN INT(RND*3)+1:LOCATE 3
,5
7070 PRINT "UNE AUTRE PARTIE : TOUCHE R":GOTO 70
50
7970 '
7980 '   Le brigand se sauve
7990 '
8000 FOR I=1 TO 3:IF D(I) THEN X=I:GOTO 4100 ELSE
NEXT:GOTO 4010

```

ANAGRAMME

Une remarque préliminaire : si vous avez déjà tapé le jeu du pendu, vous n'aurez pas beaucoup de mal à entrer ce programme dans votre ordinateur. Une bonne partie des lignes sont en effet identiques dans les deux cas.

Le jeu maintenant : demandez à l'ordinateur de choisir dans sa mémoire un mot, simple ou compliqué selon votre goût. Admettons qu'il ait choisi le mot *sucre* et voyons ce qu'il va en faire. Il va mélanger avec soin toutes les lettres qui constituent ce mot et proposer à votre sagacité le résultat de cette opération. Les cinq lettres qui seront alors affichées dans des cases bleues seront, peut-être, les suivantes : *ucser*. Et ce sera à vous, comme vous l'aviez probablement deviné, de remettre de l'ordre dans tout ceci. Vous ferez porter votre choix ¹ sur l'une des lettres, par exemple le S. Immédiatement, la lettre en question se mettra à clignoter et l'ordinateur vous demandera de choisir une des cinq cases vides qu'il a dessinées un peu plus bas sur l'écran. Si vous désignez la première d'entre elles, la lettre S s'y placera. Vous pourrez ensuite vous occuper des autres lettres, jusqu'à ce que vous ayez reconstitué la totalité du mot *sucre* dans les cases du bas. Il reste à savoir ce qui se serait passé si vous aviez indiqué une mauvaise case. Pas grand'chose, il ne se serait pas passé grand'chose. L'ordinateur vous aurait fait remarquer gentiment que vous veniez de commettre une faute et aurait passé l'éponge. Et même, il aurait pris le soin d'aller placer lui-même la lettre dans la bonne case. Sympathique, notre machine, non ? Enfin, sympa, oui, au début, car si l'on se trompe plus de fois qu'elle ne le juge acceptable, elle arrête le jeu et déclare que l'on a perdu. Sans autre forme de procès !

■ FIN DU JEU

On ne peut gagner une partie qu'en retrouvant le mot de l'ordinateur. Le nombre d'erreurs qui sont tolérées est affiché sur l'écran ; il varie en fonction de la difficulté des mots à reconstituer.

Le niveau de difficulté est lié à la complexité des mots que l'ordinateur choisit. Au niveau 1, par exemple, il ne propose que des mots

1. Il suffit de déplacer, avec les flèches horizontales, le petit repère vert clignotant jusqu'à ce qu'il se trouve sous la lettre voulue. A ce moment-là, on appuie sur la barre d'espacement.

d'un usage très courant, et, au niveau 4, il nous demande de reformer des mots tout à fait invraisemblables.

■ PHASES ESSENTIELLES

Sous-programme 2000

Choix d'une des lettres proposées par l'ordinateur.

2000 à 2020 Le programme indique, par un message de différentes couleurs, que c'est l'une des cases du haut qui doit être désignée par le joueur.

2030 Si cela est nécessaire, on déplace le petit repère clignotant sous une nouvelle lettre.

2040 à 2050 La lettre choisie est colorée en jaune, et le message est effacé.

| Variables | Fonction | Valeurs possibles |
|-----------|--|------------------------|
| A\$ | Touche enfoncée | Touche quelconque |
| E | Compteur pour l'affichage d'un message | 0 à 20 |
| X puis XO | Numéro de la première lettre désignée | 1 à 13 |
| A + 4 | Abscisse de la première case utilisée | 2 à 13 |
| A\$(X) | Lettre contenue dans la case X | Caractère alphabétique |
| G | Numéro d'une couleur | 1 ou 3 |
| XP et YP | Coordonnées du repère clignotant | 2 à 39 et 11 à 20 |

Sous-programme 3000

Recherche du bon emplacement pour une lettre.

3000 à 3030 Le joueur est prévenu que c'est l'une des cases du bas qui doit, cette fois, être retenue. En même temps, la lettre précédemment choisie se met à clignoter.

- 3040 à 3050 Si les flèches horizontales sont enfoncées, on déplace le repère, et si c'est la barre d'espacement qui est actionnée, on vérifie que le coup est valide.
- 3060 à 3090 Une lettre a été mise au bon endroit. Elle est effacée de la case d'où elle venait et affichée à sa place réelle.
- 3100 à 3120 On a essayé de mettre une lettre à un mauvais emplacement. L'ordinateur le fait savoir en faisant clignoter un point d'interrogation...
- 3130 à 3160 ... puis nous indique la bonne solution. Il vérifie alors que l'on ne s'est pas trompé plus de fois qu'il n'est autorisé ; auquel cas, il arrête le jeu et nous déclare perdant.

| Variables | Fonction | Valeurs possibles |
|-----------|---------------------------------------|------------------------|
| X puis X1 | Numéro de la deuxième lettre désignée | 1 à 13 |
| B\$(X) | Lettre contenue dans la case X | Caractère alphabétique |
| C | Nombre de lettres trouvées | 1 à 13 |
| L | Longueur du mot à retrouver | 1 à 13 |
| N | Nombre d'erreurs | 1 à 4 |
| NV | Niveau de difficulté | 1 à 4 |

■ MODIFICATIONS POSSIBLES

Augmentation du nombre de mots connus par l'ordinateur

80 de ceux-ci sont contenus dans le programme et il n'y aurait pas de difficulté majeure à en ajouter de nouveaux.

Instauration d'un chronomètre

Pour gagner une partie, il faudrait reconstituer la réponse en un temps choisi par l'ordinateur en fonction du niveau de difficulté.

```

50 '
60 '
70 '           ANAGRAMME
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    0:PAPER 0
110 PEN 2:LOCATE 1,8:PRINT "TOUCHE 1: L'ORDINATEUR
    CHOISIRA UN MOT"
120 LOCATE 1,19:PRINT "TOUCHE 2: L'UN DE VOUS CHO
    ISIT UN MOT"
130 PRINT "QUE LES AUTRES DEVRONT RETROUVER":PE
    N 3
140 A$=INKEY$:IF A$="" THEN X=RND:GOTO 140
150 IF A$="2" THEN 210 ELSE IF A$<>"1" THEN 140
160 LOCATE 1,19:PRINT SPACE$(72):LOCATE 1,11
170 PRINT "CHOISISSEZ LA DIFFICULTE ( DE 1 A 4 )"
180 A$=INKEY$:IF A$="" THEN 180 ELSE NV=VAL(A$):IF
    NV<1 OR NV>4 THEN 180
190 X=INT(RND*20)+1:X=X+(NV-1)*20:FOR I=1 TO X:READ
    M$:NEXT
200 IF LEN(M$)>13 THEN RESTORE:GOTO 190 ELSE 1000
210 LOCATE 1,22:LINE INPUT "MOT A RETROUVER ";M$
220 IF LEN(M$)>13 OR LEN(M$)<3 THEN RUN ELSE NV=3
970 '
980 '           Decor
990 ' ,
1000 CLS:PAPER 1:WINDOW 27,39,1,3:CLS:PAPER 0:PEN 2
1010 LOCATE 2,2:PRINT " ANAGRAMME ":L=LEN(M$):A=3*
    INT((13-L)/2)-2
1020 PAPER 1:WINDOW A+3,A+3*L+2,7,9:CLS:WINDOW A+3,
    A+3*L+2,16,18
1030 CLS:WINDOW 1,40,1,25:XP=A+4:YP=11:X=1:G=1
1040 FOR I=47 TO 575 STEP 48:MOVE I,112:DRAWR 0,19
    2,0
1050 MOVE I+2,112:DRAWR 0,192,0:NEXT:DIM A$(L),B$(L)
1060 FOR I=1 TO L:A$(I)=MID$(M$,I,1):B$(I)=A$(I):NEXT
1070 FOR R=1 TO 30:I=INT(RND*L)+1:J=INT(RND*L)+1
1080 A$=A$(I):A$(I)=A$(J):A$(J)=A$:NEXT:PEN 0:FOR I=1
    TO L
1090 LOCATE A+3*I+1,8:PRINT A$(I):NEXT:PAPER 0:PEN
    2
1100 LOCATE 2,22:PRINT "NOMBRE D'ERREURS COMMISES

```

```

: 0":PEN 3
1110 LOCATE 2,24:PRINT "NOMBRE D'ERREURS MAXIMUM
:":NV
1970 '
1980 '      Lettre du haut
1990 '
2000 A$=INKEY$:IF A$<>"" THEN 2030 ELSE E=E+1:IF
E<60 THEN 2000
2010 G=4-G:E=0:PAPER 0:PEN (G-1)/2+2:LOCATE 11,5
2020 PRINT "DESIGNEZ UNE CASE":LOCATE XP,11:PRINT
CHR$(240):GOTO 2000
2030 IF A$<>CHR$(32) THEN GOSUB 4000:GOTO 2000 EL
SE PAPER 1:PEN 3
2040 IF A$(X)="" THEN 2000 ELSE LOCATE A+3*X+1,8:P
RINT A$(X):PAPER 0
2050 LOCATE XP,11:PRINT " ":LOCATE 11,5:PRINT SPAC
E$(17):E=60:YP=20:X0=X
2970 '
2980 '      Lettre du bas
2990 '
3000 A$=INKEY$:IF A$<>"" THEN 3040 ELSE E=E+1:IF
E<60 THEN 3000
3010 G=4-G:E=0:PAPER 0:PEN (G-1)/2+2:LOCATE 11,14
3020 PRINT "DESIGNEZ UNE CASE":LOCATE XP,20:PRINT
CHR$(240)
3030 PEN G:PAPER 1:LOCATE A+3*X0+1,8:PRINT A$(X0):G
OTO 3000
3040 X1=X:IF A$<>CHR$(32) THEN GOSUB 4000:GOTO 30
00
3050 IF B$(X)="1" THEN 3000 ELSE PAPER 1:PEN 0
3060 IF B$(X)<>A$(X0) THEN 3100 ELSE LOCATE A+3*X+
1,17
3070 PRINT A$(X0):SOUND 1,200,9:A$(X0)="" :LOCATE A+3
*X0+1,8:PRINT " "
3080 B$(X)="1":C=C+1:IF C=L THEN 6030 ELSE PAPER 0
:YP=11:X=X1
3090 LOCATE XP,20:PRINT " ":LOCATE 11,14:PRINT SPA
CE$(17):GOTO 2000
3100 PAPER 1:FOR I=2 TO 13:PEN I:SOUND 1,400+10*I,
8
3110 IF SQ(1)>127 THEN 3110 ELSE LOCATE A+3*X+1,1
7
3120 PRINT "?":NEXT:FOR X=1 TO L:IF B$(X)<>A$(X0) TH

```

```

EN NEXT
3130 FOR I=1 TO 500:NEXT:FOR I=0 TO 12:SOUND 1,30
0+5*I,8
3140 IF SQ(1)>127 THEN 3140 ELSE PEN 1:LOCATE A+3
*X+1,17:PRINT A$(X0)
3150 NEXT:A$(X0)="" :N=N+1:PAPER 0:PEN 2:LOCATE 29,2
2:PRINT N
3160 PAPER 1:LOCATE A+3*X0+1,8:PRINT " ":IF N<=NV
THEN 3080 ELSE 6000
3970 '
3980 '      Déplacement repere
3990 '
4000 AS=ASC(A$):IF AS<242 OR AS>243 THEN RETURN E
LSE PAPER 0
4010 LOCATE XP,YP:PRINT " ":IF AS=242 AND X>1 THE
N XP=XP-3:X=X-1
4020 IF AS=243 AND X<L THEN XP=XP+3:X=X+1
4030 LOCATE XP,YP:PEN 3:PRINT CHR$(240):RETURN
4970 '
4980 '      Mots de l'ordinateur
4990 '
5000 DATA SUCRE,VOITURE,BALANCE,ECOLE,RENARD,ENFA
NT,GLACE,ARDOISE
5010 DATA LIVRE,CERCLE,POMME,BRIQUE,ORANGE,FOUDRE,
ARMOIRE,MUSIQUE
5020 DATA CHAISE,NAVIRE,CABINE,TIGRE,APPETIT,CAFETI
ERE,FAUVETTE
5030 DATA DECIMETRE,GOUVERNEUR,OMBRAJE,DEGRADATI
ON,OSSATURE
5040 DATA ARCHIDUC,FEUILLETON,BANDEROLLE,FAINEANT
,ESCALIER
5050 DATA ARRIMAGE,DECAGONE,CHAMPAGNE,GAZOMETRE,A
RISTOCRATE
5060 DATA HARMONICA,ECHANTILLON,CARICATURE,ELECTR
OSCOPE,COAGULATION
5070 DATA APPRECIATION,DEMINERALISATION,BARYTON,H
IPPOCAMPE
5080 DATA BRANDEBOURG,DESSICCATION,HEMISPHERIQUE,
CAHOTEMEMT
5090 DATA DEHARNACHEMENT,GRAPHOLOGUE,HOMONYME,CL
AVECINISTE
5100 DATA GALACTOMETRE,BELLIGERANT,ECCLESIASTIQUE
,DYNAMITAGE

```

```

5110 DATA ASSUJETTISSEMENT,ZIRCONIUM,CASTANEACEES
,FORMALDEHIDE
5120 DATA BUTYROMETRE,SYNALLAGMATIQUE,TRANSGANG
ETIQUE,EMPHYTEOTIQUE
5130 DATA ZINGLBERACEES,BALENOPTERE,XEROPHTALMIE,
TETRASYLLABIQUE
5140 DATA CEPHALOTHORAX,ZINCOGRAVURE,ECHINODERME
,ASPARAGINEES
5150 DATA FLUORESCINE,ELLIPSOGRAPHE,HYPOCONDIAQ
UE
5160 DATA CARBORUNDUM,HIPPOCASTANEES
5970 '
5980 '      Fin du jeu
5990 '
6000 WINDOW 15,21,3,5:PAPER 1:CLS:PEN 0:LOCATE 2,2
6010 PRINT "PERDU":WINDOW 1,40,1,25:PAPER 0:PEN 3
6020 LOCATE 1,1:PRINT "REPOSE : ";M$:GOTO 6040
6030 WINDOW 15,21,3,5:PAPER 1:CLS:PEN 0:LOCATE 2,2:P
RINT "GAGNE"
6040 WINDOW 1,40,1,25:FOR I=1 TO 3000:NEXT:PAPER 0:
PEN 2
6050 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>" "
THEN CLS:END
6060 FOR I=1 TO 80:NEXT:PEN INT(RND*3)+1:LOCATE 5,
14
6070 PRINT "UNE AUTRE PARTIE: TOUCHE R":GOTO 605
0

```

AU SUIVANT

Dans une grille de six carrés sur quatre est apparu un petit bonhomme vert, et l'ordinateur vous demande de choisir ¹ une case. Vous n'avez, pour l'instant, pas d'autre possibilité que de désigner le petit personnage. Faites-le donc et regardez la suite. Le premier bonhomme s'efface mais réapparaît aussitôt sur l'écran, accompagné maintenant d'un bateau. C'est sur ce dernier que va devoir porter votre choix ; vous avez, en effet, à respecter une règle de jeu plutôt facile à comprendre : on doit toujours retrouver, pour gagner, le dessin qui a été ajouté aux précédents. Ainsi donc, quand l'ordinateur vous proposera, au troisième tour, un bonhomme, un bateau et un hélicoptère, vous retiendrez l'hélicoptère. D'accord ?

Vous trouvez que c'est un jeu plutôt facile ? Hum, oui, au début, quand il n'y a que quelques dessins sur l'écran. Mais lorsque l'on arrive aux alentours du dixième coup, les affaires se compliquent. Surtout que l'ordinateur ne s'est pas cassé la tête à composer 24 dessins différents ; non, au bout d'un moment il nous ressort tranquillement les graphiques qu'il a déjà utilisés. Allez donc savoir, quand il y a par exemple deux girafes sur la grille, si c'est plutôt l'une ou plutôt l'autre qui vient d'être ajoutée à la liste des dessins précédents...

■ FIN DU JEU

Lorsque l'on se trompe de case, l'ordinateur nous montre, en le coloriant en bleu, le dessin qu'il aurait fallu choisir et note, dans un coin de l'image, que l'on vient de commettre une erreur. Le jeu se poursuit alors car ce n'est qu'au bout de quatre erreurs que la partie s'arrête et que le tour passe au joueur suivant. Le maximum de points que l'on puisse gagner dans une partie est égal à 24 et correspond au fait que l'on a retrouvé, sans aucune faute, la totalité des cases de la grille.

Un chronomètre est déclenché au niveau de difficulté numéro 2 et le temps de réflexion est alors limité à 15 secondes.

1. Il suffit de déplacer, avec les quatre flèches, le petit repère clignotant jusqu'à ce qu'il se trouve dans la case voulue. A ce moment-là, on appuie sur la barre d'espace.

■ PHASES ESSENTIELLES

Sous-programme 1000

Fabrication de la grille.

1000 à 1010 Un message d'attente indique à l'un des joueurs que son tour est arrivé.

1020 à 1030 Chacune des 24 cases de la grille se voit attribuer une valeur allant de 1 à 24. Si l'on s'arrêtait là, les dessins apparaîtraient toujours au même endroit de l'écran.

1040 à 1050 Les différents graphiques sont mélangés.

1060 à 1110 Une partie du décor est construite.

| Variables | Fonction | Valeurs possibles |
|-----------|---|-------------------|
| R | Numéro du joueur | 1 à 3 |
| A\$(R) | Nom du joueur R | Chaîne caractères |
| B(I,J) | Numéro du dessin contenu dans la case (I,J) | 1 à 24 |
| A(I,J) | Numéro du dessin affiché dans la case (I,J) | 1 à 24 |
| X et Y | Coordonnées d'une première case | 1 à 6 et 1 à 4 |
| Z et T | Coordonnées d'une deuxième case | 1 à 6 et 1 à 4 |
| B | Variable intermédiaire | 1 à 24 |
| S(I) | Score du joueur numéro I | 0 à 24 |
| XP et YP | Coordonnées du repère clignotant | 5 à 30 et 7 à 22 |
| U et V | Numéro de la case dans laquelle se trouve le repère | 1 à 6 et 1 à 4 |
| H | Nombre d'erreurs commises | 0 à 4 |

Sous-programme 2000

Affichage des divers graphismes.

- 2000 à 2010 Les 24 cases de la grille prennent la couleur noire et les images qu'elles contenaient disparaissent.
- 2020 à 2030 Le programme recherche les coordonnées du dessin qui va être ajouté à ceux qui ont déjà été affichés.
- 2040 à 2070 Le tableau A contient soit la valeur 0 (aucun dessin n'est rendu visible), soit une valeur comprise entre 1 et 24 (le dessin correspondant apparaît alors).

| Variables | Fonction | Valeurs possibles |
|-----------|---|-------------------|
| L | Numéro du nouveau dessin | 1 à 24 |
| G\$(X,Y) | Dessin dont le numéro est égal à A(X,Y) | Chaîne caractères |
| D | Temps accordé à un joueur | 15 à 0 |

Sous-programme 3000

Le joueur choisit une case.

- 3000 à 3040 Une phrase se colore en vert puis en jaune et demande au joueur de désigner un dessin. Au niveau 2, le compteur de temps est décrémenté et sa valeur est affichée.
- 3050 à 3060 Si l'une des flèches est enfoncée, le repère clignotant change de place. Lorsque la barre d'espacement est utilisée et que l'on désigne une case vide, la tentative est ignorée par l'ordinateur.
- 3070 à 3090 On a trouvé la solution et le programme nous le fait savoir ; le score est aussitôt augmenté d'une unité.
- 3100 à 3150 Par contre, ici, on a choisi une mauvaise case. Le jeu se poursuit tout de même si l'on a commis moins de quatre erreurs.

| Variables | Fonction | Valeurs possibles |
|-----------|-----------------|-------------------|
| A\$ | Touche enfoncée | Touche quelconque |

| Variables | Fonction | Valeurs possibles |
|-----------|--|-------------------|
| E | Compteur pour l'affichage d'un message | 0 à 80 |
| C | Numéro d'une couleur | 2 ou 3 |
| NV | Niveau de difficulté | 1 ou 2 |
| X et Y | Coordonnées de la bonne case | 1 à 6 et 1 à 4 |

■ MODIFICATIONS POSSIBLES

Augmentation du nombre de dessins différents

L'ordinateur n'aurait pas alors à proposer la même case deux fois de suite.

Augmentation de la taille de la grille

Il s'agirait toujours de retrouver 24 dessins mais le nombre de cases vides serait beaucoup plus élevé, rendant le repérage de la case ajoutée plus difficile.

Modification de l'ordre dans lequel apparaissent les motifs

Ceci est une idée pour rendre ce jeu encore plus difficile.

```

50 '
60 '
70 '          AU   SUIVANT
80 '
90 '
100 MODE 1:INK 0,0:INK 1,10:INK 2,18:INK 3,24:BORDER
    0:PAPER 0:SYMBOL AFTER 203
110 SYMBOL 204,3,3,3,3,1,1,63,63:SYMBOL 205,192,192,1
    92,192,128,128,252,252
120 SYMBOL 206,3,3,3,3,12,12,48,48:SYMBOL 207,192,192
    ,192,192,48,48,12,12
130 SYMBOL 208,4,6,7,7,7,7,7:SYMBOL 209,0,0,0,128,19
    2,224,240,248
140 SYMBOL 210,7,7,4,255,127,63,31,15:SYMBOL 211,252,
    254,0,255,254,252,248,240
150 SYMBOL 212,0,0,0,255,4,4,63,235:SYMBOL 213,0,0,0,2

```

```

24,0,0,0,129
160 SYMBOL 214,127,63,31,0,0,0,0:SYMBOL 215,255,129
,0,0,0,0,0
170 SYMBOL 216,24,56,48,48,24,139,103,31:SYMBOL 217,
24,28,12,12,24,209,230,248
180 SYMBOL 218,7,63,71,135,27,32,64,64:SYMBOL 219,22
4,252,226,225,216,4,2,2
190 SYMBOL 220,15,15,7,3,1,97,145,9:SYMBOL 221,224,22
4,192,128,0,3,4,8
200 SYMBOL 222,5,5,3,1,1,1,1:SYMBOL 223,16,32,32,64,
64,128,0,0
210 SYMBOL 224,32,80,136,136,4,4,2,2:SYMBOL 225,8,20,
36,36,64,64,128,128
220 SYMBOL 226,3,7,5,7,7,3,3:SYMBOL 227,128,192,64,1
92,192,192,128,128
230 SYMBOL 228,15,31,63,127,255,63,63,38:SYMBOL 229,
248,248,252,254,255,252,252,100
240 SYMBOL 230,38,38,38,62,62,62,255,255:SYMBOL 231,
100,100,100,124,124,124,255,255
250 SYMBOL 232,0,0,0,0,0,15,63,255:SYMBOL 233,0,0,0,0,
0,1,194,254
260 SYMBOL 234,255,63,15,0,0,0,0:SYMBOL 235,254,194
,1,0,0,0,0,0
270 SYMBOL 236,60,60,195,195,195,195,3,3:SYMBOL 237,
0,0,0,12,14,15,12,12
280 SYMBOL 238,12,12,48,48,48,48,15,15:SYMBOL 239,60,
63,62,60,252,252,252,248
290 SYMBOL 240,0,0,0,0,0,0,0:SYMBOL 241,14,12,8,24,1
6,48,96,96
300 SYMBOL 242,0,32,31,31,22,16,16,16:SYMBOL 243,192,
192,128,128,128,128,128,128
310 SYMBOL 244,0,0,48,240,48,48,63,31:SYMBOL 245,0,0,
0,0,0,2,252,252
320 SYMBOL 246,13,12,12,12,12,12,0,0:SYMBOL 247,236,1
2,12,12,12,12,0,0
330 SYMBOL 248,0,0,0,1,3,6,255,255:SYMBOL 249,48,96,1
92,128,1,2,252,252
340 SYMBOL 250,6,3,1,0,0,0,0:SYMBOL 251,2,1,128,192,
96,48,0,0
370 K=204:DIM G$(24):FOR I=1 TO 12
380 G$(I)=CHR$(K)+CHR$(K+1)+CHR$(8)+CHR$(8)+CHR$(10)+
CHR$(K+2)+CHR$(K+3)
390 K=K+4:NEXT:PEN 3:LOCATE 2,5:PRINT "NOMBRE DE J

```

```

OUEURS (MAX 3)"
400 A$=INKEY$:IF A$="" THEN X=RND:GOTO 400
410 A=VAL(A$):IF A<1 OR A>3 THEN 400 ELSE PEN 2
420 FOR I=1 TO A:LOCATE 2,6+2*I:PRINT "NOM DU JOU
EUR";I;
430 LINE INPUT A$(I):A$(I)=LEFT$(A$(I),9):NEXT
440 PEN 3:LOCATE 2,20:PRINT "CHOISISSEZ LA DIFFICU
LTE ( 1 OU 2 )"
450 A$=INKEY$:IF A$="" THEN 450 ELSE NV=VAL(A$):IF
NV<1 OR NV>2 THEN 450
460 FOR I=1 TO 12:G$(I+12)=G$(I):NEXT
970 '
980 '      Fabrication de la grille
990 '
1000 PAPER 0:PEN 2:CLS:R=R+1:LOCATE 1,10:PRINT A$(R
);" ....."
1010 PRINT:PRINT "C'EST A VOUS MAINTENANT DE JOU
ER"
1020 K=1:FOR J=1 TO 4:FOR I=1 TO 6:B(I,J)=K:K=K+1
1030 A(I,J)=0:NEXT:NEXT:FOR I=1 TO 40:X=INT(RND*6)+1
1040 Y=INT(RND*4)+1:Z=INT(RND*6)+1:T=INT(RND*4)+1
1050 B=B(X,Y):B(X,Y)=B(Z,T):B(Z,T)=B:NEXT:CLS:PEN 1:FOR I
=1 TO A
1060 LOCATE 13*I-12,25:PRINT A$(I);LOCATE 13*I-3,25
:PRINT S(I);
1070 NEXT:LOCATE 33,10:PRINT "ERREURS":LOCATE 33,1
4
1080 PRINT "(MAXI:3)":PAPER 1:WINDOW 34,40,3,5:CLS
1090 PEN 2:LOCATE 1,2:PRINT "SUIVANT":WINDOW 1,31,3
,23:CLS
1100 PAPER 0:WINDOW 1,40,1,25:PEN 3:LOCATE 13*R-12,
25:PRINT A$(R);
1110 XP=15:YP=12:U=3:V=2:L=0:C=2:H=0
1970 '
1980 '      Affichage des dessins
1990 '
2000 L=L+1:PAPER 0:FOR J=4 TO 19 STEP 5:FOR I=2
TO 27 STEP 5
2010 WINDOW I,I+3,J,J+3:CLS:NEXT:NEXT:WINDOW 1,40,1,2
5
2020 FOR I=1 TO 500:NEXT:FOR Y=1 TO 4:FOR X=1 TO
6
2030 IF B(X,Y)<>L THEN NEXT:NEXT ELSE A(X,Y)=L

```

```

2040 PEN 2:FOR Y=1 TO 4:FOR X=1 TO 6
2050 IF A(X,Y) THEN LOCATE 5*X-2,5*Y:PRINT G$(A(X,Y)
)
2060 NEXT:NEXT:D=15
2070 IF INKEY$<>" " THEN 2070
2970 '
2980 '      Le joueur choisit une case
2990 '
3000 A$=INKEY$:IF A$<>" " THEN 3050 ELSE E=E+1:IF
E<80 THEN 3000
3010 E=0:C=5-C:PEN C:PAPER 0:LOCATE 3,1
3020 PRINT "DESIGNEZ LE DESSIN NOUVEAU":PEN C-1
3030 LOCATE XP,YP:PRINT "+":IF NV=1 THEN 3000 EL
E D=D-1
3040 PEN 2:PAPER 0:LOCATE 35,20:PRINT D:IF D>0 THE
N 3000 ELSE 3100
3050 IF A$<>CHR$(32) THEN GOSUB 4000:GOTO 3000
3060 IF A(U,V)=0 THEN 3000 ELSE IF A(U,V)<>L THEN 3
100 ELSE PAPER 1
3070 FOR I=0 TO 13:PEN I:LOCATE XP-3,YP+1:PRINT "G
AGNE":SOUND 1,150-6*I,2
3080 FOR J=1 TO 10:NEXT:NEXT:S(R)=S(R)+1:PEN 1:PAPE
R 0
3090 LOCATE 13*R-3,25:PRINT S(R):IF L<24 THEN 2000
ELSE 5000
3100 PAPER 1:FOR I=0 TO 13:PEN I:LOCATE XP-3,YP+1
3110 PRINT "PERDU":SOUND 1,300+10*I,2:FOR J=1 TO
50:NEXT:NEXT
3120 H=H+1:PAPER 0:LOCATE 35,12:PRINT H:PEN 1:PAPE
R 0
3130 FOR X=1 TO 6:FOR Y=1 TO 4:IF A(X,Y)<>L THEN N
EXT:NEXT
3140 LOCATE 5*X-2,5*Y:PRINT G$(L):FOR I=1 TO 2000:
NEXT
3150 IF H<4 THEN 2000 ELSE 5000
3970 '
3980 '      Deplacement repere
3990 '
4000 AS=ASC(A$):IF AS<240 OR AS>243 THEN RETURN
4010 PAPER 0:LOCATE XP,YP:PRINT " "
4020 IF AS=242 AND U>1 THEN XP=XP-5:U=U-1
4030 IF AS=243 AND U<6 THEN XP=XP+5:U=U+1
4040 IF AS=240 AND V>1 THEN YP=YP-5:V=V-1

```

```

4050 IF AS=241 AND V<4 THEN YP=YP+5:V=V+1
4060 PEN 2:LOCATE XP,YP:PRINT "+":RETURN
4970 '
4980 '     Fin du jeu
4990 '
5000 PEN 1:PAPER 0:LOCATE 13*R-12,25:PRINT A$(R);I
F R<A THEN 1000
5010 IF INKEY$<>"" THEN 5010 ELSE PAPER 0
5020 A$=INKEY$:IF A$="R" THEN RUN ELSE IF A$<>""
THEN CLS:END
5030 FOR I=1 TO 100:NEXT:PEN INT(RND*3)+1:LOCATE 3
,1
5040 PRINT "UNE AUTRE PARTIE : TOUCHE R":GOTO 50
20

```

POUR UN CATALOGUE COMPLET DE NOS PUBLICATIONS

FRANCE

6-8, Impasse du Curé
75881 PARIS CEDEX 18
Tél. : (1) 42.03.95.95
Télex : 211801

U.S.A.

2021 Challenger Drive, 100
Alameda, CA 94501
Tél. : (415) 523-8233
Télex : 336311

ALLEMAGNE

Vogelsanger. Weg 111
4000 Düsseldorf 30
Tel. : (211) 61.80.2-0
Telex : 8588163



Paris • San Francisco • Düsseldorf • Londres

Cet ouvrage contient 20 programmes de jeux de réflexion écrits pour les ordinateurs Amstrad CPC 464, 664, 6128. De thèmes classiques, comme *le pendu* ou *les tours de Hanoi* ou résolument nouveaux, comme *le Mistigri* ou *la mare aux diables*, ils permettent de jouer seul ou à plusieurs, avec l'aide de l'ordinateur ou contre lui.

Chaque programme est accompagné d'une analyse pédagogique de la structure des phases essentielles et de tableaux résumant la fonction et les valeurs possibles des principales variables. De quoi comprendre, en s'amusant, comment se réalisent des programmes de jeux.

L'AUTEUR

GEORGES FAGOT-BARRALY, professeur de mathématiques et d'informatique, est spécialiste de l'utilisation des micro-ordinateurs dans l'enseignement. Il a écrit plusieurs livres de jeux et il est également l'auteur de nombreuses publications sur la programmation en assembleur.





ANVÄNDRINGEN AV REFLEKTERADE LÖSEN



Document numérisé avec amour par

AMSTRAD

CPC 

MÉMOIRE ÉCRITE



<https://acpc.me/>